
Введение
Язык программирования Паскаль (Pascal) в настоящее время следует рассматривать как учебное средство, позволяющее, при своем простом синтаксисе, сосредоточиться на алгоритмической стороне программирования, не вдаваясь в детали разработки сложных пользовательских интерфейсов и структур данных. Таким образом, пособие рассчитано на базовый курс подготовки по основам программирования и тех студентов, для которых оно не является основной специальностью, и начинающих студентов-программистов. Оно может также оказаться полезно инженерам и специалистам, решившим познакомиться с искусством программирования.
Пособие отражает, в основном, мой личный опыт преподавания основ программирования и не претендует на "всеохватность". Так, в нем практически не затронуты работа с динамической памятью, объектно-ориентированное программирование и ряд других тем, без которых профессиональное владение любым языком едва ли возможно.
Курс построен из двух частей, содержание которых примерно соответствует двухсеместровой программе изучения основ программирования из расчета одной пары в неделю.

Пособие содержит как лекционный материал, так и большое количество исходных текстов программ-примеров. Рекомендуемая среда для работы с примерами -- Turbo Pascal 7.1 , скачать компактный дистрибутив которого можно, например, по ссылке http://pers.narod.ru/distr/TP71Setup.zip (770 Кб, внутри установщик и readme.txt). Вполне подойдёт и бесплатный Free Pascal, очень похожий в управлении и по возможностям, но являющийся полноценным 32-разрядным приложением. Буду признателен за письма с Вашими отзывами, пожеланиями и замечаниями. Все авторские права принадлежат мне, однако я не возражаю против свободного использования этого учебника в образовательных целях при условии сохранения его текста неизменным. 
Часть 1. Основы языка Паскаль

Первая часть пособия предназначена для изучения базовых приемов программирования. Она последовательно знакомит читателя с азами программистского искусства, начиная с простейших расчетов, состоящих из нескольких формул, и заканчивая применением циклов и массивов.

1. Алгоритм и программа

Написанию программы всегда предшествует разработка некоторого плана (алгоритма) решения задачи. Кратко опишем основные понятия теоретической информатики, связанные с алгоритмами.
1.1. Алгоритм

Алгоритм -- это однозначно определенная последовательность действий, записанная на понятном исполнителю алгоритмическом языке и определяющая процесс перехода от исходных данных к результату.

В этом определении уже указаны основные свойства алгоритма. Во-первых, алгоритм состоит из конечного набора инструкций или шагов, во-вторых, каждый шаг трактуется исполнителем единственным образом, что позволяет гарантированно получить решение для некоторого набора входных данных, в-третьих, алгоритм всегда сводится к некоторому преобразованию исходных данных в результат или результаты. В этом смысле формулы для решения квадратного уравнения или даже четко составленную инструкцию по варке кофе можно считать алгоритмами, выполнимыми исполнителем-человеком. Для машины, разумеется, требуется более четкая формализация задачи, чем для человека, понимать естественный язык компьютеры пока неспособны, отсюда необходимость учета при составлении алгоритма ограниченного набора инструкций ЭВМ. 
1.2. Свойства алгоритма

Перечислим выделенные в п. 1.1 свойства алгоритма.

· Дискретность -- алгоритм состоит из отдельных инструкций (шагов).
· Однозначность  -- каждый шаг понимается исполнителем единственным образом.
· Массовость -- алгоритм работает при меняющихся в некоторых пределах входных данных.
· Результативность -- за конечное число шагов достигается некоторый результат.
1.3. Формы записи алгоритма

Принято выделять 2 основных формы записи алгоритма.

Графическая форма записи (блок-схема) характерна тем, что отдельные шаги алгоритма изображаются геометрическими фигурами, а последовательность выполнения шагов -- связями между этими фигурами. На рис. 1.1 указаны основные элементы блок-схем.

Рис. 1.1. Основные элементы блок-схем
Указанные на рис. 1.1 геометрические фигуры интерпретируются так:

Прямоугольник -- любая последовательность действий; внутри прямоугольника записываются формулы или словесное описание выполняемых действий;

Ромб -- блок проверки условия; так как любое условие может быть только истинно или ложно, у блока 1 вход и 2 выхода, соответствующие действиям, выполняемым в случаях, когда условие истинно и когда оно ложно. Выходы подписывают символами "+" и "-", "да" и "нет", "1" и "0" и т. п.

Параллелограмм -- блок ввода исходных данных. Внутри фигуры обычно пишется, какие именно данные должны быть введены;
Лист с разрывом -- блок вывода данных. Внутри блока указывается, какие данные или сообщения программа выводит для представления пользователю;
Закругленные прямоугольники -- необязательные блоки начала и конца программы, внутри блоков обычно указываются ключевые слова "нач" и "кон" соответственно;
Последняя фигура служит для изображения циклов, как правило, у нее 2 входа (первый и повторный вход в цикл) и 1 выход, соответствующий завершению циклического процесса.

На рис. 1.2 приведен пример блок-схемы, иллюстрирующей известный процесс решения квадратного уравнения.
Язык блок-схем довольно громоздок, как правило, он не применяется профессионалами, однако, на начальном этапе обучения программированию планирование несложных программ в виде блок-схем может оказаться весьма полезным.

Текстовая форма записи алгоритма (псевдокод) характерна тем, что шаги алгоритма и последовательность их выполнения задаются с помощью набора специальных ключевых слов. Эта форма ближе к реальным языкам программирования. Существует много различных вариантов псевдокода, например, в русскоязычной литературе по программированию распространен следующий вариант псевдокода:

· нач
 -- начало программы;
· кон 
 -- конец программы;
· если ... то ...иначе  -- проверка условия;
· ввод  -- ввод данных;
· вывод  -- вывод данных;
· для ... от .. до ... нц ... кц
 -- цикл со счетчиком (нц -- начало цикла, кц -- конец);
· пока ... нц ...кц
 -- цикл с предусловием;
· нц ... кц ... пока
 -- цикл с постусловием.

Рис. 1.2. Блок-схема решения квадратного уравнения

Все алгоритмические конструкции, соответствующие ключевым словам псевдокода, будут изучены нами в этом пособии. Как правило, программисты используют элементы псевдокода при планировании частей своей будущей программы.
1.4. Программа и программное обеспечение

Программа -- это реализация алгоритма на конкретном языке программирования. Совокупность существующих программ образует программное обеспечение (ПО). ПО принято делить на 2 вида.
Системное ПО обеспечивает работу компьютера и внешних устройств, а также поддержку прикладных программ. Оно разрабатывается квалифицированными специалистами на машинно-ориентированных языках программирования, дающих доступ к аппаратуре компьютера. Примерами системного ПО могут служить операционная система Windows (или любая другая), драйверы внешних устройств компьютера, утилиты для его технического обслуживания, системы программирования, применяемые для разработки собственных приложений.

Прикладное ПО предназначено для решения конкретных задач пользователя. Оно разрабатывается на языках высокого уровня, облегчающих процесс программирования за счет множества готовых решений. К одному из таких языков относится и Паскаль, который мы будем изучать.
1.5. Этапы разработки программы

Разработка любой программы, от несложной учебной задачи до профессионального приложения, может быть разбита на ряд этапов. Кратко опишем и охарактеризуем их.

1. Определение входных и выходных данных, требований к программе.

На первом этапе определяются входные и выходные данные программы, способ ее взаимодействия (интерфейса) с пользователем, язык и среда программирования, в которой она будет разрабатываться, а также требования к аппаратному и системному программному обеспечению компьютеров, на которых будет работать приложение.
2. Разработка алгоритма.

На этом шаге производится определение последовательности действий, ведущих к решению задачи и запись их в одной из указанных в п. 1.3 форм.
3. Кодирование (программирование).

Третий этап -- это перевод алгоритма на язык программирования и создание исходного текста программы в одной из систем программирования. Программа на любом языке состоит из операторов -- так называются отдельные действия, разрешенные в языке. Число операторов в любом языке ограничено и правила их написания жестко заданы.
4. Компиляция и отладка.

Исходный текст на Паскале не будет непосредственно исполняться компьютером -- для работы программы ее требуется откомпилировать, т. е., перевести в машинный код. Эту работу выполняет специальная  программа-компилятор или оболочка языка. Оболочка Паскаля, с помощью которой мы будем разрабатывать свои программы, называется Turbo Pascal 7.1, она разработана компанией Borland International в 1983-97 гг. В результате преобразования компилятором исходного текста программы в машинный код получается исполняемый файл с расширением exe, который можно запустить (выполнить) в той операционной системе (ОС), для которой разработан компилятор. Наша оболочка Паскаля создавалась для ОС MS-DOS, однако, в современных ОС семейства Windows программа, написанная на Паскале, работать все же будет, правда, без удобных интерфейсных возможностей Windows.

Итак, компиляция -- это процесс преобразования программы в машинный код. Программа, которую удалось откомпилировать, не обязательно работает правильно. Она может содержать ошибки, для выявления которых предназначен этап отладки -- поиска ошибок в программе. Как правило, компиляция и отладка выполняются программистом в тесной взаимосвязи.

Возможны программные ошибки трех видов:
· синтаксические (ошибки в правилах языка);

· алгоритмические (ошибки в логике программы);

· ошибки времени исполнения, возникающие в процессе работы запущенной программы.
Компилятор способен найти только синтаксические ошибки, для выявления же алгоритмических ошибок служит этап тестирования программы. Ошибки времени исполнения возникают как результат некорректных действий пользователя, недопустимых операций над данными (например, попытки извлечь квадратный корень из отрицательного числа, поделить на ноль) или ошибок программного и аппаратного обеспечения ЭВМ. Об их обработке будет рассказано в гл. 8.

5. Тестирование.

Тестированием называют проверку правильности работы программы на наборах "пробных" (тестовых) данных с заранее известным результатом. Конечно же, тестирование всей программы сразу возможно лишь для несложных учебных задач. Реальные программы, как правило, тестируются "по частям" -- отдельными функциями и модулями.

6. Документирование и поддержка.

Этот этап включает в себя создание справочной системы и документации к программе, возможно, расширение ее функциональности, выпуск новых версий, исправление ошибок, которые практически неизбежны в любой сложной программной системе. В наших учебных задачах этап поддержки будет отсутствовать. 
2. Данные в языке Паскаль

Любая программа выполняет над исходными данными некоторые расчеты. При этом, как и переменные или константы в математике, отдельные элементы данных обозначаются даваемыми программистом именами (идентификаторами). Любые идентификаторы в языке Паскаль строятся по следующим правилам:

· имена могут включать латинские буквы, цифры и знак подчеркивания (для простоты опустим некоторые другие символы, разрешенные в именах);

· имя состоит из одного слова; если требуется пробел в имени, он заменяется на подчеркивание: так, My_1 будет правильным идентификатором, а My 1 -- нет;

· имя всегда начинается с буквы: возможен объект с именем A1, но не 1A; прописные и строчные буквы в именах не различаются Паскалем: x1 и X1 -- это одна и та же величина;

· имена не могут совпадать с зарезервированными в языке служебными словами, обозначающими разрешенные в языке операции над данными: например, нельзя назвать Begin или BEGIN ни одну величину в программе, так как begin --зарезервированное служебное слово, а прописные и строчные буквы в служебных словах также не различаются. Познакомиться с большинством служебных слов мы сможем в процессе изучения языка.

2.1. Константы

Константой называют величину, значение которой не меняется в процессе выполнения программы.

Числовые константы служат для записи чисел. Различают следующие их виды:

Целочисленные (целые) константы: записываются со знаком + или -, или без знака, по обычным арифметическим правилам:

-10

+5

5

Вещественные числа могут записываться в одной из двух форм:

обычная запись: 2.5

-3.14

2.
 -- обратите внимание, что целая часть отделяется от дробной символом точки;
экспоненциальная ("научная") форма: в этой записи вещественное число представляется в виде m*10p, где m -- мантисса или основание числа, 0.1≤|m|≤1, p -- порядок числа, заданный целочисленной константой. Действительно, любое вещественное число можно представить в экспоненциальной форме:

-153.5

-0.1535*103
99.005

0.99005*102
Во всех IBM-совместимых компьютерах вещественные числа хранятся как совокупность мантиссы и порядка, что позволяет упростить операции над ними, используя специальную арифметику, отдельно обрабатывающую мантиссу и порядок. Для программной записи числа в экспоненциальной форме в качестве обозначения "умножить на 10 в степени" используется символ E или e (латинские):

-153.5
 -0.1535*103
-0.1535E3 или -1.535E02

99.005
 0.99005*102
0.99005E+2 или 9.9005e+01

Без принятия специальных мер, программа на Паскале будет выводить на экран и принтер вещественные числа именно в такой форме. Кроме того, такая форма удобна для записи очень маленьких и очень больших чисел:

1030

1e30

-1020

-1E20

10-30

1E-30

Поскольку размер памяти, отводимой под мантиссу и порядок, ограничен, то вещественные числа всегда представляются в памяти компьютера с некоторой погрешностью. Например, простейшая вещественная дробь 2/3 дает в десятичном представлении 0,666666... и, независимо от размера памяти, выделяемой для хранения числа, невозможно хранить все его знаки в дробной части. Одной из типичных проблем программирования является учет возможных погрешностей при работе с вещественными числами.

Кроме числовых констант существуют и другие их виды:

логические константы служат для проверки истинности или ложности некоторых условий в программе и могут принимать только одно из двух значений: служебное слово true обозначает истину, а false -- ложь;

символьные константы могут принимать значение любого печатаемого символа и записываются как символ, заключенный в апострофы ('одинарные кавычки'):

'Y'   
'я'
    ' '

В последнем случае значение символьной константы равно символу пробела. Если требуется записать сам символ апострофа как символьную константу, внутри внешних апострофов он удваивается: ''''.
К символьным также относятся константы вида #X, где X -- числовое значение от 0 до 255 включительно, представляющее собой десятичный ASCII-код символа. Таблицы ASCII-кодов, используемых операционными системами DOS и Windows, приведены в Приложении 1. Например, значение #65 будет соответствовать коду буквы 'A' латинской. Обработка нажатий клавиш и расширенные коды клавиатуры описаны в гл. 24 и Приложении 5.

Строковые константы -- это любые последовательности символов, заключенных в апострофы. Как правило, строковые константы служат для записи приглашений к вводу данных, выдаваемых программой, вывода диагностических сообщений и т. п.:

'Введите значение X:'

'Ответ='

Если в строковой константе необходимо записать сам символ апострофа, это делается так же, как для символьных констант.

Именованные константы перечисляются в разделе описаний программы оператором следующего вида:

const Имя1=Значение1;

 Имя2=Значение2;

 ...

 ИмяN=ЗначениеN;

Ключевое слово const показывает начало раздела описаний именованных констант. Ясно, что зачастую удобнее обращаться к константе по имени, чем каждый раз переписывать ее числовое или строковое значение. Приведем пример раздела:

const e=2.7182818285;

  lang='Turbo Pascal 7.1';

Здесь описана числовая константа e со значением основания натурального логарифма и строковая константа с именем lang, содержащая строку 'Turbo Pascal 7.1'. 
Каждое даваемое программистом имя должно быть уникальным в пределах одной программы. Если мы включим этот раздел в свою программу, мы уже не сможем создать в ней других объектов с именами e и lang.
2.2. Переменные и типы переменных

Переменными называют величины, значения которых могут изменяться в процессе выполнения программы. Каждая переменная задается своим уникальным именем, построенным по правилам, указанным в начале главы. Максимально возможная длина имени зависит от реализации Паскаля, теоретически можно давать переменным имена вплоть до 63 символов длиной, что едва ли актуально -- обычно имена не длиннее 5-10 символов.

Поскольку любые данные в памяти компьютера хранятся в числовой форме и двоичной системе счисления, кроме имени, переменной обязательно следует присвоить и тип, определяющий диапазон значений, принимаемых переменной, и способ ее обработки машиной. Поясним сказанное на примере. Как видно из Приложения 1, большая латинская буква 'A' имеет десятичный код 65, или 01000001 в двоичном представлении. Без дополнительной информации о типе данных, хранящихся в некоторой ячейке памяти, компьютеру было бы невозможно решить, что именно представляют из себя эти данные -- число 65, код символа 'A' или что-то еще. В любом языке программирования, в том числе и в Паскале, существует стандартный набор типов, к которым может быть отнесена та или иная совокупность ячеек памяти. Информацию о типах данных Паскаля удобно свести в таблицу. Строки этой таблицы будут упорядочены по старшинству типов, от самого "младшего", требующего наименьшее число байт для представления, и, соответственно, представляющего наименьший диапазон возможных значений, до самого "старшего", представляющего наибольший диапазон значений. В табл. 2.1 представлены не все возможные, а лишь основные типы данных Паскаля.
Табл. 2.1. Основные типы данных Паскаля

	Ключевое слово Паскаля
	Название и описание типа
	Объем памяти, байт
	Диапазон возможных значений

	boolean
	Логический: хранит одну логическую переменную
	1
	true и false

	char
	Символьный: хранит код одного  символа из набора ASCII-кодов
	1
	от 0 до 255 включительно (28=256)

	integer
	Целочисленный
	2
	±215

	word
	Целочисленный без знака
	2
	±216 - диапазон вдвое больше, так как 16-й бит не занят под знак числа

	longint
	Длинное целое: для представления больших целочисленных значений
	4
	±231

	real
	Вещественное число с точностью представления до 11-12 знака в дробной части
	6
	~ 2.9*10-39 - 1.7*1038

	double
	Вещественное число с точностью представления до 15-16 знака в дробной части
	8
	~ 5*10-324 – 1.7*10308

	string
	Последовательность символов типа char длиной от 1 до 255
	2-256 (данные строки + 1 байт для хранения ее длины)
	Любые строки текста, состоящие из печатаемых символов


Теоретически для записи переменной типа boolean было бы достаточно 1 бита, но минимальная адресуемая единица памяти -- 1 байт (см. Приложение 1). В этом же Приложении уточните, как именно объем памяти в байтах, выделяемой под переменную, влияет на диапазон представляемых ей значений. 
Целочисленные и символьные типы обобщенно называют порядковыми, подчеркивая этим, что данные типы имеют конечный набор значений, которые могут быть упорядочены или перечислены. Напомним, что вещественные значения хранятся в памяти компьютера иначе, чем целые -- а именно, как совокупность мантиссы и порядка. 
Задача правильного выбора типов данных целиком ложится на программиста. Например, если некоторый счетчик в вашей программе может принимать целочисленные значения от 1 до 100 000, неправильно было бы описывать его как переменную типа integer – ведь 215=32768 и при достижении счетчиком этой величины произойдет сброс его значения, которое станет равно -32768. Разумным в данном случае было бы описание счетчика как переменной типа longint.

Переменные описываются в программе оператором следующего вида:

var
 Список1:Тип1;


 Список2:Тип2;


 ...


 СписокN:ТипN;

Здесь список -- набор имен переменных, разделенных запятыми (или одна переменная), а тип -- любой из рассмотренных выше типов данных. Например, конструкция

var t,r:real;

 i:integer;

описывает 2 вещественных переменных с именами t и r, а также целочисленную переменную с именем i. Ключевое слово var можно продублировать, но обычно такой необходимости нет. Сокращение var образовано от английского "variable" (переменная).
3. Арифметические выражения

Арифметические выражения (АВ) строятся из операндов, которыми могут быть константы, переменные и стандартные функции. В АВ также входят арифметические операции и круглые скобки. В языке Паскаль определено 6 арифметических операций, перечислим их в соответствии с приоритетом, то есть, старшинством (табл. 3.1). Операции с одинаковым приоритетом равноправны между собой и выполняются слева направо, как и в математике.
Табл. 3.1. Арифметические операции языка Паскаль

	Приоритет
	Знак операции
	Описание операции

	1
	*
	умножение

	
	/
	деление

	
	div
	деление 2 целых значений с отбрасыванием остатка

	
	mod
	взятие остатка от деления 2 целых значений

	2
	+
	сложение

	
	-
	вычитание


Операции div и mod определены только для целочисленных операндов. Приведем пример их использования:

var y,c,n:integer;

. . .

y:=2009;

c:=y div 100;

n:=y mod 100;

Здесь переменная c получит значение 20, а n -- значение 9.

Примеры арифметических выражений мы приведем после рассмотрения стандартных функций языка Паскаль.

Стандартные функции служат для выполнения элементарных математических расчетов, часто требуемых при написании программ. Разработчики Паскаля стремились сделать его программное ядро максимально компактным, поэтому в него не вошел ряд функций, обычно имеющихся в других языках, таких, как вычисление максимума и минимума, возведение числа в произвольную степень и др. Физически коды стандартных функций хранятся в стандартной библиотеке Паскаля -- файле с именем turbo.tpl. Все функции оформляются одинаково: после имени функции следует ее аргумент, заключенный в круглые скобки. Если аргументов несколько, они разделяются запятыми. Информация об основных стандартных функциях представлена в табл. 3.2. 
Табл. 3.2. Стандартные функции языка Паскаль

	Математи-ческая запись
	Запись на Паскале
	Пояснение
	Тип

аргумента и результата 

	|x|
	abs(x)
	Модуль аргумента x
	Integer (I) или Real (R)

	x2
	sqr(x)
	Квадрат аргумента x
	аргумент – I или R, результат – r

	sin x
cos x
arctg x
	sin(x)

cos(x)

arctan(x)
	Остальные тригонометрические функции выражаются через эти
	аргумент – I или R, результат – R

	ex
ln x
	exp(x)

ln(x)
	Экспонента и натуральный логарифм
	аргумент – I или R, результат – R

	sqrt(x)
	Квадратный корень от аргумента x
	аргумент – I или R, результат – R

	(
	pi
	Функция без аргументов, вернет число (
	R

	
	trunc(x)
	Функция отбрасывает дробную часть аргумента, аргумент не округляется
	аргумент R, результат I

	
	frac(x)
	Функция выделяет

дробную часть своего 
аргумента
	R

	
	round(x)
	Округление вещественного числа до ближайшего целого
	аргумент R, результат I


В табл. 3.2 x обозначает любую подходящую по типу переменную, либо результат вычисления выражения соответствующего типа, либо соответствующий по типу результат, вычисленный другой стандартной функцией. 
Приведем примеры арифметических выражений.
1. Возвести величину x в пятую степень. Выражение может быть записано как x*x*x*x*x или  sqr(x)*sqr(x)*x или sqr(sqr(x))*x, последнее показывает, что результаты одних функций могут быть аргументами других -- это называют вложением функций. Разумеется, тип результата, возвращаемый вложенной функцией, должен быть подходящим для аргумента внешней функции.

2. Возвести величину a в произвольную степень x. Так как в Паскале нет функции возведения в произвольную степень, воспользуемся формулой ax=ex*ln a:
a:=2.5; x:=0.25;

ax:=exp(x*ln(a));

Обратите внимание, что все круглые скобки в выражении должны быть парными. Другой пример применения этого способа: [image: image4.wmf]3

3. Вычислить sin2x . Запись на Паскале: sqr(sin(x)). Сравните с выражением sin x2, которое записывается как sin(sqr(x)).

4. Вычислить k=tg(t). Т. к. функции тангенса в Паскале нет, распишем тангенс в виде k:=sin(t)/cos(t);.
5. При необходимости изменить обычное старшинство операций в записи выражения используются дополнительные круглые скобки. Например, правильная запись выражения 

 выглядит как y:=(a+b)/2;. Запись y:=a+b/2; неверна, т. к. это означает 

6. В записи выражений нельзя пропускать знак *, как часто делается в математике: b2-4ac записывается как sqr(b)-4*a*c. Нельзя писать sin*x или sin x, после имени функции может следовать только ее аргумент в круглых скобках.

Тип выражения определяется старшим из типов входящих в него операндов (т. е., стандартных функций, переменных, констант). Старшинство типов можно определить по табл. 2.1, напомню, что в первой строке таблицы находится самый младший тип. 
Например, для переменных
var i,j:integer; f:real;

выражение i+4*j имеет целый тип, и его результат можно записать в целую переменную. Выражение f+i*0.5 дает вещественный, результат, который должен быть записан в вещественную переменную.
Операция деления / в Паскале всегда дает вещественное число. Для деления целых чисел с целым результатом (остаток отбрасывается) используйте div, для взятия остатка от деления двух целых -- mod.
4. Линейный вычислительный процесс

Линейный вычислительный процесс (ЛВП) представляет собой набор операторов, выполняемых последовательно, один за другим. Основу программы ЛВП составляют операторы присваивания, ввода и вывода данных.
4.1. Оператор присваивания

Оператор присваивания используется для сохранения результата вычисления арифметического выражения в переменной. Он имеет следующий общий вид:

переменная := выражение;

Знак := читается как "присвоить".

Оператор присваивания работает следующим образом: сначала вычисляется выражение, стоящее справа от знака :=, затем результат записывается в переменную, стоящую слева от знака. Например, после выполнения оператора

k:=k+2;

текущее значение переменной k увеличится на 2.

Тип переменной слева от знака присваивания должен быть не младше типа выражения. В частности, это означает, что если выражение дает целое число, результат можно писать и в целую, и в вещественную переменную, если результат вычисления выражения вещественный, писать его в целую переменную нельзя, т. к. может произойти потеря точности.

Приведем примеры.
1. Записать оператор присваивания, который позволяет вычислить расстояние между двумя точками на плоскости с координатами (x1, y1) и (x2, y2).
Оператор будет иметь вид 
d:=sqrt(sqr(x1–x2)+sqr(y1–y2));

2. Записать последовательность операторов присваивания, обеспечивающих обмен значениями переменных x и y в памяти компьютера.
c:=x; x:=y; y:=c;

Здесь с -- дополнительная переменная того же типа, что x и y, через которую осуществляется обмен. Грубой ошибкой было бы, например, попытаться выполнить обмен операторами x:=y; y:=x; -- ведь уже после первого из них мы имеем два значения y, а исходное значение x потеряно.
4.2. Оператор ввода

Базовая форма оператора ввода позволяет пользователю ввести с клавиатуры значения одной или нескольких переменных. Оператор ввода с клавиатуры может быть записан в одной из следующих форм:
read(список_переменных);

readln(список_переменных);

Имена переменных в списке перечисляются через запятую. Здесь и далее список данных, передаваемых любому оператору (а позднее и написанным нами подпрограммам), мы будем называть параметрами. Таким образом, параметрами оператора (точней, стандартной процедуры) read являются имена переменных, описанных ранее в разделе var.

По достижении оператора ввода выполнение программы останавливается и ожидается ввод данных пользователем. Вводимые значения переменных разделяются пробелом или переводом строки (нажатием Enter). После ввода значений всех переменных из списка работа программы продолжается со следующего оператора. 
Оператор readln отличается от read только тем, что все переменные должны быть введены в одну строку экрана, клавиша Enter нажимается один раз по окончании ввода. Форма записи readln используется, в основном, для ввода строк текста, для ввода числовых значений лучше использовать read, т. к. в этом случае пользователь может вводить данные более свободно (и в одну, и в несколько строк экрана). 
Если пользователь вводит данные недопустимого типа (например, строку текста вместо числа), то выводится системное сообщение об ошибке и работа программы прерывается.

В качестве примера организуем ввод исходных данных для решения квадратного уравнения:
var a,b,c:real;

...

read (a,b,c);

Для задания значений a=1, b=4, c=2.5 на экране вводится:

1_4_2.5¬

Здесь и далее "_" означает пробел, а "¬" -- нажатие Enter. Другой вариант ввода с клавиатуры:
1¬
4¬

2.5¬

Третий вариант:
1¬

4_2.5¬

Во всех вариантах пробелов может быть и несколько, лишние будут проигнорированы оператором.
Как правило, перед оператором ввода ставится оператор вывода, служащий приглашением к вводу и поясняющий пользователю, что именно следует сделать (см. п. 4.3).
4.3. Оператор вывода

Базовая форма оператора вывода позволяет отобразить на экране значения переменных, АВ или констант, а также строки текста в апострофах. Оператор записывается в одной из следующих форм:
write(список);

writeln(список);

Элементы списка перечисляются через запятую.
Элементы списка выводятся в пользовательское консольное окно программы. Вещественные значения выводятся в экспоненциальной форме. Строки выводятся "как есть". После вывода работа программы продолжается со следующего оператора.

Оператор writeln отличается от write тем, что после вывода значения последнего элемента списка выполняется перевод курсора на следующую строку экрана.

Приведем примеры. 
1. Нужно дать пользователю возможность ввести с клавиатуры число, затем программа возведет это число в квадрат и выведет результат на экран.

var a,a2:integer;

...

writeln ('Введите целое число:');
{это приглашение к вводу}

read (a);

a2:=sqr(a);

writeln ('Квадрат числа=',a2);
Если ввести значение a=2, на экране будет напечатано

Квадрат числа=4

|

Символ |здесь и далее обозначает курсор. Видно, что оператор writeln перевел курсор на следующую строку.
После вывода результата выполнение программы продолжится, а если оператор writeln был в ней последним, то и завершится. Чтобы пользователь успел прочитать результат, следует в конце программы добавить оператор

readln;

который будет ждать нажатия клавиши Enter.

2. Требуется вывести на экран результаты решения квадратного уравнения: значения x1=1.5 и x2=2.5:
write ('x1=',x1,'_x2=',x2);

Пробел в строкой константе '_x2=' нужен, чтобы значение x1 не слилось со строкой 'x2='. На экране будет напечатано:

x1= 1.5000000000E+00 x2= 2.5000000000E+00|

Курсор остался в конце строки, т.к. использована форма оператора write.
Вещественные числа читать в подобной форме неудобно, для их вывода используйте решение из следующего раздела.

4.4. Управление выводом данных

В операторе write или writeln вещественное значение (а также целое или строковое) зачастую удобнее записывать в виде:

переменная:ширина:точность

Здесь ширина -- целое положительное число, определяющее, сколько экранных позиций отводится для вывода всего числа. Ширина определена для числовых значений любого типа и строк.

Точность -- целое положительное число, определяющее, сколько цифр из ширины отводится на вывод дробной части числа. Значение точности определено только для вещественных чисел. Оно не учитывает позицию десятичной точки. Разумные значения точности -- от 0 до ширина-2 включительно. Недопустимые значения ширины и точности не будут учтены при выводе.

В качестве примера выведем на экран значения нескольких переменных:
var x1,p:real;

i:integer;

...

x1:=2.5; p:=-3.175; i:=2;

writeln ('x1=',x1:8:2,'_p=',p:9:4);

write ('I=','_':5,i:2);

На экране будет напечатано:

x1=____2.50_p=__-3.1750

I=______2

4.5. Вывод на печать

Иногда требуется, чтобы программа вывела результаты своей работы на принтер. Для этого достаточно выполнения двух условий. Первым оператором раздела описаний программы следует указать оператор uses printer;, подключающий стандартную библиотеку для работы с принтером, а первым параметром оператора write или writeln указать символическое имя принтера lst, описанное в библиотеке printer:

write ('Hello');

— строка 'Hello' выведена на экран,

write (lst,'Hello');
— строка выведена на принтер.
Отличие между write и writeln сохраняется при выводе на принтер -- то есть, при использовании writeln позиция печати на принтере будет переведена на следующую строку.

Здесь не приводится код, позволяющий проверить, готов ли принтер к печати и удалась ли операция вывода данных на него. Подобные проверки мы научимся делать, изучив стандартную функцию IoResult.
5. Структура простой программы на Паскале

Программа на Паскале не просто состоит из операторов -- порядок следования этих операторов не случаен и образует определенную структуру. Структура простейшей программы описана в табл. 5.1.
Табл. 5.1. Структура простой программы на Паскале

	Название раздела
	Операторы раздела

	Заголовок программы (необязателен)
	program ИмяПрограммы;

	Раздел описаний – необязателен, но, как правило, присутствует
	const список констант;
var список переменных;

	Тело программы – обязателен, содержит операторы программы
	begin

 операторы;

end.


Пара операторов begin и end называется операторными скобками, они служат для того, чтобы объединить группу операторов, выполняемых вместе, например, в цикле или по условию. Ключевые слова begin и end следует рассматривать как единый оператор, поэтому после begin точка с запятой не ставится, а количество begin и end в программе всегда одинаково. Таким образом, тело программы заключено в операторные скобки, объединяющие все ее операторы.

Только последний оператор программы завершается точкой: end. , все остальные -- символом ;.
Если в программе нет констант, в ней будет отсутствовать раздел const, если нет и переменных -- раздел var.

При написании текста программы следует соблюдать несложные правила, облегчающие его последующие чтение и модификацию:

· внутри ключевых слов или идентификаторов не должно быть пробелов и других разделителей, таких как табуляция или перевод строки; во всех иных случаях не запрещено разрывать оператор, однако, делать это следует лишь тогда, когда написание оператора в одну строку затрудняет восприятие текста программы;

· на каждой строке обычно пишется один оператор (это облегчает и отладку программы);

· операторы одного уровня вложенности пишутся с одинаковым отступом слева; например, хорошим тоном считается после начала каждого блока (begin) отступать на следующей строке на символ или несколько символов вправо, а закрывать блок так, чтобы соответствующий end; находился под своим begin. Приведем пример неправильного и правильного структурирования:

program p1; var

a,b,c:real;  begin

writeln ('Введите значения A и B:');read(a,b);

 c:=a+b; writeln ('A+B=',c); c:=a-b;

writeln ('A-B=',c); end.

Текст этой программы структурирован явно неудачно, гораздо лучше он воспринимается так:

program p1;

var a,b,c:real;

begin
  writeln ('Введите значения A и B:');

  read (a,b);

  c:=a+b; 
  writeln ('A+B=',c); 
  c:=a-b;

  writeln ('A-B=',c); 
end.

· основные действия программы комментируются: комментарием в Паскале считается любой  текст, ограниченный фигурными скобками { ... } или символами (* ... *). Количество комментариев в программе никак не влияет на объем генерируемого машинного кода, они призваны, прежде всего, облегчить последующее чтение и модификацию исходного текста программы.

В качестве примера приведем законченную программу на Паскале, вычисляющую вещественные корни произвольного квадратного уравнения.

program Equation;

var a,b,c,d,x1,x2:real; begin

 writeln;

 writeln ('Введите коэффициенты a,b,c:');

 read (a,b,c);

 d:=sqr(b)-4*a*c;

 x1:=(-b+sqrt(d))/(2*a);

 x2:=(-b-sqrt(d))/(2*a);

 writeln ('Корни уравнения');

 writeln (x1:10:2,x2:10:2);

 readln; readln;

end.

В разделе описаний программы всем переменным, требуемым для решения задачи, присвоен тип real, и этот выбор вполне очевиден -- коэффициенты a, b и с -- не обязательно целые значения. "Лишний" оператор writeln; перед приглашением к вводу -- гарантия того, что приглашение будет напечатано с начала пустой строки, ведь мы пока не умеем очищать экран, и, возможно, при запуске нашей программы курсор находится не в начале строки. После вычисления дискриминанта и корней x1, x2 (условие d≥0 мы пока не проверяем), на экран печатается информационное сообщение "Корни уравнения", а затем с новой строки выводятся значения x1 и x2 с соблюдением указанных ширины и точности вывода. Наконец, два оператора readln; в конце программы позволяют ей дождаться, пока пользователь не нажмет клавишу Enter. "Удвоение" оператора здесь связано с тем,  что один раз мы уже нажимали Enter после ввода данных, и первый readln; прочитает именно это нажатие, а второй будет ждать еще одного. Будь ввод данных организован в виде a:=1; b:=2; c:=0; или readln(a,b,c);, нам хватило бы и одного readln;. Однако, оператор readln(a,b,c); потребовал бы от пользователя ввести все 3 числа в одной строке, а задание фиксированных значений a, b, c уменьшило бы до нуля ее полезность. В дальнейшем мы узнаем более гибкие способы программирования реакции программы на действия пользователя.
6. Компилятор и оболочка Turbo Pascal
Кратко рассмотрим основные действия с этой программой. После установки программы из папки Паскаля или с помощью ярлыка запускается файл с именем turbo.exe. Основные  элементы окна Турбо Паскаля приведены на рис. 6.1.

Рис. 6.1. Окно программы Turbo Pascal
Как видно на рис. 6.1, устройство окна несколько отличается от стандартного окна Windows. 
Для входа в верхнее меню следует нажать клавишу F10 или сделать щелчок мышью на нужном пункте. Если ни одно окно не открыто или необходимо окно для новой программы, в меню File выберите пункт New. Для открытия ранее сохраненной программы в этом же меню выберите Open или просто нажмите F3 из основного окна. Выбрать нужную программу можно в появившемся диалоговом окне, возможно, для этого потребуется сменить папку.

Если открыто сразу несколько окон, переключаться между ними можно, нажимая при нажатой левой Alt цифровую клавишу с номером нужного окна (от 1 до 9). Получить список всех окон можно комбинацией клавиш Alt+0, закрыть текущее окно -- Alt+F3

После ввода программы (а лучше несколько раз в процессе ввода) ее следует сохранить на диске. Для этого в меню File достаточно выбрать команду Save или нажать F2 из окна программы. При первом сохранении программе нужно дать имя. Помните, что Паскаль разрабатывался для операционной системы MS-DOS и даваемые файлам имена должны включать в себя только латинские буквы и цифры, а по длине не превышать 8 символов. Тип файла .pas можно не указывать, он добавится к имени автоматически.

Для сохранения файла с программой на дискету или открытия его с дискеты проще всего в окне ввода указать a: и нажать Enter. Аналогично можно перейти к другим сменным носителям, введя их системное имя диска.

После сохранения программы, для ее компиляции и выполнения достаточно нажать комбинацию клавиш Ctrl+F9. Если в процессе компиляции найдена синтаксическая ошибка, компилятор сообщит о ней и установит курсор на строку, содержащую ошибку. После исправления ошибки нажмите Ctrl+F9 еще раз. По завершении программы, если в ней не предусмотрено никакого останова, произойдет автоматический возврат в окно с исходным текстом. Вернуться к окну вывода программы после ее выполнения можно, нажав Alt+F5.

Вследствие алгоритмических ошибок или аппаратных проблем запущенная программа может "зависнуть", например, бесконечно выполняя неправильно запрограммированный цикл. В этом случае, при работе из оболочки Паскаля, можно попытаться прервать ее сочетанием клавиш Ctrl+Break.

Переключение на русский язык и обратно из оболочки Турбо Паскаля зависит от настроек ОС, уточните их  у преподавателя или оператора.

Переключение в полноэкранный режим и обратно, как и для других приложений DOS, выполняется сочетанием клавиш Alt+Enter.
По умолчанию программа Turbo Pascal не создает исполняемых файлов *.exe. Чтобы она начала это делать, достаточно в верхнем меню Compile установить пункт-переключатель Destination в значение Disk (значение по умолчанию -- Memory).

Оболочка Паскаля включает удобные средства отладки программ, основные требуемые команды собраны в меню Debug. Для выполнения программы по строкам достаточно нажимать F7 или F8. Разница между назначениями этих клавиш в том, что нажатие F7 пошагово выполняет программу с входом во все возможные подпрограммы, а F8 -- нет. Пока тема "Подпрограммы" не изучена, разницы в действии клавиш не будет заметно. Строка, которая будет выполняться следующей, выделена светло-зеленым цветом. Перейти сразу к нужному месту в программе можно, установив курсор на соответствующую строку и нажав клавишу F4. Выйти из режима пошагового выполнения и прервать работу отладчика позволяет сочетание клавиш Ctrl+F2.

В процессе пошагового выполнения можно посмотреть и даже изменить значения любых переменных. Для этого достаточно установить курсор на имя нужной переменной и нажать сочетание клавиш Ctrl+F4. На экране должно появиться диалоговое окно "Evaluate and Modify". В поле ввода Expression уже показано имя переменной, на которой стоял курсор. Если это не так, здесь можно задать имя любой доступной переменной или ввести произвольное выражение на Паскале. Нажатие Enter или кнопки Evaluate выводит результат в поле Result. В поле New Value можно, не прерывая пошагового выполнения, изменить значение переменной. Если предложенное изменение возможно, после нажатия кнопки Modify оно будет показано в поле Result, в противном случае там выведется сообщение "Cannot be modified". Таким образом, окно "Evaluate and Modify" позволяет гибко тестировать поведение программы при различных входных данных.

Наконец, в Паскаль встроена мощная система помощи. Нажатие клавиши F1 вызывает окно помощи по редактору, а Shift+F1 -- индекс оглавления справочной системы. В тексте помощи ссылки на другие разделы выделены желтым цветом, а переходить по ссылкам можно клавишами Tab и Shift+Tab или клавишами со стрелками. Нажатие Alt+F1 возвращает к чтению предыдущей статьи. Пользуясь выделением с помощью мыши или клавишами со стрелками при нажатой Shift, информацию из окна помощи можно копировать во встроенный буфер обмена и затем вставлять в новый или существующий файл. Соответствующие команды доступны из меню Edit. Дополнительные возможности системы помощи доступны из меню Help.
Оболочка программы Free Pascal отличается размером окна консоли, но очень похожа по управлению, так что рассматривать её отдельно не будем.
Подробнее о работе с этой и другими оболочками компании Borland можно узнать в специальной литературе.

7. Разветвляющийся вычислительный процесс и условный оператор

Главное, чего недостает нашим первым программам -- гибкости и умения принимать решения. Ведь уже несложный алгоритм решения квадратного уравнения предусматривает два варианта расчета, реальные же алгоритмы могут выдавать результаты, зависящие от десятков и сотен условий.

Разветвляющийся вычислительный процесс (РВП) реализуется по одному из нескольких направлений вычисления (ветвей алгоритма). Выбор одной из ветвей зависит от истинности или ложности некоторого условия (логического выражения), включенного в состав условного оператора. Программа должна учитывать все возможные ветви вычислений. При запуске программы, в зависимости от данных, выполняется только одна из возможных ветвей.
7.1. Логические выражения 
Логические выражения (ЛВ) строятся из АВ, операций отношения, логических операций и круглых скобок. 
Результатом вычисления ЛВ является одно из двух значений: true или false.
7.2. Операции отношения 
Операции отношения (сравнения) имеют следующий общий вид:

АВ1 ОО АВ2

где АВ -- арифметические выражения, ОО -- один из следующих знаков операций:

<
<=
>
>=
=
<> 
Последний знак обозначает отношение "не равно". Обратите также внимание на запись отношений "меньше или равно", "больше или равно".

В любое логическое выражение должна входить хотя бы одна операция отношения.
Приведем примеры ЛВ, включающих одну ОО:
d<0
  -- выбор ветви вычислений зависит от значения d;
sqr(x)+sqr(y)<=sqr(r) -- результат будет равен true для точек с координатами (x, y), лежащих внутри круга радиуса R с центром в начале координат;
cos(x)>1 -- результат этого ЛВ всегда равен false.
К вещественным значениям в общем случае неприменима операция = ("равно") из-за неточного представления этих значений в памяти компьютера. Поэтому для вещественных переменных отношение вида a=b часто заменяется на abs(a–b)<eps, где eps -- малая величина, определяющая допустимую погрешность.
7.3. Логические операции
Логические операции применимы только в логических выражениях и служат для составления сложных условий, требующих более одной операции отношения. В Паскале определены логические операции, описанные в табл. 7.1.
Табл. 7.1. Логические операции языка Паскаль
	Математическая запись
	Запись на Паскале
	Название

	not
	Отрицание

	and
	Операция "И" (логическое умножение)

	or
	Операция "ИЛИ" (логическое сложение)

	xor
	Операция "исключающее ИЛИ"


Операция NOT применима к одному логическому выражению (является унарной). Ее результат равен true, если выражение ложно и наоборот. 
Например, выражение NOT (sin(x)>1) всегда даст значение true.
Операция AND связывает не менее двух логических выражения (является бинарной). Ее результат равен true, если все выражения истинны или false, если хотя бы одно из выражений ложно.
В качестве примера распишем выражение 

. Т. к. операции принадлежности в Паскале нет, используем операцию AND и операции отношения: (x>=a) and (x<=b).
Математическое выражение a,b,c>0 (одновременно) будет иметь вид (a>0) and (b>0) and (c>0).
Операция OR также связывает не менее двух логических выражений. Ее результат равен true, если хотя бы одно выражение истинно и false, если все выражения ложны.
Распишем выражение 

. На Паскале оно будет иметь вид(x<a) or (x>b). Другой способ связан с применением операции NOT: not ((x>=a) and (x<=b)).
Условие "хотя бы одно из значений a,b,c положительно" может быть записано в виде (a>0) or (b>0) or (c>0) .
Условие "только одно из значений a,b,c положительно" потребует объединения возможностей операций AND и OR:
(a>0) and (b<=0) and (c<=0) or 
(a<=0) and (b>0) and (c<=0) or 
(a<=0) and (b<=0) and (c>0).
Операция XOR, в отличие от OR, возвращает значение "ложь" (false) и в том случае, когда все связанные ей логические выражения истинны. Чтобы лучше уяснить это отличие, составим так называемую таблицу истинности двух логических операций (табл. 7.2). Для краткости значение false обозначим нулем, а true -- единицей. Для двух логических аргументов возможно всего 4 комбинации значений 0 и 1.
Табл. 7.2. Таблица истинности операций OR и XOR
	Аргумент A
	Аргумент B
	A or B
	A xor B

	0
	0
	0
	0

	0
	1
	1
	1

	1
	0
	1
	1

	1
	1
	1
	0


В качестве примера использования операции XOR запишем условие "только одно из значений a,b положительно":

(a>0) xor (b>0).
К сожалению, записать условие "только одно из значений a,b,c положительно" в напрашивающемся виде (a>0) xor (b>0) xor (c>0) нельзя -- результат этого выражения будет равен true и в случае, когда все три значения положительны. Связано это с тем, что при последовательном расчете логических выражений слева направо (1 xor 1) xor 1 будет равно 0 xor 1 = 1.

С помощью xor удобно организовывать различного рода переключатели, которые последовательно должны принимать одно из двух состояний:

x := x xor true; writeln ('x=', x);

x := x xor true; writeln ('x=', x);

Независимо от начального значения логической переменной x, второе выведенное на экран значение будет логическим отрицанием первого. В реальной практике конструкции подобные x := x xor true; не дублируются в коде многократно, а применяются внутри цикла (см. гл. 9).

Приоритет логических операций следующий:  самая старшая операция -- not, затем and, следующие по приоритету -- or и xor (равноправны между собой), самый низкий приоритет имеют операции отношения. Последнее служит причиной того, что в составных условиях отдельные отношения необходимо заключать в круглые скобки, как и сделано во всех примерах раздела.
7.4. Короткий условный оператор

Это первый вид условного оператора, позволяющий программе выполнить или пропустить некоторый блок вычислений. Общий вид короткого условного оператора следующий:
if логическое_выражение then оператор1;

Сначала вычисляется логическое выражение, если оно имеет значение true, то выполняется оператор1, иначе оператор1 игнорируется. Блок-схема соответствующего вычислительного процесса представлена на рис. 7.1.

Рис. 7.1. Блок-схема короткого условного оператора

Если по условию требуется выполнить несколько операторов, их необходимо заключить в операторные скобки begin...end;, образуя единый составной оператор:

if d>0 then begin

x1:=(-b+sqrt(d))/(2*a);

x2:=(-b-sqrt(d))/(2*a);


writeln (x1:8:3,x2:8:3);

end;

Здесь по условию d>0 выполняется 3 оператора, первые два из которых вычисляют корни x1 и x2 квадратного уравнения, а последний выводит на экран найденные значения корней.
Следующий пример иллюстрирует поиск значения y=max(a,b,c). Поскольку стандартной функции для нахождения максимума в Паскале нет, применим 2 коротких условных оператора:
y:=a;

if b>y then y:=b;

if c>y then y:=c;

Вообще, для условной обработки N значений требуется N-1 короткий условный оператор.

7.5. Полный условный оператор

Эта форма условного оператора позволяет запрограммировать 2 ветви вычислений. Общий вид полного условного оператора следующий:

if логическое_выражение then оператор1

else оператор2;

Оператор работает следующим образом: если логическое выражение имеет значение true, то выполняется оператор1, иначе выполняется оператор2. Всегда выполняется только один из двух операторов. Перед ключевым словом else ("иначе") точка с запятой не ставится, т.к. if-then-else -- единый оператор.
Вычислим значение m=min(x,y) с помощью полного условного оператора:
if x<y then m:=x else m:=y;

В следующем примере выполним обработку двух переменных: если значения a и b одного знака, найти их произведение, иначе заменить нулями.

if a*b>0 then c:=a*b

else begin

  a:=0; b:=0;

end;

Из примера видно, что к ветви алгоритма после ключевого слова else, состоящей более чем из одного оператора, также применяются операторные скобки.
7.6. Составной условный оператор 
Эта форма условного оператора применяется, когда есть более 2 вариантов расчета. Общий вид составного оператора может включать произвольное число условий и ветвей расчета:
if логическое_выражение1 then оператор1

else if логическое_выражение2 then оператор2

...

else if логическое_выражениеN then операторN
else оператор0;

При использовании оператора последовательно проверяются логические выражения 1, 2, ... ,N, если некоторое выражение истинно, то выполняется соответствующий оператор и управление передается на оператор, следующий за условным. Если все условия ложны, выполняется оператор0 (если он задан). Число ветвей N неограниченно, ветви else оператор0; может и не быть.
Существенно то, что если выполняется более одного условия из N, обработано все равно будет только первое истинное условие. В показанной ниже программе составной условный оператор неверен, если ее разработчик хотел отдельно учесть значения x, меньшие единицы по модулю:
var x:real;

begin

 write ('Введите x:'); readln (x);

 if x<0 then writeln ('Отрицательный')

 else if x=0 then writeln ('Ноль')

 else if abs(x)<1 then 
  writeln ('По модулю меньше 1')

 else writeln ('Больше 1');

end.
Условие x<0 сработает, например, для значения x=-0.5, что не позволит программе проверить условие abs(x)<1.
Еще одну распространенную ошибку работы с составным условным оператором показывает произведенный ниже расчет знака n переменной a:
if a<0 then n:=-1;

if a=0 then n:=0

else n:=1;

Применение одного короткого и одного полного условных операторов является здесь грубой ошибкой -- ведь после завершения короткого условного оператора для всех ненулевых значений a будет выполнено присваивание n:=1. Правильных вариантов этого расчета, по меньше мере, два:
if a<0 then n:=-1;

if a=0 then n:=0;
if a>0 then n:=1;
— с помощью 3 коротких условных операторов, вариант не очень хорош тем, что проверяет лишние условия даже тогда, когда знак уже найден.

if a<0 then n:=-1;

else if a<0 then n:=1;
else n:=0;
— с помощью составного условного оператора, этот вариант лучше. На практике следует помнить, что для вещественных значений сравнение с нулем может быть проблематично.
Можно сделать вывод, что при программировании многовариантных расчетов следует соблюдать осторожность, чтобы не допустить "потерянных" или неверно срабатывающих ветвей алгоритма.
В качестве еще одного примера рассчитаем значение функции, заданной графически (рис. 7.2).

Рис. 7.2. Функция, заданная графически

Перепишем функцию в аналитическом виде:

Одним из вариантов запрограммировать вычисление y(x) мог бы быть следующий:
if abs(x)>1 then y:=0

else if x<0 then y:=x+1

else y:=1-x;

Возможна и последовательная проверка условий слева направо, которая породит немного избыточный, но легче воспринимающийся код:
if x<-1 then y:=0

else if x<0 then y:=x+1

else if x<1 then y:=1-x

else y:=0;

7.7. Вложенные условные операторы

Когда после ключевых слов then или else вновь используются условные операторы, они называются вложенными. Число вложений может быть произвольно, при этом действует правило: else всегда относится к ближайшему оператору if, для которого ветка else еще не указана. Часто вложением условных операторов можно заменить использование составного.
В качестве примера рассмотрим программу для определения номера координатной четверти p, в которой находится точка с координатами (x,y). Для простоты примем, что точка не лежит на осях координат. Без использования вложений основная часть программы может иметь следующий вид:
if (x>0) and (y>0) then p:=1

else if (x<0) and (y>0) then p:=2

else if (x<0) and (y<0) then p:=3

else p:=4;

Однако использование такого количества условий представляется явно избыточным. Перепишем программу, используя тот факт, что по каждое из условий x>0, x<0 оставляет в качестве значения p только по 2 возможных четверти из 4:
if x>0 then begin

 if y>0 then p:=1

 else p:=4;

end

else begin

 if y>0 then p:=2

 else p:=3;

end;

В первом фрагменте программе проверяется от 2 до 6 условий, во втором -- всегда только 2 условия. Здесь использование вложений дало существенный выигрыш в производительности.
Рассмотренный в п. 7.6 пример с определением знака числа может быть переписан и с использованием вложения:
if a>0 then n:=1

else begin

 if a<0 then n:=-1
 else n:=0;

end;

Однако, как эти операторы, так и составной условный оператор из п. 7.6 проверяют не более 2 условий, так что способы примерно равноценны.
7.8. Оператор выбора 
Для случаев, когда требуется выбор одного значения из конечного набора вариантов, оператор if удобнее заменять оператором выбора (переключателем) case:

case выражение of
 список1: оператор1;

 список2: оператор2;

 . . .

 списокN: операторN;

 else оператор0;

end;

Оператор выполняется так же, как составной условный оператор.

Выражение должно иметь порядковый тип (целый или символьный). Элементы списка перечисляются через запятую, ими могут быть константы и диапазоны значений того же типа, что тип выражения. Диапазоны указываются в виде:

Мин.значение .. Макс.значение

Оператор диапазона записывается как два рядом стоящих символа точки. В диапазон входят все значения от минимального до максимального включительно.
В качестве примера по номеру месяца m определим число дней d в нем:
case m of

 1,3,5,7..8,10,12: d:=31;

 2: d:=28;

 4,6,9,11: d:=30;

end;

Следующий оператор по заданному символу c определяет, к какой группе символов он относится:
case c of

 'A'..'Z','a'..'z': 
   writeln ('Латинская буква');

 'А'..'Я','а'..'п','р'..'я': 
   writeln ('Русская буква');

 '0'..'9': 
   writeln ('Цифра');

 else writeln ('Другой символ');

end;

Здесь отдельные диапазоны для русских букв от "а" до "п" и от "р" до "я" связаны с тем, что между "п" и "р" в кодовой таблице DOS находится ряд не-буквенных символов (см. Приложение 1).

Если по ветви оператора case нужно выполнить несколько операторов, действует то же правило, что для оператора if, т. е. ветвь алгоритма заключается в операторные скобки begin ... end;.

7.9. Примеры программ с условным оператором

Приведем несколько примеров законченных программ, использующих РВП.

1. Проверить, может ли быть построен прямоугольный треугольник по длинам сторон a, b, c.

Проблема с решением этой задачи -- не в проверке условия теоремы Пифагора, а в том, что в условии не сказано, какая из сторон может быть гипотенузой. Подходов возможно несколько -- запрашивать у пользователя ввод данных по возрастанию длины сторон, проверять все три возможных условия теоремы Пифагора и т. п. Используем наиболее естественное решение -- перед проверкой условия теоремы Пифагора упорядочим величины a, b, c так, чтобы выполнялись соотношения a≤b≤c. Для этого применим прием с обменом значений переменных из п. 4.1.
var a,b,c, {Длины сторон}

 s:real;{Буферная переменная для обмена}

begin
{ Секция ввода данных }

 writeln;

 write ('Введите длину 1 стороны:');

 readln (a);

 write ('Введите длину 2 стороны:');

 readln (b);

 write ('Введите длину 3 стороны:');

 readln (c);

{ Сортируем стороны по неубыванию }

 if (a>b) then begin

    s:=a; a:=b; b:=s;

 end;

 if (a>c) then begin

    s:=a; a:=c; c:=s;

 end;

 if (b>c) then begin

    s:=b; b:=c; c:=s;

 end;

{ Проверка и вывод }

 if abs(a*a+b*b-c*c)<1e-8 then writeln
  ('Прямоугольный треугольник ',

    'может быть построен!')

 else writeln('Прямоугольный треугольник ',

    'не может быть построен!')

end.

2. Определить, попадает ли точка плоскости, заданная координатами (a, b) в прямоугольник, заданный координатами двух углов (x1, y1) и (x2, y2).

Как и в предыдущей задаче, было бы не совсем корректно требовать от пользователя вводить данные в определенном порядке -- гораздо лучше при необходимости поменять x- и y-координаты прямоугольника так, чтобы пара переменных  (x1, y1) содержала координаты левого нижнего угла прямоугольника, а (x2, y2) -- правого верхнего.

var x1,y1,x2,y2,a,b:real;

begin
 writeln ('Введите координаты 1 угла:');

 read (x1,y1);

 writeln ('Введите координаты 2 угла:');

 read (x2,y2);

 if x1>x2 then begin

  a:=x1; x1:=x2; x2:=a;

 end;

 if y1>y2 then begin

  a:=y1; y1:=y2; y2:=a;

 end;

 writeln ('Введите координаты точки:');

 read (a,b);

 if (x1<=a) and (a<=x2) 
    and (y1<=b) and (b<=y2) then writeln 
   ('Точка попадает в прямоугольник')

 else writeln 
   ('Точка не попадает в прямоугольник');

end.

3. Вводится денежная сумма в рублях и копейках. Программа печатает введенную сумму с правильной формой слов "рубли" и "копейки", например, "123 рубля 15 копеек".

Окончание, используемое для слов "рубли" и "копейки", зависит от последней цифры суммы, которую можно получить, взяв остаток от деления на 10 (1058 рублей, 38 рублей и т.д.). Исключения -- суммы с последними двумя цифрами от 11 до 19 включительно, которые всегда произносятся "рублей" и "копеек" (511 рублей, но 51 рубль). Используя эту информацию, составим программу.

var r,k,o10,o100:integer;

begin
 writeln;

 write ('Введите количество рублей, ',

     'затем пробел и количество копеек:');

 read (r,k);

 writeln;

 o10:=r mod 10;   {Взяли последнюю цифру}

 o100:=r mod 100; {...и 2 последних цифры}

 write ('Правильно сказать: ',r,' '); 
 {Печатаем число рублей, затем пробел}

 if (o100>10) and (o100<20) 
     or (o10>4) or (o10=0) then 
  write ('рублей')

 else if (o10>1) and (o10<5) then 
  write ('рубля')

 else 
  write ('рубль');

 {аналогично для копеек:}

 o10:=k mod 10;

 o100:=k mod 100;

 write (' ',k,' '); 
 {печатаем число копеек с пробелами}

 if (o100>10) and (o100<20) or 
     (o10>4) or (o10=0) then 
  write ('копеек')

 else if (o10>1) and (o10<5) then 
  write ('копейки')

 else write ('копейка');

end.
8. Директивы компилятора и обработка ошибок ввода

Компилятор Паскаля -- сложное приложение, имеющее множество настроек. При написании учебных программ большинство этих настроек не имеют значения, но некоторые из них окажутся нам полезны. Для управления компилятором существует 2 основных возможности: настройка режимов работы с помощью верхнего меню Options оболочки Turbo Pascal и настройка конкретной программы с помощью директив компилятора, которую мы кратко рассмотрим. В общем виде директива компилятора представляет собой конструкцию вида {$X+} или {$X-}, где X -- латинская буква. Вариант со знаком "+" включает некоторый режим работы компилятора (например, строгий контроль программой соответствия типов данных, вывод системных диагностических сообщений и т. д.), а вариант со знаком "-" выключает его. Расположение директив, в общем, произвольно, однако, директивы, влияющие на всю программу, принято располагать в самом начале файла с исходным текстом. Фигурные скобки комментария { ... } необходимы как часть синтаксиса директивы.

Подробную информацию о назначении всех директив можно получить в справочной системе оболочки. Основные директивы компилятора также кратко описаны в Приложении 2.

Наиболее   полезной   для   нас   выглядит   директива  {$I-}/{$I+}, соответственно, выключающая и включающая автоматический контроль программой результатов операций ввода/вывода (в/в). К операциям в/в относятся, в числе прочего, ввод данных пользователем, вывод строки на принтер, открытие файла для получения или вывода данных и т. п. Понятно, что даже несложная учебная программа выглядит лучше, если она умеет реагировать на неправильные действия пользователя или возникающие ошибки не просто выводом маловразумительного системного сообщения на английском языке, а доступным неискушенному пользователю текстом. По умолчанию контроль в/в включен и системные сообщения об ошибках генерируются автоматически. Все они кратко приведены в Приложении 3. Для замены системной диагностики своей собственной следует, во-первых, отключить директиву контроля оператором {$I-}, а во-вторых, сразу же после оператора, который мог породить ошибку, проверить значение, возвращаемое системной функцией IoResult. Эта функция возвращает ноль, если последняя операция в/в прошла успешно, в противном случае возвращается ненулевое значение. После завершения "критического" оператора директиву следует включить снова, чтобы не создавать потенциально опасных ситуаций в коде, который будет писаться далее. Приведем пример, написав "расширенную" программу решения квадратного уравнения, корректно реагирующую на возникающие ошибки:

uses printer;

var a,b,c,d,x1,x2:real;

begin

  writeln;

  writeln ('Введите коэффициенты a,b,c:');

  {$I-} read (a,b,c); {$I+}

  if IoResult<>0 then begin 
   {Возникла ошибка!}

   writeln ('Вы не ввели 3 числа, ',

    'это что-то другое!');

    reset (input); {очищаем стандартный 
 поток ввода перед ожиданием нажатия Enter}

    readln;

    halt; {а этим оператором можно 
      аварийно завершить программу}

  end;

  d:=sqr(b)-4*a*c;

  if d<0 then begin

    writeln ('Ошибка - дискриминант<0');

    reset (input); readln; halt;

  end;

  x1:=(-b+sqrt(d))/(2*a);

  x2:=(-b-sqrt(d))/(2*a);

  {$I-}

  writeln (lst,'x1=',x1:8:2,' x2=',x2:8:2);

  {$I+}

  if IoResult<>0 then 
   writeln ('Не удалось напечатать')

  else writeln ('Результаты напечатаны');

  reset (input); readln; halt;

end.

Специальной директивы для контроля математических ошибок в Паскале не предусмотрено, но это почти всегда можно сделать обычными проверками корректности данных. Обратите внимание на альтернативное решение проблемы "двух readln" в этом коде, а также на новый оператор halt и способ контроля того, удалось ли вывести строку на принтер.

В дальнейшем мы не всегда будем выполнять подобные проверки, однако, их никогда не помешает делать, если вы хотите создать удобные для пользователя и понятные для разработчика программы.
Для работы с вещественными числами с двойной точностью (тип double) может также понадобиться указать перед программой директиву {$N+}, позволяющую сгенерировать код для аппаратной обработки таких чисел.

9. Оператор цикла. Циклы с предусловием и постусловием

Циклический вычислительный процесс (ЦВП) характеризуется повторением одних и тех же вычислений над некоторым набором данных. Числом повторений цикла управляет специальная переменная, называемая его счетчиком или управляющей переменной цикла. На счетчик накладывается условие, определяющее, до каких пор следует выполнять цикл.

Повторяемый блок вычислений называют телом цикла. В теле цикла должно быть обеспечено изменение значения счетчика, чтобы он мог завершиться. Если тело цикла состоит более чем из одного оператора, оно заключается в операторные скобки begin ... end;. Однократное выполнение тела цикла называют его шагом.

Таким образом, для программирования цикла достаточно определить условие, управляющее числом его повторений и описать операторы, образующие тело цикла. С этой точки зрения, теоретически возможны всего два вида циклов -- проверка условия либо предшествует выполнению тела цикла, либо происходит после него. Изобразим эти циклы в виде блок-схем (рис. 9.1).
В цикле с предусловием сначала проверяется условие, затем, в зависимости от того, истинно оно или ложно, либо выполняется тело цикла, либо следует переход к оператору, следующему за телом цикла. После завершения тела цикла управление вновь передается на проверку условия. Естественно, предполагается, что в теле цикла было обеспечено некоторое изменение входящих в условие переменных -- в противном случае произойдет зацикливание и программа "зависнет".

Рис. 9.1. Блок-схемы циклов с предусловием и постусловием

Для цикла с постусловием сначала выполняется тело цикла, затем управление передается на проверку условия. В зависимости от истинности или ложности условия, тело цикла выполняется повторно или же происходит переход к оператору, следующему за телом цикла. Всё, сказанное о возможном зацикливании для цикла с предусловием, справедливо и для цикла с постусловием.
Исходя из приведенных блок-схем, очевидно основное различие двух циклов: цикл с постусловием гарантированно выполняется хотя бы раз, а цикл с предусловием может не выполняться ни разу, если условие сразу же окажется ложным. 
В языке Паскаль реализованы оба вида циклов. Цикл с предусловием имеет следующий общий вид:

while логическое_выражение do begin

 {операторы тела цикла}

end;

Работу цикла можно описать словами: "пока логическое выражение истинно, повторяется тело цикла". 
Логическое выражение строится по правилам, изученным в гл. 7. Тело цикла могут образовывать любые операторы Паскаля. Если в цикле находится всего один оператор, операторные скобки, показывающие начало и конец тела цикла, можно не писать.

Общая запись цикла с постусловием следующая:

repeat

 {операторы тела цикла}

until логическое_выражение;

Работает цикл с постусловием следующим образом: "тело цикла повторяется до тех пор, пока логическое выражение не станет истинным". Обратите внимание, что, в отличие от while, цикл repeat в Паскале работает, пока условие ложно. Это отличие подчеркивается использованием ключевого слова until ("до тех пор, пока не") вместо while ("до тех пор, пока"). Кроме того, в виде исключения, тело цикла repeat, даже если оно состоит из нескольких операторов, можно не заключать в операторные скобки.

Довольно часто циклы взаимозаменяемы. Представим, например, что для каждого из значений переменной x=1, 2, ... ,20, нужно выполнить некоторый расчет (математически этот закон изменения x можно записать как 

). Это можно сделать как в цикле while:
x:=1;

while x<=20 do begin
 {операторы расчета}

 x:=x+1;

end;
так и с помощью repeat:
x:=1;

repeat

 {операторы расчета}

 x:=x+1;

until x>20;
Как видно из листинга, управляющей переменной x в обоих случаях присвоено начальное значение 1, оба цикла изменяют значение x и, соответственно, условие цикла, оператором x:=x+1;,  но для цикла repeat условие "перевернуто" ("пока x не станет больше 20"), а тело не заключено в операторные скобки.

Зачастую использование одного из циклов выглядит предпочтительней. Например, обработка ввода пользователя с клавиатуры удобней с помощью repeat (сначала пользователь должен нажать клавишу, затем следуют проверки и обработка).
10. Цикл со счетчиком и досрочное завершение циклов

Прежде, чем перейти к примерам, обсудим еще ряд проблем, связанных с циклами. Как для while, так и для repeat, во-первых, нигде в явном виде не задается число шагов цикла (хотя его обычно можно вычислить), во-вторых, при использовании обоих циклов программист должен заботиться об изменении управляющей переменной. Между тем, весьма распространены задачи, где объем последовательно обрабатываемых данных известен заранее (а значит, известно и требуемое число шагов цикла), а управляющая переменная меняется с шагом, равным единице. Рассмотренный выше пример с двадцатью значениями x относится именно к таким задачам. Поэтому для обработки заранее известного объема данных с шагом по управляющей переменной, равным единице, вместо цикла while используется цикл со счетчиком (цикл for). Его общий вид следующий:

for счетчик := НЗ to КЗ do begin

 {операторы тела цикла}

end;

Здесь счетчик -- целочисленная переменная, НЗ (начальное) и КЗ (конечное) значения -- целочисленные выражения или константы. Тело цикла образовано не менее чем одним оператором, если этот оператор единственный, операторные скобки можно не писать. Работает цикл for следующим образом: счетчик автоматически меняется от начального значения до конечного включительно, для каждого значения счетчика повторяется тело цикла. После каждого шага цикла значение счетчика автоматически увеличивается на единицу. Если требуется, чтобы значение счетчика уменьшалось, а не увеличивалось, вместо ключевого слова to используется downto.

Подобно while, цикл for может не выполниться и ни разу -- если начальное значение управляющей переменной сразу же больше конечного (при использовании to) или меньше (при использовании downto).

Запишем рассмотренный выше цикл по переменной x с помощью оператора for:

for x:=1 to 20 do begin

 {операторы тела цикла}

end;

Удобства очевидны -- границы изменения x заданы сразу же при входе в цикл, а выполнять шаг по x отдельным оператором не требуется. Понятны и ограничения -- x должен быть описан с типом данных integer, а в случае изменения значения x с шагом, не равным единице, использовать for вместо while не удалось бы.

Еще одна проблема связана с тем, что при выполнении программы довольно часто возникает необходимость завершить цикл досрочно -- например, если искомое в нем значение уже найдено или возникла ошибка, из-за которой дальнейшие шаги становятся бессмысленными. Теоретически цикл можно было бы завершить, присвоив управляющей переменной значение, выходящее за пределы ее изменения:

x:=1;

while x<10 do begin

 y:=ln(x);

 if y>2 then x:=10; 
  {При y>2 цикл нужно завершить}

 x:=x+0.5;

end;

Однако, во избежание трудноуловимых ошибок, управляющую переменную не принято менять иначе, чем для выполнения шага цикла. Например, после оператора if y>2 then x:=10; в нашем листинге выполнение текущего шага продолжится, что чревато лишними или неправильными вычислениями. Кроме того, текст такой программы воспринимается нелегко.

Поэтому для досрочного выхода из цикла существует оператор break (от англ. "to break" -- прервать), немедленно прекращающий его выполнение:

x:=1;

while x<10 do begin
 y:=ln(x);

 if y>2 then break; 
 x:=x+0.5;

end;

Указание break здесь передаст управление на оператор, следующий непосредственно за циклом. В отличие от предыдущего примера, здесь не будет выполняться часть тела цикла, следующая за break;.

Для немедленного продолжения цикла со следующего шага используется оператор continue (от англ. "to continue" -- продолжить):

var n:integer;

begin

 repeat

  writeln ('Введите положительное число:');

  read (n);

  if n<1 then continue; 

  {Если введено n<1, снова запросить число}

  {Операторы обработки числа}

  break; {Выход из цикла обработки}

 until false;

end.

В этом примере оператор continue использован для повторного перехода к вводу n, если введено n<1. Так как цикл обработки здесь -- бесконечный, для выхода из него необходим break;. Кроме того, пример показывает одну из возможностей контроля правильности ввода данных. Указав директиву {$I-}, изученную в гл. 8, мы могли бы защитить программу и от ввода пользователем нечисловых значений в качестве n.

В теме "Кратные циклы" второй части курса будет рассмотрен оператор goto, также способный решить проблему аварийного завершения циклов.
11. Типовые алгоритмы табулирования функций, вычисления количества, суммы и произведения

Итак, основное назначение циклов -- обработка большого объема данных. Математически эта обработка зачастую сводится к поиску, выбору и статистической обработке нужных величин. Практически в любой реальной задаче мы ищем максимальные и минимальные значения в наборе данных, суммируем или перемножаем требуемые данные, определяем арифметическое среднее или количество элементов, отвечающих условию. Для решения всех этих распространенных задач существуют типовые алгоритмы, задающие правила выполнения соответствующих расчетов. Изучением этих алгоритмов мы займемся в гл. 11 и 12.

Разумеется, настоящие задачи, встающие перед программистами, значительно сложнее, чем приведенные далее примеры, но из типовых алгоритмов, как из кирпичиков, строится здание любой сложной программы.

11.1. Алгоритм табулирования 
Применяется для составления всевозможных таблиц, которыми могут быть как абстрактная таблица значений математической функции, так и конкретная таблица стоимости товара или платежей, совершенных абонентом сотового оператора.

В общем виде алгоритм можно описать так:

1. до цикла задается начальное значение управляющей переменной, условием выхода из цикла служит достижение управляющей переменной конечного значения;

2. в теле цикла на каждом шаге вычисляется очередное значение функции, зависящее от управляющей переменной, затем формируется строка таблицы;
3. в конце шага цикла значение управляющей переменной (обозначим ее x) изменяется оператором вида x:=x+d;, где d -- заданный шаг по управляющей переменной.

В качестве примера составим таблицу синусов в пределах от 0 до π с шагом по аргументу 0.25. Обозначим аргумент как x, значение синуса от x обозначим как y. В простейшем случае программа табулирования может выглядеть так:

var x,y:real;

begin

 writeln('x':10,'sin(x)':10);
  {печать заголовка таблицы до цикла}

 x:=0; {начальное значение аргумента}

 while x<=pi+1e-6 do begin

   y:=sin(x); {вычисление функции}

   writeln (x:10:2, y:10:2);

    {печать строки таблицы}

   x:=x+0.25; {шаг по x}

 end;

end.

"Расширим" задачу за счет использования произвольных границ изменения аргумента и произвольного шага, а также выполнения всех необходимых проверок корректности. Пусть, например, требуется составить таблицу значений следующей функции:

 значения a, b вводятся пользователем.

Напишем текст программы, сопроводив его соответствующими комментариями.

var x,f,a,b,dx:real;
 n:integer; {счетчик выведенных строк}

begin

 repeat {Цикл ввода с контролем 
  правильности значений: a,dx,b должны быть 
  числами, dx>0, a+dx должно быть меньше b}

  writeln ('Введите a,dx,b:');

  {$I-}read (a,dx,b);{$I+}

  if IoResult <> 0 then begin

   writeln ('Вы не ввели 3 числовых ',

    'значения, попробуем еще раз');

   continue;

  end;

  if (dx<=0) or (a+dx>=b) then begin

   writeln ('Вы не ввели допустимые ',

    'данные, попробуем еще раз');

   continue;

  end

  else break;

 until false;

 {Печать заголовка таблицы}

 writeln;

 writeln ('x':10,'f(x)':10);

 x:=a;

 n:=2; {2 строки уже использованы}

 while x<=b+1e-6 do begin

   {в условии цикла учитываем возможную 
    погрешность работы с real!}

  if x<=0 then f:=sqr(x)*x

  else f:=exp(1/3*ln(abs(x))); 
   {корень 3 степени взяли через exp и ln}

  writeln (x:10:2,f:10:2);

  n:=n+1;

  if n=24 then begin 
  {На экране консоли по умолчанию 25 строк}

   write ('Нажмите Enter...');

   reset (input); readln;

   n:=1;

  end;

  x:=x+dx;

 end;

 writeln ('Таблица выведена');

 reset (input); readln;

end.

Как видно из примера, основной порядок действий -- такой же, как в предыдущей задаче. Так как экран консоли по умолчанию содержит всего 25 строк, с помощью переменной n мы дополнительно контролируем число уже выведенных строк и делаем по заполнении экрана паузу до нажатия пользователем клавиши Enter.

Разумеется, другие изученные нами виды циклов также могут применяться при табулировании. Рассмотрим в качестве примера следующую задачу.
Известна стоимость единицы товара. Составить таблицу стоимости 1, 2, ..., K единиц товара, значение K вводится. 
Так как число единиц товара -- заведомо целое, при программировании задачи будет удобен цикл for:

 var t:real;

    i,k:integer;

begin

 writeln;

 writeln ('Стоимость единицы товара:');

 read (t);

 writeln ('Количество единиц товара:');

 read (k);

 writeln ('Единиц':10,'Стоимость':10);

 for i:=1 to k do

  writeln (i:10,(i*t):10:2);

end.

Здесь для простоты мы исключили сделанные в предыдущем примере проверки. Стоимость единицы товара обозначена t, переменная i необходима для перебора возможных значений единиц товара в цикле for. Поскольку счетчик цикла for автоматически меняется с шагом 1, а оператором writeln можно выводить не только значения переменных, но и выражения, основной цикл программы состоит из одного оператора и не нуждается в операторных скобках.
11.2. Алгоритм организации счетчика

Этот алгоритм применяется, когда требуется подсчитать количество  элементов данных, отвечающих какому-либо условию или условиям. В общем виде алгоритм описывается следующим образом:

1. в разделе var описать переменную целочисленного типа, с помощью которой будет вестись подсчет;

2. до цикла присвоить ей начальное значение 0;

3. в теле цикла, если очередной элемент данных отвечает условию подсчета, увеличить эту переменную на 1 оператором вида k:=k+1;.

Необходимость присваивания начальных значений на шаге 2 этого и последующих алгоритмов связана с тем, что после описания в разделе var значение переменной еще не определено. "Пока мы не начали подсчитывать количество, оно равно нулю" -- этот очевидный для человека факт не очевиден для компьютера! Поэтому любой переменной, которая может изменяться в теле цикла, необходимо присвоить до цикла начальное значение, что и делает оператор вида k:=0;. 
Рассматриваемый нами алгоритм очень часто встречается в самых различных задачах, поэтому для "быстрой" записи операции по увеличению счетчика (она называется инкремент) или его уменьшению (декремент) существуют специальные стандартные процедуры:

Inc(X,N); -- увеличивает значение переменной.

Здесь параметр X -- переменная порядкового типа, а N -- переменная или выражение целочисленного типа. Значение X увеличивается на 1, если параметр N не определен, или на N, если параметр N определен, то есть Inc(X); соответствует X:=X+1;, а Inc(X,N); соответствует X:=X+N;.

Dec(X,N);  -- уменьшает значение переменной.

Параметр X -- также переменная порядкового типа, N -- целочисленное значение или выражение. Значение X уменьшается на 1, если параметр N не определен, или на N, если параметр N определен, то есть Dec(X); соответствует X:=X-1;, а Dec(X,N); соответствует X:=X-N;.

С помощью Inc и Dec генерируется более оптимизированный код, особенно полезный в сложных циклах. Возможно, мы будем использовать их не во всех примерах, но вам советую о них не забывать.

В качестве примера реализации алгоритма рассмотрим следующую задачу.
 Последовательность z(i) задана соотношениями 

, i=1,2,...,100. Найти количество элементов последовательности, больших значения 0.5.

Обозначив искомое количество за k, составим программу:

var z:real;

 i,k:integer;

begin

 k:=0;

 for i:=1 to 100 do begin

  if i mod 2 = 0 then z:=sqr(i)*cos(i)

  else z:=sin(i/2);

  if z>0.5 then inc(k);

 end;

 writeln ('Количество=',k);

end.

Так как шаг по переменной i равен 1, в программе использован цикл for, для проверки того, является ли значение i четным, использована операция mod.

В следующей задаче займемся обработкой данных по мере их ввода пользователем.
Известны оценки за экзамен по информатике для группы из n студентов, 2≤n≤25. Оценить количественную и качественную успеваемость группы по формулам:

, где k1 -- количество "троек", "четверок" и "пятерок", k2 -- количество только "четверок" и "пятерок".

Для ввода текущей оценки используем целочисленную переменную a, в качестве счетчика цикла for введем переменную i ("номер студента"), остальные величины описаны в условии задачи. При вводе значения n и очередного значения a для простоты не будем контролировать корректность вводимых данных.

var a,i,n,k1,k2:integer;

    ykol,ykach:real;

begin

 writeln;

 writeln ('Введите количество студентов:');

 read (n);

 k1:=0;

 k2:=0;

 for i:=1 to n do begin

  write ('Введите оценку ',i,' студента:');

  read (a);

  if a>2 then begin

   inc(k1);

   if a>3 then inc(k2);

  end;

 end;

 ykol:=k1/n*100;

 ykach:=k2/n*100;

 writeln 
('Количественная успеваемость=',ykol:6:2);

 writeln 
('Качественная успеваемость  =',ykach:6:2);

 reset (input); readln;

end.
11.3. Алгоритмы накопления суммы и произведения

Данные алгоритмы применяются, когда требуется сложить или перемножить выбранные данные. В общем виде эти широко применяемые алгоритмы можно описать так:

1. для подсчета каждой суммы или произведения описать по одной переменной того же типа, что суммируемые или перемножаемые данные;

2. до цикла переменной-сумме присвоить начальное значение 0, а произведению -- значение 1;

3. в теле цикла, если очередной элемент данных t отвечает условию суммирования или перемножения, сумма накапливается оператором вида s:=s+t;, а произведение -- оператором вида p:=p*t;
Очевидно, почему начальное значение произведения -- 1, а не 0. После оператора p:=0; оператор p:=p*t;, расположенный в теле цикла, будет возвращать только нули.

Рассмотрим типовую задачу. Для функции 

 найти арифметическое среднее ее положительных значений и произведение ненулевых значений.

Для поиска арифметического среднего необходимо сначала найти сумму s и количество k положительных значений функции. Составим следующую программу:

var x,f,s,p:real;

    k:integer;

begin

 s:=0; k:=0; p:=1;

 x:=-5;

 while x<=5+1e-6 do begin

  if x<0 then f:=sqr(ln(abs(x)))

  else if x>0 then f:=sin(sqr(x))

  else f:=0;

  if f>0 then begin

   s:=s+f;

   k:=k+1;

  end;

  if f<>0 then p:=p*f;

  x:=x+0.5;

 end;

 s:=s/k; {теперь в s - искомое среднее}

 writeln 
 ('Среднее положительных =',s:10:6);

 writeln 
 ('Произведение ненулевых=',p:10:6);

 reset (input); readln;

end.

В следующей задаче также применяется алгоритм накопления суммы. 
Требуется написать программу, имитирующую работу кассового аппарата: пользователь в цикле вводит цену очередного товара или 0 для завершения ввода, программа суммирует цены. По завершении цикла ввода программа начисляет скидку с общей стоимости товара по правилам: скидки нет, если общая стоимость покупки -- менее 10000 руб.; скидка равна 5%, если общая стоимость -- от 10000 до 20000 руб.; скидка равна 7%, если общая стоимость -- свыше 20000 руб. После начисления скидки выводится окончательная стоимость покупки.

Обозначив общую стоимость покупки s, а цену очередного товара -- t, напишем следующую программу:

var s,t:real;

begin

 writeln;

 s:=0; {начальное значение суммы!}

 repeat

  writeln ('Введите стоимость товара или '

   '0 для завершения ввода:');

  {$I-}read(t);{$I+}

  if (IoResult<>0) or (t<0) then begin

   writeln ('Ошибка! Повторите ввод');

   continue;

  end;

  if t=0 then break;

  {Округляем t до 2 знаков после запятой –
   на случай, если есть копейки}

  t:=round (t*100) / 100;

  s:=s+t; {накопление суммы}

 until false;

 {Начисление скидки и вывод ответа}

 writeln ('Стоимость без скидки:',s:8:2);

 if s>20000 then s:=s-s*0.07

 else if s>10000 then s:=s-s*0.05;

 writeln ('Стоимость со скидкой:',s:8:2);

 writeln ('Спасибо за покупку!');

 reset (input); readln;

end.

Тип данных real выбран для s и t не случайно -- выбор integer ограничил бы диапазон обрабатываемых значений и не позволил в удобном виде ввести копейки. Проверки корректности ввода, делаемые программой, знакомы по предыдущим примерам и поэтому не закомментированы.
12. Типовые алгоритмы поиска максимума и минимума

В этой главе мы изучим простейшие статистические алгоритмы, главный из которых -- определение максимального и минимального значений на множестве данных.

Рассмотрим алгоритм в общем виде:

1. описать для каждого максимума и минимума по одной переменной того же типа, что анализируемые данные;

2. до цикла максимуму присваивается либо заведомо малое для анализируемых данных значение, либо первый элемент данных; минимуму присваивается либо заведомо большое для анализируемых данных значение, либо первый элемент данных;

3. в теле цикла каждый подходящий для поиска элемент данных t обрабатывается операторами вида:
if t>max then max:=t; -- для максимума;
if t<min then min:=t; -- для минимума,
где max и min -- переменные, введенные для величин максимума и минимума соответственно.

Шаг 2 этого алгоритма требует комментариев, которые мы сделаем на примере поиска максимума. Очевидно, что сам алгоритм несложен -- каждый элемент данных t последовательно сравнивается с ячейкой памяти max и, если обнаружено значение t, большее текущего значения max, оно заносится в max оператором max:=t;. Как мы помним, после описания на шаге 1 переменной max, ее значение еще не определено, и может оказаться любым, откуда следует необходимость задания начального значения. Представим, что после выбора начального значения max, равного нулю, при анализе встретились только отрицательные значения элементов t. В этом случае условие t>max не выполнится ни разу и ответом будет max, равное нулю, что неправильно. Выбор заведомо малого начального значения max (например, значение -1E30, т. е., -1030, вряд ли встретится в любых реальных данных) гарантирует, что условие t>max выполнится хотя бы раз и максимум будет найден. Альтернативный способ -- присвоить переменной max значение отдельно вычисленного первого элемента последовательности данных. В этом случае ответ либо уже найден, если первый элемент и есть максимальный, либо будет найден в цикле.

Аналогичные рассуждения помогают понять, почему минимуму следует присваивать в качестве начального значения заведомо большое число.

Перейдем к примерам. Для функции y(x)=sin2(x), 

 найти минимальное среди положительных и максимальное значения.

Обозначив искомые значения min и max соответственно, напишем следующую программу:

var x,y,max,min:real;

begin

 x:=-pi/3;

 max:=-2;

 min:=2; {эти начальные значения 
  - заведомо малое и большое для синуса}

 while x<=pi/3+1e-6 do begin

  y:=sqr(sin(x));

  if y>0 then 
   {ищем min только среди положительных!}

   if y<min then min:=y;

  if y>max then max:=y;

  x:=x+pi/24;

 end;

 writeln ('Минимум =',min:8:2);

 writeln ('Максимум=',max:8:2);

 reset (input); readln;

end.

В следующем примере дополнительно сохраним значения аргументов функции, для которых найдены минимум и максимум.

Последовательность T(k) задана соотношениями T(k)=max(sin k, cos k), k=1, 2, ... ,31. Найти номера максимального и минимального элементов последовательности.

Поиск номеров не избавит нас от необходимости поиска значений. Поэтому, кроме переменных min и max, нам понадобятся две целочисленные переменные для хранения номеров минимального и максимального значений, обозначим их kmin и kmax соответственно. Обратите также внимание, что на каждом шаге цикла дополнительно потребуется находить максимальное из значений sin(k) и cos(k), для занесения его в переменную t.

var t,max,min:real;

    k,kmin,kmax:integer;

begin

 min:=1e30;

 max:=-1e30;
 {задаем "надежные" значения, 
  близкие к плюс и минус бесконечности}

 for k:=1 to 31 do begin

  if sin(k)>cos(k) then t:=sin(k)

  else t:=cos(k);

  if t<min then begin

   {по условию нужны 2 оператора – 
    сохранение нового мин. значения

    и сохранение номера элемента, 
    отсюда операторные скобки!}

   min:=t; kmin:=k;

  end;

  if t>max then begin

   max:=t; kmax:=k;

  end;

 end;

 writeln ('Номер мин. элемента =',kmin);

 writeln ('Номер макс. элемента=',kmax);

 reset (input); readln;

end.
13. Решение учебных задач на циклы

Применяя полученные навыки работы с типовыми алгоритмами, рассмотрим несколько учебных задач различных типов.

На практике нередко встречаются задачи, для которых число шагов цикла заранее неизвестно и не может быть вычислено. Как правило, в этом случае расчетный цикл  работает до выполнения некоторого условия, определяющего достижение требуемого результата. Приведем пример подобной задачи.

Задана функция и ее разложение в ряд:

 n=0, 1, 2, ... (здесь n! обозначает факториал числа n, равный 1*2*3*...*n; при этом 0!=1). Найти число элементов ряда k, требуемое для достижения заданной точности ε.

Перед нами -- типичная задача на ряды. С ростом величины n слагаемые 

становятся все меньше, равно как и модуль разности между двумя соседними слагаемыми. Поэтому под достижением заданной точности будем понимать выполнение условия 

 где sn, sn-1  -- суммы ряда, вычисленные на текущем и предыдущем шагах цикла, а значение ε задается малым числом, от 10-6 и ниже. Для вычисления x2n введем переменную xn, которую на каждом шаге цикла будем домножать на x2, аналогично, для вычисления текущего значения факториала используем переменную nf. Иной подход потребовал бы большого количества повторных вычислений на каждом шаге цикла, плюс был бы чреват большими потерями точности. Для вычисления (-1)n было бы странно использовать формулу ax=exln a -- вполне достаточно завести целочисленную переменную znak, равную 1, которая на каждом шаге цикла будет менять свое значение на противоположное оператором znak:=-znak;. Кроме того, так как требуемое число шагов заранее неизвестно, используем для всех основных переменных более точный тип double вместо real.

Реализуем все сказанное в следующей программе:

{$N+} {Совместимость с процессором 80287 –

 для использования double}

var x,sn,sn1,xn,nf,eps:double;

    k,n:longint;

    znak:integer;

begin

 writeln ('Введите значение x:');

 read (x);

 writeln ('Введите требуемую точность:');

 read (eps);

 sn1:=0;

 sn:=0;

 k:=0;

 n:=0;

 xn:=1;

 nf:=1;

 znak:=1;

 repeat

  sn1:=sn; {Текущая сумма стала 
            предыдущей для следующего шага}

  sn:=sn+znak*xn/nf; {Выполняем шаг цикла}

  {Меняем переменные для следующего шага:}

  znak:=-znak;

  xn:=xn*sqr(x);

  nf:=nf*(n+1)*(n+2);

  n:=n+2;

  k:=k+1;

 until abs(sn1-sn)<eps;

 writeln ('Значение =',cos(x):20:16);

 writeln ('Предыдущая сумма=',sn1:20:16);

 writeln ('Текущая сумма=',sn:20:16);

 writeln ('Число шагов=',k);

 reset (input); readln;

end.

Даже не владея приемами работы с графикой (см. гл. 25), можно наглядно проиллюстрировать результаты своих расчетов. Рассмотрим в качестве примера задачу.
Сформировать на экране изображение функции

f(x)= cos(x)+ln(x)

на интервале [1, 5] с шагом 0.2.

Программа с комментариями приводится далее.
var x,y,{ Значение аргумента и функции }

    a,b,{ Интервал }

    ymin,ymax:real; { Мин. и макс. y }

    cy, { Позиция столбца на экране }

    i:integer;
begin

 a:=1;

 b:=5;

 { Первый цикл нужен, чтобы определить
   минимальное и максимальное значения

   функции на заданном интервале }

 x:=a;

 ymin:=1e20; ymax:=1e-20;

 while x<=b do begin 
  y:=cos(x)+ln(x);
  if y<ymin then ymin:=y
  else if y>ymax then ymax:=y;
  x:=x+0.2;
 end; 
 { Во втором цикле обработки можно выводить
   на экран значения функции }

 x:=a;

 while x<=b do begin

  y:=cos(x)+ln(x);

  cy:=round(8+(y-ymin)*(70/(ymax-ymin)));

  { Чтобы пересчитать y из границ

    [ymin,ymax] в границы [8,78] (столбцы
    экрана, в которые выводим график),
    используем формулу cy=  8 + (y-ymin)*
     (78-8)/(ymax – ymin) }

  writeln;
  write (x:7:3,' '); 
   { Выводим очередной x и пробел }

  for i:=8 to cy do write ('*'); 
   { Рисуем звездочками значение y }

  x:=x+0.2; { Переходим к следующему x  }
 end;

end.

В заключительном примере этого пункта проверим, является ли введенное положительное целое число N простым. 
Простое число делится без остатка только на единицу и само на себя. В цикле последовательно проверим остатки от деления N на числа 2, 3, ... ,N/2. Если найден хотя бы один остаток, равный нулю, число не является простым и дальнейшие проверки бессмысленны. Ясно также, что не имеет смысла проверять, есть ли целые делители, которые больше половины исходного числа, поэтому верхняя граница цикла = N/2 (для деления используем операцию div). Обратите внимание на использование логической переменной-флага s в этой программе.

var n,i:longint; s:boolean;

begin

 write ('N='); readln (n);

 s:=true;

 for i:=2 to n div 2 do

  if n mod i = 0 then begin

   s:=false; break;

  end;

 if s=true then writeln ('простое')

 else writeln ('НЕ простое');

end.

14. Одномерные массивы. Описание, ввод, вывод и обработка массивов на Паскале

Массивом называют упорядоченный набор однотипных переменных (элементов). Каждый элемент имеет целочисленный порядковый номер, называемый индексом. Число элементов в массиве называют его размерностью. Массивы используются там, где нужно обработать сразу несколько переменных одного типа -- например, оценки всех 20 студентов группы или координаты 10 точек на плоскости. Строку текста можно рассматривать как массив символов, а текст на странице -- как массив строк.

Массив описывается в разделе var оператором следующего вида:

var ИмяМассива: array [НИ .. ВИ] of Тип;

Здесь

НИ (нижний индекс) -- целочисленный номер 1-го элемента массива;
.. -- оператор диапазона Паскаля (см. п. 7.8);
ВИ (верхний индекс) -- целочисленный номер последнего элемента;
Тип -- любой из известных типов данных Паскаля. Каждый элемент массива будет рассматриваться как переменная соответствующего типа.

Опишем несколько массивов разного назначения.
var a: array [1..20] of integer;

Здесь мы описали массив с именем A, состоящий из 20 целочисленных элементов;

var x,y : array [1..10] of real;

Описаны 2 массива с именами x и y, содержащие по 10 вещественных элементов;

var t : array [0..9] of string;

Массив t состоит из 10 строк, которые занумерованы с нуля.

Легко увидеть, что размерность (число элементов) массива вычисляется как ВИ – НИ + 1.
Для обращения к отдельному элементу массива используется оператор вида ИмяМассива [Индекс].
Здесь Индекс -- целочисленный номер элемента (может быть целочисленным выражением или константой). Индекс не должен быть меньше значения нижнего или больше верхнего индекса массива, иначе возникнет ошибка "Constant out of range". Отдельный элемент массива можно использовать так же, как переменную соответствующего типа, например:
A[1]:=1;

x[1]:=1.5; y[1]:=x[1]+1;

t[0]:='Hello';

В этой главе мы изучаем одномерные массивы, в которых каждый элемент имеет один номер (индекс), характеризующий его положение в массиве. В математике понятию одномерного массива из n элементов соответствует понятие вектора из n компонент: A = {Ai}, i=1, 2 ,..., n.
Как правило, ввод, обработка и вывод массива осуществляются поэлементно, с использованием цикла for.
Простейший способ ввода -- ввод массива с клавиатуры:
const n = 10;

var a: array [1..n] of real;

 i:integer;

begin
writeln ('Введите элементы массива');

for i:=1 to n do read (A[i]);

Размерность массива определена константой n, элементы вводятся по одному в цикле for -- при запуске этой программы пользователю придется ввести 10 числовых значений. При решении учебных задач вводить массивы "вручную", особенно если их размерность велика, не всегда удобно. Существуют, как минимум, два альтернативных решения.
Описание массива констант удобно, если элементы массива не должны изменяться в процессе выполнения программы. Как и другие константы, массивы констант описываются в разделе const. Приведем пример такого описания:

const a:array [1..5] of real=(

 3.5, 2, -5, 4, 11.7

);

Как видно из примера, элементы массива перечисляются в круглых скобках через запятую, их количество должно соответствовать его размерности.

Формирование массива из случайных значений уместно, если при решении задачи массив служит лишь для иллюстрации того или иного алгоритма, а конкретные значения элементов несущественны. Для того чтобы получить очередное случайное значение, используется стандартная функция random(N), где параметром N передается значение порядкового типа. Она вернет случайное число того же типа, что тип аргумента и лежащее в диапазоне от 0 до N-1 включительно. Например, оператор вида a[1]:=random(100); запишет в a[1] случайное число из диапазона [0,99].
Для того чтобы при каждом запуске программы цепочка случайных чисел была новой, перед первым вызовом random следует вызвать стандартную процедуру randomize;, запускающую генератор случайных чисел. Приведем пример заполнения массива из 20 элементов случайными числами, лежащими в диапазоне от -10 до 10:

var a:array [1..20] of integer;

 i:integer;

begin

 randomize;

 for i:=1 to 20 do begin

  a[i]:=random(21)-10;

  write (a[i]:4);

 end;

end.

Еще более удобный путь -- чтение элементов массива из текстового или двоичного файла. Об этом рассказывается в гл. 21 и 22.

К массивам применимы все типовые алгоритмы, изученные в теме "Циклы". Приведем один пример, в котором вычисляется сумма s положительных элементов массива.
var b:array [1..5] of real;

s:real; i:integer;

begin
writeln ('Введите 5 элементов массива');

for i:=1 to 5 do read (b[i]);

s:=0;

for i:=1 to 5 do if b[i]>0 then s:=s+b[i];

Вывод массива на экран также делается с помощью цикла for.
for i:=1 to 5 do write (b[i]:6:2);

Здесь 5 элементов массива b напечатаны в одну строку. Для вывода одного элемента на одной строке можно было бы использовать оператор writeln вместо write.
Существенно то, что если обработка массива осуществляется последовательно, по 1 элементу, циклы ввода и обработки зачастую можно объединить, как в следующем примере.
Найти арифметическое среднее элементов вещественного массива t размерностью 6 и значение его минимального элемента.

var b:array [1..6] of real;

 s, min:real;

 i:integer;

begin

s:=0; min:=1e30;

writeln ('Ввод B[6]');

for i:=1 to 6 do begin

 read (b[i]);

 s:=s+b[i];

 if b[i]<min then min := b[i];

end;
writeln ('min=',min,' s=', s/6);

end.

Теоретически в этой программе можно было бы обойтись и без массива -- ведь элементы b[i] используются только для накопления суммы и поиска максимума, так что описание массива вполне можно было заменить описанием вещественной переменной b. Однако, в реальных задачах данные, как правило, обрабатываются неоднократно и без массивов обойтись трудно. Приведем пример учебной задачи, где использование массива дает выигрыш за счет уменьшения объема вычислений, выполняемых программой.

Задана последовательность Ti = max {sin i, cos i}, i= -5, -4, ..., 5. Найти элемент последовательности, имеющий минимальное отклонение от арифметического среднего положительных элементов.

Здесь в первом цикле можно сформировать массив по заданному правилу и найти арифметическое среднее положительных элементов. Во втором цикле, когда среднее известно, можно искать отклонение. Без использования массива нам пришлось бы считать элементы последовательности дважды.

var t : array [-5..5] of real;

 i,k:integer;
 s, ot : real;

begin
 s:=0; k:=0;

 for i:=-5 to 5 do begin

  t[i]:=sin(i);

  if t[i]<cos(i) then t[i]:=cos(i);

  if t[i]>0 then begin

   k:=k+1; s:=s+t[i];

  end;

 end;

 s:=s/k;

 ot:=1e30;

 for i:=-5 to 5 do begin

  if abs(t[i]-s)<ot then ot:= abs(t[i]-s);

 end;

 writeln ('Ot=', ot:8:2);

end.

Распространена обработка в одной задаче сразу нескольких массивов. Приведем пример.
Координаты 10 точек на плоскости заданы массивами x={xi}, y={yi}, i=1, 2, ..., 10. Найти длину ломаной, проходящей через точки (x1, y1), (x2, y2), ..., (x10, y10), а также номер точки, лежащей дальше всего от начала координат.

При решении задачи используем формулу для нахождения расстояния между 2 точками на плоскости, заданными координатами (x1,y1) и (x2,y2): 

Обозначим через r расстояние между текущей точкой и следующей, Len искомую длину ломаной, Dist -- расстояние от текущей точки до начала координат, max -- максимальное из этих расстояний, Num -- искомый номер точки. 
var x,y : array [1..10] of real;

I, num:integer;

r, Len, Dist, max : real;

begin
 {найдем max при вводе данных}

 max:=0; {т.к. расстояние не м.б. <0 } 
 num:=1; {на случай, если все точки (0,0)}

 writeln ('Введите координаты 10 точек');

 for i:=1 to 10 do begin    
  read (x[i], y[i]);

  Dist:=sqrt (sqr(x[i]) + sqr (y[i]));

  if dist > max then begin

    max:=dist; {запомнили новое расстояние}
    Num:=i;    {и номер точки}

  end;

 end;

 writeln ('Номер точки=',num, 
  ' расстояние=',dist:8:2);

 Len:=0;{длина ломаной - сумма длин сторон}

 for i:=1 to 9 do begin 
         {у 10-й точки нет следующей!}

   r:= sqrt (sqr(x[i]-x[i+1])+
             sqr(y[i]-y[i+1]));

   Len:=Len+r;

 end;

 writeln ('Длина ломаной=',len:8:2);

end.

Приведем пример задачи формирования массива по правилу. 
Задан массив x из 8 элементов. Сформировать массив y по правилу

и найти количество его положительных элементов.

var x,y: array [1..8] of real;

 i,k:integer;

begin
 writeln ('Введите массив x из 8 эл.');

for i:=1 to 8 do begin

  read (x[i]);

  if i mod 2 =0 then y[i]:=4*x[i]

  else y[i]:=cos(2*x[i]);

end;

K:=0;

writeln ('Массив y');

 for i:=1 to 8 do begin

    if y[i]>0 then k:=k+1;

   write (y[i]:8:2);

 end;

 writeln;

 writeln ('K=',k);

end.

15. Решение типовых задач на массивы

В гл. 14 мы подчеркивали, что типовые алгоритмы, изученные в теме "Циклы", в полной мере применимы и к массивам. Занимая определенный объем оперативной памяти, однажды полученные элементы массива остаются доступными весь сеанс работы программы и не требуют повторного вычисления или чтения из файла. Это порождает круг приложений, связанных с типовой обработкой наборов данных -- вычисление математических и статистических характеристик векторов, объединение массивов или поиск нужных значений в них и т. д. Другие интересные примеры, такие как подсчет частоты встречаемости элементов в массиве, задача сортировки (упорядочения) данных, будут рассмотрены в гл. 16.

1. Найти сумму, скалярное произведение и длину двух векторов произвольно заданной размерности. Размерность не может превышать значения 100.

Предельную размерность массивов зададим константой size=100. Размерность реальных векторов a и b, с которыми работает программа, не обязательно столь велика, обозначим ее n и позаботимся о том, чтоб при вводе значения n соблюдалось соотношение 2≤n≤size. Введем вектор a с клавиатуры, а вектор b сгенерируем из случайных вещественных чисел, принадлежащих диапазону от 0 до 10. Для этого достаточно умножить на 10 значение, возвращаемое стандартной функцией random, если она вызвана без параметров: b[i]:=random*10;.
При написании кода мы применим изученный в гл. 8 прием контроля правильности вводимых данных с помощью директивы компилятора и стандартной функции IoResult.
Как известно из математики, сумма векторов a и b ищется поэлементно по правилу ci=ai+bi, скалярное произведение может быть вычислено по формуле 

const size=100;

var a,b,c:array [1..size] of real;

    n,i:integer;

    la,lb,s:real;

begin
 repeat
  writeln;

  write ('Введите размерность A и B, ',

   'значение от 2 до ',size,':');

  {$I-}readln (n);{$I+}

  if (IoResult<>0) or (n<2) or (n>size) 
   then
   writeln ('Неверный ввод, повторите')

  else break;

 until false;

 writeln ('Введите вектор A из ',
  n,' элементов:');

 for i:=1 to n do begin

  repeat

   write ('A[',i,']=');

   {$I-}readln (a[i]);{$I+}

   if IoResult=0 then break;

  until false;

 end;

 writeln ('Генерируется вектор B из ',
  n,' элементов:');

 randomize;

 for i:=1 to n do begin

  b[i]:=random*10;

  write (b[i]:8:2);

 end;

 la:=0;

 lb:=0;

 s:=0;

 writeln;

 writeln ('Вектор c=A+B:');

 for i:=1 to n do begin

  c[i]:=a[i]+b[i];

  write (c[i]:8:2);

  la:=la+sqr(a[i]);

  lb:=lb+sqr(b[i]);

  s:=s+a[i]*b[i];

 end;

 writeln;

 writeln ('Длина вектора A:       ', 
  sqrt(la):8:2);

 writeln ('Длина вектора B:       ',
  sqrt(lb):8:2);

 writeln ('Скалярное произведение:',s:8:2);

 reset (input); readln;

end.
Думаю, при анализе листинга вы обратили внимание еще на одну особенность программы -- на каждом шаге цикла ввода вектора a дополнительно выполняется цикл repeat...until, предназначенный для контроля правильности ввода элементов массива. Подробнее такие конструкции, называемые кратными циклами, будут изучены в гл. 16.

2. Задана выборка из n случайных чисел. Для n=100 определить математическое ожидание mx и дисперсию dx выборки по формулам: 

Две подсчитываемые величины -- самые распространенные статистические характеристики набора случайных данных. Фактически, математические ожидание характеризует арифметическое среднее данных выборки, а дисперсия -- среднеквадратичное отклонение данных от среднего значения. Для хранения данных используем массив x из 100 элементов, сами данные сгенерируем из случайных чисел, находящихся в диапазоне от -1 до 1 включительно. Поскольку алгоритм накопления суммы предполагает последовательное и однократное использование каждого элемента массива, в цикле генерации элементов массива могут быть подсчитаны значения 

, которые затем позволят найти mx и dx.

const n=100;

var x:array [1..100] of real;

    s,d:real;

    i:integer;

begin

 writeln;

 writeln ('Массив x[100]:');

 randomize;

 s:=0; d:=0;

 for i:=1 to n do begin

  x[i]:=random*2-1;

  write (x[i]:8:3);

  s:=s+x[i];

  d:=d+sqr(x[i]);

 end;

 s:=s/n; {теперь в s - мат. ожидание,}

 d:=d/(n-1)-sqr(s)/(n*(n-1)); 
         {а в d - дисперсия}

 writeln;

 writeln ('s=',s:8:4);

 writeln ('D=',d:8:4);

 reset (input); readln;

end.

3. Объединить 2 упорядоченных по возрастанию массива A и B в массив C.

Для решения этой задачи тоже достаточно одного прохода по массивам. Действительно, заведя для каждого из массивов a, b и c по собственному счетчику (обозначим их ia, ib и i соответственно), мы можем, в зависимости от истинности или ложности соотношения aia≤bib переписывать в элемент ci значение aia или bib. Остается проконтролировать, чтобы ни один из счетчиков не вышел за границу размерности своего массива. Увидеть детали вам поможет внимательный анализ листинга.

Для простоты не будем просить пользователя вводить с клавиатуры две неубывающих последовательности чисел -- ввод заменен генерацией случайных элементов, совмещенной с выводом генерируемых значений.

program m_concat;

const size=10;

      step=5;

var a,b:array [1..size] of integer;

    c:array [1..2*size] of integer;

    i,n1,n2,ia,ib,ic:integer;

begin

 writeln;

 repeat

  write('Размерность 1 массива ',

        '(от 2 до ',size,'):');

  read (n1);

 until (n1>1) and (n1<=size);

 randomize;

 a[1]:=random(step);

 write ('A= ',a[1],' ');

 for i:=2 to n1 do begin

  a[i]:=a[i-1]+random(step);

  write (a[i],' ');

 end;

 writeln;

 repeat

  write('Размерность 2 массива ',

        '(от 2 до ',size,'):');

  read (n2);

 until (n2>1) and (n2<=size);

 b[1]:=random(step);

 write ('B= ',b[1],' ');

 for i:=2 to n2 do begin

  b[i]:=b[i-1]+random(step);

  write (b[i],' ');

 end;

 writeln;

 ia:=1; ib:=1;

 write ('c= ');

 for i:=1 to n1+n2 do begin

  if a[ia]<=b[ib] then begin

   c[i]:=a[ia];

   if ia<n1 then Inc(ia)

   else begin

    a[n1]:=b[ib];

    if ib<n2 then Inc (ib);

   end;

  end

  else begin

   c[i]:=b[ib];

   if ib<n2 then Inc(ib)

   else begin

    b[n2]:=a[ia];

    if ia<n1 then Inc(ia);

   end;

  end;

  write (c[i],' ');

 end;

 writeln;

end.

Часть 2. Элементы профессионального программирования на Паскале

Часть 2 предназначена для углубленного изучения основ программирования на Паскале. Это углубление связано с такими актуальными темами, как применение вложенных циклов и многомерных массивов, написание сложных программ, состоящих из процедур и функций, обработка символьных и строковых величин, обмен данными с текстовыми и бинарными файлами, основы компьютерной графики.

16. Кратные циклы

Вложение одних операторов цикла в другие позволяет обработать многомерные данные, что необходимо в большинстве реальных задач.

16.1. Двойной цикл и типовые задачи на двойной цикл

В теле цикла может находиться любой оператор, в том числе и другой оператор цикла. Такая конструкция называется кратными или вложенными циклами. Число вложений циклов не ограничено. Если цикл содержит только один вложенный цикл, то он называется двойным циклом.

Цикл, который содержит вложенный цикл, называется внешним, вложенный цикл называется внутренним. Управляющая переменная внутреннего цикла всегда меняется быстрее, чем переменная внешнего. Это означает, что для каждого значения переменной внешнего цикла перебираются все значения переменной внутреннего. В случае вложения нескольких циклов это правило также действует: быстрее всего меняется переменная самого внутреннего цикла.

Поясним сказанное на примере. В конструкции, описанной ниже:

for i:=1 to 5 do begin

 {начало внешнего цикла}

 j:=-5;

 while j<=5+1e-6 do begin

  {начало внутреннего цикла}

  {тело двойного цикла}

  j:=j+0.5;

 end; {конец внутреннего цикла}

end; {конец внешнего цикла}

при i=1 переменная j последовательно принимает значения -5, -4.5, ... , 5, затем процесс повторится при i=2, 3, 4 и 5. Таким образом, за время выполнения каждого шага внешнего цикла for внутренний цикл while выполнится 21 раз.

Двойной цикл особенно широко распространен в программировании, так как идеально подходит для обработки табличной информации, работы с матрицами, задач сортировки и т. д. Рассмотрим несколько типовых задач, связанных с двойными циклами.

1. Для функции двух переменных f(x,y) = sin 2x + cos 4y построить таблицу ее значений при значениях аргументов x и y, меняющихся от 0 до 2 с шагом, равным 0.25.

Решение задачи возможно, как минимум, в трех вариантах. Они схематично изображены на рис. 16.1.
Варианты 1 и 2 отличаются лишь порядком следования внешнего и внутреннего циклов. Рассмотрим возможное решение для варианта 1.

Рис. 16.1. Таблицы значений функции 2 переменных

var x,y,f:real;

begin
 writeln ('x':8,'y':8,'f':8); 
 x:=0;

 while x<=2 do begin

  y:=0;

  while y<=2 do begin

   f:=sin(2*x)+cos(4*y);
   writeln (x:8:2,y:8:2,f:8:2);
   y:=y+0.25;

  end;

  x:=x+0.25;

 end;

end.

Как видно из примера, при решении задачи соблюдены все правила алгоритма табулирования из первой части курса: на каждом шаге двойного цикла выполняется вычисление значения функции, затем печать строки таблицы и переход к следующему значению аргумента. Обратите внимание, что внутренний цикл полностью содержится во внешнем, а инициализация переменной внутреннего цикла происходит на каждом шаге внешнего. Попробуйте переписать эту программу для реализации второго варианта расчета.

Вариант 3 удобен "естественным"  представлением таблицы значений функции двух переменных, однако, помешать его реализации может ограниченная ширина окна консоли (80 символов). Ниже приводится одна из возможных реализаций такой программы:

var x,y,f:real;

begin
 writeln; write (' ':6); 
  {оставляем место для печати значений y}

 x:=0;

 while x<=2 do begin 
  { для печати строки значений x 
    потребовался отдельный цикл }

  write (x:6:2);

  x:=x+0.25;

 end;

 y:=0;

 while y<=2 do begin

  writeln;{перевод строки для очередного y}

  write (y:6:2); {печать значения y}

  x:=0; {реализация внутреннего цикла по x}

  while x<=2 do begin
   f:=sin(2*x)+cos(4*y);

   write (f:6:2);

   x:=x+0.25;

  end;

  y:=y+0.25;

 end;

end.

Как видно из листинга, код несколько усложнился за счет дополнительного цикла формирования строки значений x, для того, чтобы строка таблицы помещалась на строке экрана, пришлось также уменьшить ширину поля вывода с 8 до 6 символов. Однако, выигрыш несомненен -- теперь вся таблица поместилась на одном текстовом экране. Ограничение на ширину экрана консоли не имело бы значения, например, при выводе таблицы значений в текстовый файл.

2. Не обязательно границы внешнего и внутреннего циклов четко фиксированы. На практике распространены задачи, когда граница внутреннего цикла переменна и зависит от текущего значения управляющей переменной внешнего цикла. Рассмотрим такую задачу.
Напечатать на первой строке вывода один символ звездочки, на второй -- два, на строке номер N -- N символов. Значение N вводится с клавиатуры, 1<N<25.

В общем виде условие можно сформулировать так: в строке номер i печатается i звездочек, при этом i меняется от 1 до N включительно. Разумеется, для перебора значений i мы используем цикл for. Однако для печати очередной строки звездочек нам потребуется вложенный цикл. Обозначив его счетчик как j, получим закон изменения j от 1 до i включительно:

var i,j,n:integer;

begin
 repeat {Ввод N с контролем ошибок}

  write ('N=');

  {$I-}read (n);{$I-}

  if (IoResult<>0) or (n<2) or (n>24) 
  then begin
   writeln ('Ошибка! N должно быть ',

            'числом от 2 до 24');

   continue;

  end;

  break;

 until false;

 {Двойной цикл, в котором вложенный 
  цикл имеет переменную границу}

 for i:=1 to n do begin

  writeln;

  for j:=1 to i do

   write ('*');

 end;

end.

3. Весьма распространена задача сортировки, при решении которой требуется упорядочить по возрастанию или по убыванию элементы данных. Рассмотрим типовую задачу сортировки одномерного массива, решаемую с помощью двойного цикла с переменной границей вложенного цикла.
Заданы N элементов массива A, упорядочить их по возрастанию значений.

Итак, нам требуется получить порядок элементов в массиве, при котором выполняется соотношение A1≤A2≤...≤AN. Достичь этого можно, сделав перестановки значений элементов, алгоритм перестановки нам известен еще из гл. 4. Представим, что в цикле мы сравниваем первый элемент A1 со всеми остальными Ai, i=2, 3, ... ,N, при необходимости (если A1>Ai) переставляя A1 и Ai. Таким образом, после выполнения этого цикла на первом месте в массиве гарантированно окажется его наименьший элемент. Для поиска следующего по величине элемента алгоритм можно повторить, сравнивая элемент A2 с остальными и при необходимости выполняя перестановку. Таким образом, приходим к двойному циклу следующего вида:

i=1,2,....,N-1 {до N-1 - так как у 
              элемента AN нет следующего}

 j=i+1,i+2,...,N {будем для Ai перебирать 
                все следующие элементы}

  {в теле двойного цикла сравниваем 
   Ai и Aj, при необходимости переставляем
   их значения}

Реализуем этот алгоритм в виде следующей программы:

const nmax=100; {максимальная размерность}

var a:array [1..nmax] of real;

    i,j,n:integer;

    b:real; {буферная переменная}

begin

 repeat {цикл ввода значения N с проверкой}

  writeln ('Введите N от 2 до ',nmax,':');

  {$I-}read (n);{$I+}

  if (IoResult<>0) or (n<2) or (n>nmax)
   then writeln ('Ошибка! Повторите ввод');

 until (n>1) and (n<=nmax);

 {Цикл ввода и печати исходного массива}

 randomize; 
 {Элементы массива выберем случайными}

 writeln ('Исходный массив:');

 for i:=1 to n do begin

  a[i]:= random(100);

  write (a[i]:4:0);

 end;

 {Двойной цикл сортировки}

 for i:=1 to n-1 do

 for j:=i+1 to n do

  if a[i]>a[j] then begin

    b:=a[i]; a[i]:=a[j]; a[j]:=b;

  end;

 {Цикл вывода измененного массива}

 writeln;

 writeln ('Отсортированный массив:');

 for i:=1 to n do write (a[i]:4:0);

 reset (input); readln;

end.

Сортировка массива по убыванию отличается лишь знаком в операторе сравнения элементов Ai и Aj.

Существуют и значительно более мощные алгоритмы сортировки, однако, их применение имеет смысл для действительно больших объемов данных.

4. Найти, сколько раз каждый элемент встречается в массиве.

Первое, что приходит в голову при решении этой распространенной задачи -- кроме массива данных ввести "массив частот", элементы которого будут хранить информацию о встречаемости  тех или иных значений. Однако решить проблему можно и без столь расточительных трат оперативной памяти (ведь размерность массива частот будет равна размерности исходного массива). Каждый элемент массива ai, кроме последнего, мы будем сравнивать со следующими за ним элементами aj, считая сколько раз выполнилось равенство ai=aj. Для того чтобы исключить повторную обработку уже встречавшихся элементов, учтенные значения нам придется заменять нулями. Поскольку значение ноль само может встретиться в массиве, дополнительный цикл перед обработкой подсчитает количество нулевых элементов.

Константа size носит то же назначение, что nmax в предыдущей задаче, а константа Diap задает диапазон значений, которые могут принимать элементы массива. Элементы массива следует сделать целочисленными, ведь задача сравнения вещественных чисел в общем случае некорректна (см. п. 7.2).

program m_chast;

const size=10;

      Diap=10;

var a:array [1..size] of integer;

    i,n,k,j:integer;

begin
 writeln;

 repeat
  write('Введите размерность 1 массива ',

        '(от 2 до ',size,'):');

  read (n);

 until (n>1) and (n<=size);

 randomize;

 a[1]:=random(Diap);

 write ('A= ',a[1],' ');

 for i:=2 to n do begin

  a[i]:=random(Diap);

  write (a[i],' ');

 end;

 writeln;

 k:=0;

 for i:=1 to n do if a[i]=0 then Inc(k);

 if k>0 then writeln ('0: ',k);

 for i:=1 to n-1 do if a[i]<>0 then begin

  k:=1;

  for j:=i+1 to n do if a[i]=a[j] then 
  begin

   a[j]:=0;

   Inc(k);

  end;

  writeln (a[i],': ',k);

 end;

end.

Подобный же принцип сравнения "каждый с каждым" используется при решении большого числа задач.

5. Координаты десяти точек на плоскости заданы двумя массивами xi, yi, i=1, 2, ..., 10. Найти две точки, расстояние между которыми минимально.

Действительно, нам придется искать расстояние от первой точки до всех остальных, от второй -- только до точек с номерами 2, 3, ..., 10, так как расстояние до первой уже известно и т. д. Попробуйте написать программу самостоятельно, применив двойной цикл из предыдущего примера.

6. В завершение темы рассмотрим задачу, где число вложений циклов больше двух.
Номера автобусных билетов состоят из 6 цифр. "Счастливыми" называют билеты, в номерах которых  сумма трех первых цифр равна сумме трех последних. Определить общее количество "счастливых" билетов.

Каждая из шести цифр может меняться от 0 до 9 включительно. Создав кратные циклы по шести переменным, воспользуемся известным нам алгоритмом организации счетчика:

var i,j,k,l,m,n:integer;

 kol:longint; 
    {типа integer может быть недостаточно}

begin

 kol:=0;

 for i:=0 to 9 do

 for j:=0 to 9 do

 for k:=0 to 9 do

 for l:=0 to 9 do

 for m:=0 to 9 do

 for n:=0 to 9 do

  if i+j+k = l+m+n then inc(kol);

 writeln 
  ('Количество счастливых билетов=',kol);

end.
16.2. Оператор безусловного перехода

С задачей досрочного выхода из кратных циклов свяжем изучение ранее неиспользуемого нами оператора безусловного перехода. Его общий вид следующий:

goto Метка;

Здесь Метка -- специальным образом указанное место в программе. Действие goto очень просто: независимо от каких-либо условий он осуществляет передачу управления на оператор, следующий за меткой. Каждая метка имеет вид

m:

где m -- имя, построенное по обычным правилам или целочисленная константа от 0 до 9999 включительно. Метку недостаточно поставить перед нужным оператором -- ее нужно еще и объявить в разделе описаний оператором 
label m;.
Можно объявить и несколько меток сразу, разделив их имена запятыми. Принято располагать оператор описания меток до оператора var. Таким образом, метки и оператор goto позволяют решить задачу быстрого непосредственного перехода к нужному месту в программе:

label 10,20;

var i,j:integer;

begin
 write ('Введите значение I (i>2)');

 readln (i);

 if i<3 then goto 10;

 j:=2;

 20:

 if i mod j = 0 then write (j,' ');

 j:=j+1;

 if j<i then goto 20;

10:

end.

Эта несложная программа позволяет пользователю ввести число i, большее двух, после чего печатает все числа от 2 до  i-1, на которые i делится без остатка (проверка остатка от деления i на j выполнена операцией mod). В случае ввода пользователем недопустимого значения i (меньшего трех) происходит переход на метку 10:, расположенную перед закрывающим оператором end. -- таким образом, программа сразу же завершается. Метка 20: позволяет организовать цикл без использования операторов while, repeat или for. Действительно, перед меткой 20: переменная j получает начальное значение 2, затем, если i делится на j без остатка, значение j выводится на экран, затем увеличивается на единицу и, если значение i не достигнуто, совершается переход на метку 20:, проверяющую уже новое значение j. Однако, и той, и другой цели можно было достигнуть без использования goto, применяя только операторы цикла и break;.

С правомерностью использования простого оператора goto связаны сложные и многолетние споры. Несомненно, что программу всегда можно написать, не прибегая к goto (иногда структурное программирование называют даже "программированием без goto"), не вызывает сомнения также, что использование меток и goto нарушает и усложняет восприятие структуры программы человеком. Однако автору представляется чрезмерным требование изгнать goto совсем -- в ряде ситуаций он просто удобнее, чем другие средства. Представим, что нам нужно досрочно выйти из двойного (или вообще кратного) цикла. Оператор break;, расположенный во внутреннем цикле, прервет его выполнение, но выполнение внешнего цикла продолжится со следующего шага. Без goto выйти "из всех циклов сразу" можно разве что с применением специальных переменных-флагов:

var Flag:boolean; 
{переменная-флаг}

...

Flag:=false; {флаг установлен в "ложь"}

while true do begin {внешний цикл}

 while true do begin 
{внутренний цикл}

   ...

   if Условие then begin 
   {случилось что-то, требующее выхода 
    из обоих циклов сразу}

    Flag:=true; 
    {ставим флаг, сообщая внешнему циклу,
     что требуется завершение}

 
 break; {выход из внутреннего цикла}

   end;

 end;

 if Flag=true then break; 
 {внешний цикл на каждом шаге должен
  отслеживать состояние флага}

end;

Если цикл тройной -- при таком подходе понадобится уже два флага, и так далее. Альтернативный путь -- установить  максимальные значения управляющих переменных внешних циклов -- также не соответствует канонам структурного программирования. С goto все выглядит проще и естественней:

label go1;

...

while true do begin 
 while true do begin
  ...

  if Условие then goto go1; 
   {вышли из двух циклов сразу!}

 end;

end;

Go1:

... 
Второй аспект возможного использования goto связан с написанием сложных подпрограмм (см. гл. 18), способных завершаться в различные моменты своего выполнения. Зачастую удобнее, чтоб подпрограмма имела всего одну "точку выхода" -- свой последний оператор, а досрочный выход осуществляется установкой возвращаемого значения и переходом на последний оператор с помощью goto.

Ограничения на использование goto следующие: метка должна находиться в том же блоке программы, что и оператор goto. Таким образом, нельзя перейти из одной части сложной программы (такой, как процедура или функция, изучаемые в гл. 18) в другую. Несомненно, что эти ограничения введены разработчиками, чтоб не усложнять и без того сложное восприятие программ, злоупотребляющих данным оператором.

17. Матрицы и типовые алгоритмы обработки матриц

Многие задачи связаны с обработкой многомерных массивов данных. Наиболее распространены при этом двумерные массивы.

В математике двумерные массивы представляются матрицей:

или, в сокращенной записи,

где n -- число строк матрицы, m -- число столбцов, i, j -- индексы (номера) текущих строки и столбца, на пересечении которых находится элемент 

Как и другие объекты данных, матрица описывается в разделе var. От вектора ее описание отличается тем, что в квадратных скобках перечисляются два оператора диапазона, указывающие, соответственно, нумерацию строк и столбцов матрицы:
var ИмяМатрицы: array 
  [n1..n2, m1..m2]  of  Тип;

Здесь n1..n2 -- диапазон значений номера строки, n1 и n2 -- целочисленные константы, m1..m2 -- диапазон значений номера столбца, значения m1 и m2 также целочисленные.

Как и векторы, матрицы могут быть образованы из элементов любого существующего в языке типа данных.

Под матрицу из n строк и m столбцов выделяется область памяти размером n*m*k байт, где k -- размер в байтах одного элемента. Для известных нам типов данных этот размер можно узнать из табл. 2.1. В оперативной памяти матрица хранится всегда построчно.

Например, для матрицы A, описанной оператором вида

var A:array [1..5,1..4] of real;

выделяется 20 ячеек памяти по 6 байт, причем в следующей за элементом A1,4 ячейке хранится значение элемента A2,1. 
Обращение к отдельным элементам матрицы осуществляется с помощью переменной с двумя индексами, например:

ai,j


a[i,j]

a2,1


a[2,1]

a2n,k


a[2*n,k]

Первый индекс, как и в математике, всегда показывает номер строки, а второй -- номер столбца.

Поскольку адресация памяти в любом случае линейна, следует понимать матрицу как удобный для программиста структурный тип данных. В отдельных случаях использование матрицы может быть заменено использованием вектора с тем же количеством элементов: так, матрице An,m всегда может быть сопоставлен вектор b из n*m элементов, а обращение к элементу A[i,j] при нумерации строк и столбцов с единицы может быть заменено на обращение к элементу b[(i-1)*m+j].

Подобно тому, как любая последовательная обработка вектора выполняется в цикле for, обработка матрицы осуществляется в двойном цикле for:

for i:=1 to n do
for j:=1 to m do
  {Обработка элемента A[i,j]}

Согласно правилу выполнения кратных циклов, переменная j меняется в этом двойном цикле быстрее, чем i, таким образом, обработка всех элементов матрицы будет выполнена построчно. Для последовательной обработки всех элементов матрицы по столбцам достаточно поменять местами циклы по i и j:

for j:=1 to m do
for i:=1 to n do
  {Обработка элемента A[i,j]}

Теоретически мы могли бы решить эту же задачу и перестановкой индексов в обращении к элементу матрицы (использовать запись A[j,i] вместо A[i,j]), однако, во избежание путаницы, делать этого не рекомендуется.

Приведем примеры использования двойного цикла for для ввода и вывода элементов матрицы. Пусть матрица c размерностью 4(2 (как мы помним, это означает, что в матрице 4 строки и 2 столбца) описана оператором вида

var c:array [1..4,1..2] of real;
В программе предусмотрено также 2 целочисленных счетчика для строк и столбцов матрицы:

var i,j:integer;
В этом случае типовой ввод матрицы с клавиатуры мог бы выглядеть так:

writeln ('Введите матрицу c ',

         'размерностью 4*2');

for i:=1 to 4 do
for j:=1 to 2 do read (c[i,j]);

Иногда удобнее печатать отдельное приглашение к вводу каждого элемента:

writeln ('Введите матрицу c[4,2]');

for i:=1 to 4 do

for j:=1 to 2 do begin

  write ('c[',i,',',j,']=');

  readln (c[i,j]);

end;

Например, в качестве приглашения к вводу элемента c1,1 на экране будет напечатано:

c[1,1]=

Оператор readln используется, поскольку элементы вводятся по одному.

Как и для векторов, возможны альтернативные способы ввода -- описание матрицы в разделе констант и формирование ее элементов из случайных чисел. Приведем пример описания матрицы констант размерностью 2(3:

const d:array [1..2,1..3] of integer=(

 (1,2,3),

 (4,5,6)

);

Как видно из примера, для описания матрицы констант создается список ее строк, каждая строка, в свою очередь, представляет собой список значений элементов.

Вывод на экран матрицы небольшой размерности бывает удобно реализовать в виде "одна строка матрицы на одной строке экрана". Для этого элементы матрицы во внутреннем цикле печатаются оператором write, не переводящим строку на экране, а во внешнем цикле выполняется writeln:

writeln ('Матрица c:');

for i:=1 to 4 do begin


writeln;


for j:=1 to 2 do write (c[i,j]:4:1);

end;
Разумеется, при выводе элементов матрицы можно использовать константы ширины и точности.

Базовые алгоритмы обработки матриц -- те же, что мы изучили в гл. 11-15. Отличие состоит в том, что реализацию этих алгоритмов можно условно рассматривать для двух типов задач:

· алгоритмы реализуются при обработке всех элементов матрицы;
· алгоритмы реализуются при обработке отдельных строк или столбцов матрицы.

В первом случае типовая учебная задача на матрицу отличается от аналогичной задачи с вектором лишь тем, что используются двойные циклы ввода, обработки и вывода.
1. Приведем пример реализации алгоритма первого типа.
В матрице A размерностью 4*4 найти сумму ее положительных элементов, произведение элементов, значения которых попадают в интервал [2, 10], а также отношение этих двух величин.

var a:array [1..4,1..4] of real;

    i,j:integer;

    s,p,z:real;

begin
 {Цикл ввода}

 writeln ('Ввод матрицы размерностью 4*4');

 for i:=1 to 4 do
 for j:=1 to 4 do read (a[i,j]);

 {Начальные значения и цикл обработки}

 s:=0;

 p:=1;

 for i:=1 to 4 do

 for j:=1 to 4 do begin

  if a[i,j]>0 then s:=s+a[i,j];

  if (a[i,j]>=2) and (a[i,j]<=10) 
   then p:=p*a[i,j];

 end;

 {Вывод результатов}

 writeln ('Сумма=',s:6:2);

 writeln ('Произведение=',p:6:2);
 if p<>0 then begin

  z:=s/p;

  writeln ('Отношение=',z:6:2);

 end

 else writeln ('p=0, отношение ',

               'вычислить нельзя');

 reset (input); readln;

end.

Поскольку о типе матрицы в условии задачи ничего не сказано, выбран "более общий" тип real. Способ ввода также выбран "по умолчанию" -- пользователь вводит значения элементов матрицы с клавиатуры. Искомые сумма, произведение и отношение обозначены, соответственно, s, p и z.

2. Рассмотрим пример реализации алгоритма второго типа. Отличие его в том, что алгоритм подсчета или поиска каждый раз заново реализуется во внутреннем цикле, таким образом, искомая в двойном цикле величина заново получает значение на каждом шаге внешнего цикла.

Задана матрица C размерностью 3(4. Найти номер строки, сумма элементов которой максимальна.

На этот раз заполним матрицу случайными числами. Обозначим s сумму элементов очередной строки, max -- максимальную из этих сумм. В качестве искомого номера строки введем целочисленную переменную imax, в которую будем заносить номер текущей строки i, если для этой строки выполняется соотношение s>max. Переменную s следует инициализировать на каждом шаге внешнего цикла по i (ведь сумма элементов ищется заново в каждой строке), тогда как переменная max ищется для всей матрицы и инициализируется один раз до двойного цикла:

var c:array [1..3,1..4] of real;

 i,j,imax:integer;

 s,max:real;

begin

 {Формирование матрицы с ее выводом}

 writeln;

 write ('Матрица C');

 for i:=1 to 3 do begin

  writeln;

  for j:=1 to 4 do begin

   c[i,j]:=random*10 
           {вещественные числа от 0 до 10};

   write (c[i,j]:6:2);

  end;

 end;

 {Поиск номера строки и вывод результата}

 max:=-1e30;

 for i:=1 to 3 do begin

  s:=0;

  for j:=1 to 4 do s:=s+c[i,j];

  if s>max then begin

   max:=s;

   imax:=i;

  end;

 end;

 writeln;

 writeln ('Номер строки=',imax);

end.

Аналогично могут быть реализованы типовые алгоритмы в столбце матрицы -- как описано выше, для этого достаточно поменять местами циклы по i и j. 
3. Встречаются также задачи, где результаты обработки строк или столбцов матрицы должны быть занесены в вектор. Решим задачу подобного типа.

В матрице S размерностью 20(4 приведены оценки 20 студентов за 4 экзамена последней сессии. Подчитать и вывести на экран элементы вектора, содержащего средние баллы каждого студента.

Обозначим искомый вектор как B. Его размерность очевидна, поскольку каждый элемент вектора зависит от одной строки матрицы.  Применив алгоритм обработки строк матрицы, получаем следующую программу:

var s:array [1..20,1..4] of integer;

    b:array [1..20] of real;

    i,j:integer;

begin

 {Матрица s из случайных оценок от 3 до 5}

 for i:=1 to 20 do

 for j:=1 to 4 do s[i,j]:=3+random(3);

 {Обрабатываем матрицу построчно, 
  формируя вектор B}

 for i:=1 to 20 do begin

  b[i]:=0;

  for j:=1 to 4 do b[i]:=b[i]+s[i,j];

  b[i]:=b[i]/4;

 end;

 {Выводим оценки со средними баллами}

 writeln;

 write ('Оценки':8,' Баллы');

 for i:=1 to 20 do begin

  writeln;

  for j:=1 to 4 do write (s[i,j]:2);

  write (b[i]:5:2);

 end;

end.

4. Нередко встречаются задачи, требующие обработки элементов, расположенных под или над главной диагональю матрицы. Главная диагональ характеризуется тем, что номера строк и столбцов расположенных на ней элементов совпадают. Корректно это понятие определено только для квадратных матриц с одинаковым числом строк и столбцов. Соответственно, для обработки элементов, лежащих на главной диагонали матрицы A размерностью n(n было бы достаточно цикла

for i:=1 to n do begin

 {Обработка элемента A[i,i]}

end;

Когда требуется обработать элементы, лежащие под главной диагональю (для них номер строки больше номера столбца), двойной цикл обработки мог бы выглядеть так:

for i:=1 to n do

for j:=1 to n do begin

 if i>j then begin

  {Обработка элемента A[i,j]}

 end;

end;

Приведенный код имеет тот недостаток, что в нем выполняются заведомо лишние шаги цикла. Действительно, для квадратной матрицы размерностью n(n мы выполняем n2 шагов двойного цикла, тогда как имеется всего 

 элементов под главной диагональю. Исправить ситуацию может двойной цикл с переменной границей, подобный изученному выше:

for i:=1 to n do

for j:=1 to i-1 do begin

 {Обработка элемента A[i,j]}

end;

Этот цикл выполняется ровно 

 раз. Для обработки только элементов, лежащих над главной диагональю, этот цикл будет выглядеть следующим образом:

for i:=1 to n do
for j:=i+1 to n do begin
 {Обработка элемента A[i,j]}

end;

5. Распространенный класс задач, связанных с умножением и обращением матриц, мы рассмотрим в гл. 18. Сейчас же ограничимся реализацией хорошо известного алгоритма умножения матрицы A размерностью n(m на вектор b размерности m. Напомним, что в результате умножения получается вектор размерности n (обозначим его c), а само умножение состоит в том, что каждая строка матрицы скалярно умножается на вектор по формуле 

 и результат заносится в компоненту ci. Применив известные нам типовые алгоритмы, напишем следующую программу:

const n=3; m=4; 
      {размерности матрицы и векторов}

var a:array [1..n,1..m] of real;

    b:array [1..m] of real;

    c:array [1..n] of real;

    i,j:integer;

begin
 {Формируем матрицу A из случайных чисел}

 writeln ('Матрица A');

 for i:=1 to n do begin

  for j:=1 to m do begin

   a[i,j]:=random*10;

   write (a[i,j]:6:2);

  end;

  writeln;

 end;

 {Аналогично для вектора b}

 writeln ('Вектор b');

 for j:=1 to m do begin

  b[j]:=random*10;

  writeln (b[j]:6:2);

 end;

 {Умножение совмещаем с выводом вектора с,
  это возможно, т.к. компоненты вектора
  ищутся последовательно}

 writeln ('Вектор c=A*b');

 for i:=1 to n do begin

  c[i]:=0; {c[i] - сумма}

  for j:=1 to m do

   c[i]:=c[i]+A[i,j]*b[j];

  writeln (c[i]:6:2);

 end;

end.

18. Подпрограммы

Написание любой большой программы невозможно как без разбиения задачи на менее сложные подзадачи, которые мы можем решать независимо, так и без повторного использования ранее написанного кода (представим, что каждая новая программа писалась бы "с нуля"?!). Решить эти важнейшие задачи позволяет механизм подпрограмм, имеющийся в любом языке программирования, в том числе, и в Паскале.

Подпрограммой называют независимую часть программы, предназначенную для решения некой подзадачи. Подпрограмма взаимодействует с основной программой через механизм параметров -- так называют входные и выходные данные, с которыми работает подпрограмма. Однажды написанная подпрограмма, выполненная с теми или иными значениями параметров, может решать некоторый класс задач.

Использование подпрограмм в чем-то похоже на расчеты с использованием математических или физических формул. Так, имея общие формулы решения квадратного уравнения, мы можем подставить вместо коэффициентов a, b и c любые числовые значения и решить любое конкретное уравнение. Аналогично, мы могли бы написать подпрограмму с входными параметрами a, b, c и выходными параметрами x1, x2 (найденные корни уравнения), а затем, применяя нужное число раз эту подпрограмму (однократное использование подпрограммы называется ее вызовом), найти корни любого количества квадратных уравнений.

Итак, использование подпрограмм позволяет решить следующие задачи:

· уменьшение размеров кода и экономия памяти за счет возможности неоднократного вызова одной и той же подпрограммы в рамках одной программы;

· лучшее структурирование программы за счет разбиения задачи на более простые подзадачи;

· эффективное повторное использование однажды написанного кода.

Рассмотрим общую структуру сложной программы, содержащей  две подпрограммы:

var раздел_описаний_1;

Заголовок Подпрограммы_1;

begin
 {Тело подпрограммы_1}

end;

Заголовок Подпрограммы_2;

begin
 {Тело подпрограммы_2}

end;

var раздел_описаний_2;

begin
 {Тело главной программы}

end.

Как видно из листинга, каждая подпрограмма имеет заголовок (по меньшей мере, в этом заголовке должно быть указано ее назначенное программистом имя) и тело, состоящее из операторов. Подобно телу цикла, тело подпрограммы заключено в операторные скобки begin ... end;. Обратите внимание, что в листинге два раздела описаний. Первый из них  расположен до обеих подпрограмм, второй -- после них перед телом главной программы. Данные, описанные в первом разделе -- глобальные, они доступны всем частям программы, расположенным ниже по ее тексту. Данные второго раздела описаний также глобальны, но доступны лишь главной программе, так как описаны непосредственно перед ней. Общее правило очень простое: подпрограммы "видят" все глобальные переменные, описанные выше их тела. Аналогично, без принятия специальных мер подпрограмма "видит" и может вызвать любую другую подпрограмму, расположенную выше нее по тексту программы. Здесь вторая подпрограмма  может вызвать первую, но не наоборот. Главная программа, как правило, расположенная последней, может вызвать все подпрограммы.

На практике не рекомендуется делать подпрограммы зависящими от глобальных данных, так как это снижает их переносимость (возможность повторного использования). Разумеется, любого из разделов описаний в конкретной программе может и не быть. Более того, поскольку подпрограмма -- отдельная и, в идеале, независимая часть программы, она может содержать собственный раздел описания локальных переменных, предназначенных лишь для ее нужд и невидимых из других частей программы. Например, для подпрограммы решения квадратного уравнения такой локальной переменной могла бы быть переменная d, предназначенная для вычисления дискриминанта уравнения. Поясним этот важный момент на примере:

var i:integer; 
    {глобальная переменная – 
     описана вне всех подпрограмм}

Заголовок Подпрограммы;

var i:integer; 
    {локальная переменная – 
     описана после заголовка подпрограммы}

begin
 {Тело подпрограммы}

end;

begin
{Тело главной программы}

end.

Описанная перед телом подпрограммы локальная переменная i не имеет никакого отношения к одноименной переменной, описанной выше. На время выполнения подпрограммы локальная переменная i вытесняет глобальную, делая значение последней недоступной. После завершения подпрограммы значение локальной i будет утеряно, а значение глобальной i никак от этого не изменится. Итак, глобальные переменные видимы (доступны) от точки их определения до конца файла. Локальные переменные доступны только в том блоке операторных скобок, в котором они описаны и во вложенных в него блоках. 
В Паскале существует два вида подпрограмм, которые мы изучим в этой главе -- процедуры и функции. В программе может быть произвольное количество как функций, так и процедур.
18.1. Процедуры

Общий вид подпрограммы-процедуры следующий:

procedure Имя 
           (Список формальных параметров); 
{выше - заголовок подпрограммы}

var описания локальных переменных; 
begin
 {Тело процедуры}

end;

Другие подразделы раздела описаний, такие как label, const и изучаемый в п. 18.3 оператор type, также могут присутствовать между заголовком и телом процедуры и действие их также будет локально -- то есть, определено лишь в рамках данной процедуры. Единственная правильная связь процедуры с "внешним миром", то есть, другими подпрограммами и главной программой -- указанный после имени список формальных параметров. В этом списке через запятую указываются входные и выходные параметры процедуры с указанием их типов данных. Входные параметры служат исходными данными для процедуры, а выходные определяют результаты ее вычислений, передаваемые в главную программу или другую подпрограмму. Перед выходными параметрами, измененные значения которых должны сохраняться после завершения процедуры, следует указывать ключевое слово var. Причины именно такого указания мы обсудим ниже.

Само по себе написание процедуры еще не вызывает выполнения никаких действий. Чтобы процедура сработала, ее нужно вызвать, записав в нужной точке программы имя процедуры со списком фактических параметров, которые будут подставлены на место формальных. Все это делается не так сложно, как звучит. Представим, что мы решили написать процедуру решения любого квадратного уравнения с именем Equation. Входными данными этой процедуры будут значения коэффициентов уравнения a, b и c, выходными -- найденные корни x1 и x2 (если они существуют). Кроме того, нам понадобится "временная" (а точней, локальная) переменная d, позволяющая процедуре хранить вычисленное значение дискриминанта:

procedure Equation (a,b,c:real;
                    var x1,x2:real);

var d:real;

begin
 d:=sqr(b)-4*a*c;

 if d>=0 then begin

  x1:=(-b+sqrt(d))/(2*a);

  x2:=(-b-sqrt(d))/(2*a);

 end;

end;

Поскольку все параметры этой процедуры -- вещественные, в ее заголовке нам хватило двух списков -- один для входных параметров a, b и c, другой -- для выходных x1 и x2 (обратите внимание на слово var перед этим списком). В случае отрицательного значения дискриминанта, значения x1 и x2 остаются неопределенными (что, вообще говоря, не совсем корректно), в противном случае они вычисляются и должны быть переданы в главную программу. Вызвать эту процедуру мы могли бы, например, так:

var a,b,c,d,x1,x2,x3,x4:real;

...

write ('Введите значения a,b,c:');

read (a,b,c);

Equation (a,b,c,x1,x2); 
Здесь мы попытались решить уравнение ax2+bx+c=0, значения a, b, c, ввел пользователь, ответы, если они вычислялись, будут помещены в переменные x1 и x2. При вызове вида
Equation (1,4,-3.5,x3,x4);

мы решили уравнение x2+4x-3.5=0, а ответы поместили в переменные x3 и x4. Еще один вызов 
Equation (4,1,-3.5,x3,x4);

позволяет решить уравнение 4x2+x-3.5=0, записав ответы в переменные x1 и x2. Наконец, четвертое обращение к процедуре
Equation (1,b,0,x1,x3);

решает уравнение x2+bx=0, помещая ответы в переменные x1 и x3.
И так далее. Суть в том, что при каждом вызове подпрограммы значения фактических параметров подставляются на место формальных и с ними производятся вычисления, предусмотренные операторами подпрограммы. Указанные требования называют согласованием параметров и описывают следующим образом: формальные и фактические параметры должны быть согласованы между собой по количеству, типу и порядку следования. Это означает, что количество формальных и фактических параметров должно быть одинаковым, при этом, при вызове процедуры каждый фактический параметр должен иметь тот же тип и занимать в списке то же место, что и соответствующий ему формальный параметр. Из сказанного следует, что нашу процедуру Equation можно вызвать только с пятью параметрами (а не тремя, семью или нулём), причем все эти параметры должны быть вещественными. Если формальный параметр является выходным (перед ним в заголовке процедуры указано ключевое слово var), то соответствующий фактический параметр не может быть константой (ведь значение константы нельзя изменить). Для больше наглядности опишем данное соответствие в табл. 18.1.
Табл. 18.1. Соответствие формальных и фактических параметров

	Формальный параметр в заголовке процедуры
	Соответствующий фактический параметр при вызове процедуры

	Переменная некоторого типа данных без ключевого слова var
	Переменная, константа, элемент массива или значение выражения того же типа данных

	Переменная некоторого типа данных с ключевым словом var
	Переменная или элемент массива того же типа данных

	Массив
	Массив


О массивах как параметрах подпрограмм будет подробно рассказано ниже, а сейчас обсудим более подробно различие между "входными" и "выходными" параметрами процедур. На самом деле первые называются параметрами-значениями, а вторые -- параметрами-переменными. Передача процедуре "входных" параметров называется передачей по значению -- в этом случае при входе в процедуру для фактического параметра, которым может быть константа, переменная или значение выражения, создается временная локальная переменная, в которую и помещается вычисленное значение фактического параметра. При выполнении процедуры значение этой переменной может изменяться, однако, после выхода из процедуры все изменения будут потеряны и значение фактического параметра останется без изменения:

procedure p1 (x:integer); 
      {для x создастся локальная копия}

begin

 x:=x+1; {значение копии увеличилось на 1}

 writeln ('x=',x); {и было выведено}

end;

var x:integer;

begin

 x:=3;

 p1(x);

 writeln ('x=',x);

 {после вызова с передачей параметра 
  по значению, x по-прежнему равно 3}

end.

Если формальный параметр процедуры помечен ключевым словом var как параметр-переменная, это означает, что передача фактического параметра осуществляется по адресу, то есть, в процедуру передается адрес того места в памяти, где находится фактический параметр. Таким образом, все выполняемые в процедуре изменения значения параметра-переменной будут выполнены непосредственно над значением фактического параметра. По той же причине формальному параметру-переменной может соответствовать только фактический параметр-переменная -- ведь значение константы нельзя изменить, а выражение не имеет конкретного адреса в памяти, вычисляясь каждый раз при выполнении программы:

procedure p1 (var x:integer);

 {получаем адрес переменной, 
  переданной в качестве x}

begin

 x:=x+1;   {значение x увеличилось на 1}

 writeln ('x=',x);     {и было выведено}

end;

var x:integer;

begin

 x:=3;

 p1(x);

 writeln ('x=',x);

 {после вызова с передачей параметра 
  по ссылке,x стало равно 4 – 
  оно было изменено процедурой p1}

end.

Таким образом, использование var в заголовке процедуры подчеркивает, что параметр является переменной и может изменяться этой процедурой.

Перейдем к примерам.

1. Для точки на плоскости с заданными координатами (x,y) найти расстояние l от точки до начала координат, а также длину окружности и площадь круга радиусом l с центром в начале координат. 
Обозначим длину окружности как c, а площадь круга -- s. Нетрудно заметить, что помимо выделения в отдельную процедуру расчета величин l, c и s, целесообразно написать также процедуру Get для ввода вещественного значения с предварительным выводом приглашения к вводу (нам придется вводить, по меньшей мере, 2 значения -- x и y), а также процедуру Put для печати вещественного значения с указанной шириной, точностью и строкой комментария (выводиться будут искомые величины l, c и s).

procedure Circle2 (x,y:real; 
 var l,c,s:real);

const pi=3.1415926;

 {для иллюстрации; на самом деле 
  есть функция pi}

begin
 l:=sqrt(sqr(x)+sqr(y));

 c:=2*pi*l;

 s:=pi*sqr(l);

end;

procedure Get (s:string; var x:real);

 {s - приглашение, 
  x - параметр-переменная, 
      вводимая с клавиатуры величина}

begin
 write (s,': ');

 readln (x);

end;

procedure Put (s:string; x:real);

 {s - комментарий к выводу, 
  x - выводимая величина}

begin

  writeln (s,'= ',x:8:3);

end;

var x,y,c,s,l:real;

begin

  writeln;

  Get ('Введите координату x',x);

  Get ('Введите координату y',y);

  Circle2 (x,y,l,c,s);

  Put ('Удаление от начала координат',l);

  Put ('Длина окружности',c);

  Put ('Площадь круга',s);

end.

Несмотря на то, что процедура Circle2 вызвана в этой программе однократно, она может пригодиться при повторном использовании кода. Исходными данными (параметрами-значениями) для этой процедуры являются координаты точки x и y, выходными данными (параметры-переменные) -- найденные значения l, c и s. 
При всей своей простоте, этот листинг иллюстрирует особенность любой грамотно составленной сложной программы: как правило, главная программа состоит лишь из вызовов подпрограмм, выполняющих всю необходимую работу. Кроме того, договорившись о параметрах процедур, такую программу мог составить и коллектив разработчиков.

2. Написать процедуру, печатающую первые N членов ряда Фибоначчи.

Члены ряда Фибоначчи вычисляются по формуле:

 F0  = 0,  F1  = 1,  FN  = FN-1   + FN-2,  N=2,3,...

Заметим, что кроме простой и красивой формулы, ряд Фибоначчи замечателен еще тем, что отношение двух его соседних членов дает в пределе константу "золотого сечения", широко используемую в самых различных расчетах. Для вычисления и вывода на экран N первых членов ряда Фибоначчи напишем процедуру Fibo с формальным параметром all, определяющим, сколько членов ряда требуется найти. В главной программе организуем цикл ввода значения фактического параметра N.

procedure Fibo (all:integer);

var n,n1,n2:longint;

{n – текущий элемент ряда,

 n1 – предыдущий, 
 n2 – вычисленный 2 шага назад}

    i:integer;

begin

 n1:=1; n2:=0;

 writeln;

 write (n2,' ',n1,' ');

 for i:=2 to all do begin

  n:=n1+n2;

  write (n,' ');

  n2:=n1; n1:=n; 
    {переприсвоили для следующего шага}

 end;

end;

var n:integer;

begin
 repeat
  writeln ('Введите число членов ряда ',

           'или 0 для выхода:');

  read (n);

  if n=0 then halt(1)

  else Fibo(n);

 until false;

end.

18.2. Функции

О параметрах-переменных часто говорят, что подпрограмма возвращает их значения, подчеркивая тем самым, что они являются или могут являться выходными данными некоторого вычислительного процесса. 
Распространены подпрограммы, требующие возврата всего одного выходного параметра, являющегося скаляром (то есть, единственным значением, а не вектором или матрицей). В этих случаях вместо подпрограммы-процедуры используется подпрограмма-функция, возвращающая скалярный результат своей работы в основную программу. Поэтому для функции в ее заголовке необходимо указать тип возвращаемого результата, а в теле функции должен присутствовать хотя бы один оператор присваивания, в левой части которого записывается имя функции:

function Имя 
 (Список формальных параметров) :
   ТипРезультата; 
{выше - заголовок подпрограммы}

var описания локальных переменных; 
begin
  {Тело функции}

   Имя:=Выражение1;

end;

Здесь Выражение1 должно иметь тот же тип, что и указанный в заголовке ТипРезультата, а оператор Имя:=Выражение1; не обязан быть последним в теле функции. Поскольку результат выполнения функции возвращается в основную программу через ее имя, то обращение к функции записывается аналогично стандартным функциям в виде операнда-выражения, стоящего справа от знака присваивания:

Результат:=Выражение2;

Выражение2 может состоять только из вызова функции вида Имя (список фактических параметров) или включать этот вызов как часть более сложного выражения., переменная Результат должна соответствовать типу функции. В записи выражения должны быть соблюдены все изученные ранее правила соответствия типов.

Все, сказанное выше о согласовании формальных и фактических параметров, а также о параметрах-значениях и переменных, в полной мере относится и к функциям.

Фактически, стандартные подпрограммы Паскаля также делятся на функции и процедуры, в зависимости от способа их вызова. Так, writeln относится к стандартным процедурам, а sin -- к функциям. Правда, некоторые из стандартных подпрограмм имеют переменное число параметров, что запрещено пользовательским подпрограммам.

Формально при использовании функций не запрещено возвращать более одного значения через дополнительные параметры-переменные, подобные тем, что мы изучили для процедур. Можно сказать, что функции в Паскале -- это процедуры, способные возвращать дополнительное скалярное значение. Во многих других языках программирования специального разделения на процедуры и функции нет.

Чтобы окончательно уяснить не-синтаксические различия между функциями и процедурами, обобщим их в табл. 18.2.
Табл. 18.2. Различия между процедурами и функциями

	Процедура
	Функция

	Вызывается отдельным оператором
	Вызывается из выражения справа от знака присваивания

	Использует параметры-значения и переменные
	Использует параметры-значения и переменные, дополнительно доступен параметр-переменная с именем, совпадающим с именем функции


Напишем и вызовем простейшую функцию, находящую максимальный из двух своих вещественных аргументов:

function max(a,b:real):real;

begin
 if a>b then max:=a

 else max:=b;

end;

Вызвать эту функцию мы могли бы различными способами:

var x,y,z,r,t:real;

...

read (x,y,z);

r:=max(x,y); 
С помощью функции вычислили максимальное из значений x, y и записали его в переменную r.
t:=max(max(x,y),z);
Максимальное из значений x, y, z записали в переменную t.
Последний вызов иллюстрирует, что, как и стандартные функции, функции, написанные программистом, могут вызываться из сколь угодно сложных выражений -- при условии, что будут соблюдаться правила соответствия параметров. Написанную нами функцию max нетрудно было бы реализовать и в виде процедуры:

procedure max (a,b:real; var c:real);

begin
 if a>b then c:=a

 else c:=b;

end;

Однако, при вызове этой процедуры постоянно пришлось бы передавать "лишний" параметр c, служащий только для хранения возвращаемого значения.

Мы уже знакомы с оператором halt;, позволяющим аварийно (или просто из бесконечного цикла) завершить программу. Немедленно завершить текущий блок, в том числе и подпрограмму-функцию или процедуру, позволяет оператор Exit;.

Перейдем к примерам.
1. Используя подпрограмму, вычислить сумму первых k членов ряда 1+1/n.

Сумма ряда -- это скаляр, естественным выглядит использование подпрограммы-функции. Применив известные алгоритмы, составим программу:

function sum (k:integer):real;

var i:integer; s:real;

begin

 s:=0;

 for i:=1 to k do s:=s+1/i;

 sum:=s;

end;

var k:integer; s:real;

begin
 write ('Введите число шагов:');

 readln (k);

 s:=sum(k);

 writeln ('Сумма=',s:8:3);

end.

Обратите внимание -- несмотря на то, что функция sum вычисляет единственную величину s, мы были бы не вправе написать в ней оператор 
for i:=1 to k do sum:=sum+1/i;,

поскольку sum -- это имя функции, и справа от знака присваивания оно было бы воспринято как попытка функции sum вызвать саму себя, причем, без соблюдения правил соответствия параметров. Поэтому нам понадобилась локальная переменная s.

Тем не менее, рекурсивные функции, вызывающие сами себя, существуют и будут кратко рассмотрены далее.

2. Вычислить значение выражения 

, где z(x)= sin 2x + ln |x|.

Очевидно, что повторное вычисление выражения z(x) с различными значениями аргумента x неэффективно, удобнее написать подпрограмму-функцию, считающую по формуле значение z(x).
function z(x:real):real;

begin

 z:=sin(2*x)+ln(abs(x));

end;

begin

 write ('Ответ=', (z(3)+2*sqr(z(5)))/
                  (z(0.1)+1):6:2);

 readln;

end.

Как видно из примера, использование функции позволило выполнить расчет единственным оператором (с учетом того, что оператор writeln может печатать значения вычисленных выражений).

3. В заключение раздела скажем несколько слов о рекурсии. Рекурсивными называют функции, способные повторно вызывать сами себя. С точки зрения программирования, в этом нет ничего удивительного -- просто при повторном входе в функцию в программном стеке создается ее новая копия и расчет выполняется заново с измененным значением параметра функции. Затем функция проверяет значение параметра, при необходимости изменяет его и вновь повторно вызывает сама себя. Разумеется, во избежание зацикливания, при некотором значении параметра должно быть предусмотрено завершение вычислительного процесса. Использование рекурсии целесообразно везде, где расчет следующего значения некоторой функции зависит от ее предыдущего значения. Так, классический пример на рекурсию -- расчет факториала (факториал целого положительного числа n, обозначаемый n!, равен произведению всех чисел от 1 до N включительно). Используя очевидную формулу n!=n*(n-1)!, напишем следующую рекурсивную функцию:

function Factorial (n:integer):longint;

begin

 if n<2 then Factorial:=1

 else Factorial:=n*Factorial(n-1);

end;

Тип функции longint использован, так как факториал очень быстро растет и уже 8!=40320, то есть, больше максимального значения типа integer. Эта функция, как нетрудно увидеть, при значении аргумента, меньшем двух, возвращает единицу, а в противном случае возвращает свой аргумент n, умноженный на факториал от n-1. Несложный тест функции мог бы выглядеть так:

begin
 writeln ('10!=',Factorial(10));

end.

Рекурсия удобна при составлении различного рода переборных стратегий. Так, игровая задача составления пути по "лабиринту" лучше всего решается рекурсивно, ведь всякий такой путь -- это шаг плюс новый путь, уменьшенный на один шаг.

При написании рекурсивных функций следует учитывать, что рекурсивные вызовы довольно быстро заполняют стек и грозят его переполнением. Поэтому основная проблема при написании такой функции -- отсечь заведомо лишние вложенные вызовы. Для этого используют дополнительные проверки условий, определяющих, целесообразно ли углублять дальше текущий уровень вложений.
18.3. Массивы в качестве параметров подпрограммы

Зачастую подпрограмма-процедура или функция должна обработать некоторый массив данных. Однако структурный тип массива нельзя непосредственно указать в списке формальных параметров. В этом случае тип массива объявляется заранее в разделе описаний с помощью подраздела type:

type ИмяТипа = ОписаниеМассива;

Здесь ИмяТипа -- даваемое программистом наименование нового типа данных, а ОписаниеМассива -- известный нам оператор описания вектора или матрицы. Например, если подпрограмма должна обрабатывать вещественные массивы, не превышающие по размерности десяти элементов, в разделе описаний программы имеет смысл сделать следующее объявление:

type vector = array [1..10] of real;

Поскольку новый тип данных vector предназначен для использования подпрограммами, указанное выше объявление располагается до всех подпрограмм.

В дальнейшем тип данных vector можно использовать как в разделе описания переменных
var a,b:vector;,
так и в списке формальных параметров функции или процедуры:

function work (a:vector):real;

Если по условию задачи в подпрограмме требуется при разных обращениях обрабатывать массивы различной размерности, то в разделе type объявляется тип массива с наибольшей из размерностей, встречающихся в задаче. Это чревато тем, что при обработке массивов меньшей размерности часть памяти останется неиспользованной. Кроме того, при обращении к подпрограмме в этом случае приходится отдельным параметром передавать фактическую размерность массива.

Допустим, с помощью подпрограммы требуется найти максимальные элементы матриц A3x4, B4x3, C4x5. С учетом максимальной из имеющихся размерностей, делаем объявление типа matrix в разделе описаний:

type matrix = array [1..4,1..5] of real;

Далее описываем конкретные матрицы как объекты нового типа данных matrix:

var a,b,c:matrix;

В качестве подпрограммы поиска значения максимального элемента матрицы напишем функцию max1 (функцию, а не процедуру, поскольку максимум -- скалярное значение). В эту функцию нам придется передать 2 дополнительных параметра-значения, определяющих количество строк n и количество столбцов m в конкретной матрице, переданной фактическим параметром:

function max1 (n,m:integer; 
               var a:matrix):real;

var max:real; i,j:integer;

begin
 max:=a[1,1];

 for i:=1 to n do

 for j:=1 to m do

  if a[i,j]>max then max:=a[i,j];

 max1:=max;

end;

Использование дополнительной локальной переменной max связано с теми же причинами, что в первом примере на функции. Обратите внимание, что, хотя функция max1 и не изменяет элементы своих матриц-параметров, в списке ее формальных параметров матрица указана как параметр-переменная. Для матриц и векторов это рекомендуется делать всегда, поскольку при указании параметра-переменной в матрицу передается только адрес того места памяти, где хранятся ее данные, а не копия всех элементов. В конечном счете, передача матриц и векторов формальным параметрам-переменным позволяет уменьшить размер генерируемого машинного кода.

Вызвать нашу функцию для матриц A, B и C мы могли бы так:

var ma,mb,mc:real;

...

ma:=max1(3,4,a);

mb:=max1(4,3,b);

mc:=max1(4,5,c);

Более удобную работу с массивами-параметрами предлагает механизм открытых массивов, рассмотренный в п. 18.4. Однако его непосредственно используют лишь для одномерных массивов.

Пока же рассмотрим примеры решения типовых задач.

1. Заданы вектора A и B, содержащие по 5 элементов. Используя подпрограмму, найти их скалярное произведение по формуле 

Поиск скалярного произведения реализуем в виде подпрограммы-функции scal.

type vector=array [1..5] of real;

function scal (n:integer; 
               var a,b:vector):real;

var i:integer;

    s:real;

begin

 s:=0;

 for i:=1 to n do s:=s+a[i]*b[i];

 scal:=s;

end;

var i:integer;

    a,b:vector;

    s:real;

begin
 writeln ('Вектор 1 из 5 элементов:');

 for i:=1 to 5 do read (a[i]);

 writeln ('Вектор 2 из 5 элементов:');

 for i:=1 to 5 do read (b[i]);

 s:=scal(5,a,b);

 writeln ('s=',s:10:3);

end.

2. Сформировать по введенному с клавиатуры вектору A размерности n вектор res, компонентами которого являются отклонения элементов A от их арифметического среднего (подобная задача уже решалась выше, расширим ее на случай вектора).

Задача предполагает написание, по меньшей мере, двух подпрограмм: функция Middle будет вычислять арифметическое среднее элементов вектора, а процедура Otkl – формировать по вектору A и ранее найденному среднему mid искомый вектор отклонений b. Компоненты вектора b при этом будут вычисляться по формуле 

. Поскольку о размерности векторов в задаче ничего не сказано, укажем в разделе type максимальную размерность, равную 100 элементам.

type vector= array [1..100] of real;

function Middle (n:integer; 
                 var a:vector):real;

var j:integer;

    res:real;

begin

 res:=0.0;

 for j:=1 to n do res:=res+a[j];

 Middle:=res/n;

end;

procedure Otkl (n:integer; mid:real; 
                var a,b:vector);

var j:integer;

begin

 for j:=1 to n do b[j]:=abs(a[j]-mid);

end;

var a,res: vector;

    i,n:integer;

    s:real;

begin

 write ('Размерность? ');

 readln (n);

 for i:=1 to n do begin

  write ('A[',i,']=');

  readln (a[i]);

 end;

 s:=Middle (n,a);

 Otkl (n,s,a,res);

 for i:=1 to n do 
  writeln ('res[',i,']=',res[i]:8:2);

end.

3. Используя подпрограмму, написать и проверить программу перемножения двух матриц.

Как известно, матрица A размерностью n(m может быть умножена на матрицу B размерностью m(p по следующей формуле: 

где ci,j -- элемент получающейся в результате перемножения матрицы С размерностью n(m. Из формулы видно, что для умножения двух матриц нам понадобится тройной цикл: внешний цикл по i перебирает строки матрицы A, вложенный в него цикл по j выбирает в матрице B очередной столбец, а самый внутренний цикл по l умножает строку матрицы A на столбец матрицы B, получая элемент ci,j. Напишем соответствующую процедуру mmul и тестовую программу для нее:

type matrix=array[1..10,1..10] of real;

var a,b,c: matrix;

    i,j,n,m,p: integer;

procedure mmul (n,m,k:integer; 
                var a,b,c:matrix);

var i,j,l:integer; s:real;

begin

 for i:=1 to n do

 for j:=1 to k do begin

     s:=0;

     for l:=1 to m do s:=s+a[i,l]*b[l,j];

     c[i,j]:=s;

 end;

end;

begin

 repeat

  writeln;

  write ('Введите количество строк ',

         '1 матрицы: ');

  readln (n);

  write ('Введите количество столбцов ',

         '1 матрицы: ');

  readln (m);

  write ('Введите количество столбцов ',

         '2 матрицы: ');

  readln (p);

 until (n>1) and (n<11) and (m>1)
        and (m<11) and (p>1) and (p<11);

 for i:=1 to n do begin

  writeln ('Введите строку ',i,
   ' матрицы A из',m,'элементов:');

  for j:=1 to m do read (a[i,j]);

 end;

 for i:=1 to m do begin

  writeln ('Введите строку ',i,
   ' матрицы B из',p,'элементов:');

  for j:=1 to p do read (b[i,j]);

 end;

 mmul (n,m,p,a,b,c);

 for i:=1 to n do begin

  writeln;

  for j:=1 to p do write (c[i,j]:10:3);

 end;

end.

4. Процедурно ориентированные программы для распространенных задач решения системы линейных уравнений методом  Гаусса, сортировки одномерного массива, поиска всех миноров второго порядка в квадратной матрице Вы можете найти и самостоятельно разобрать в Приложении 4.

5. В качестве еще одного развернутого примера на использование массивов в подпрограммах, разберем следующую задачу.
Имеется N городов, между которыми налажены пассажирские перевозки. Между какими городами самый активный пассажиропоток?

Количество городов обозначим константой cities. После математической формализации задачи, нетрудно заметить, что перевозки из города i в город j могут быть занесены в элемент матрицы ai,j, таким образом, требуется определить величину max(ai,j+aj,i), учитывая перевозки "туда" и "обратно". Для поиска максимального пассажиропотока достаточно двойного цикла с переменной границей по счетчику вложенного цикла. Как и в других программах, выделим в отдельные подпрограммы также типовые задачи ввода и вывода матрицы, а также ввода вещественного значения с контролем допустимых числовых границ ввода.

const cities=10;

type matrix=array [1..cities,1..cities] 
 of integer;

function max1 (n:integer; var a:matrix; 
 var imax,jmax:integer):integer;

var i,j,m,p:integer;

begin

 m:=a[1,2]; imax:=1; jmax:=2;

 for i:=1 to n do

 for j:=1 to n do

  if (i<>j) then begin

   p:=a[i,j]+a[j,i];

   if p>m then begin

    m:=p; imax:=i; jmax:=j;

   end;

  end;

 max1:=p;

end;

function readNumber (s:string;
 min,max:integer):integer;

var a:integer;

begin

 repeat

  write (s);

  {$I-}readln(a);{$I+}

  if IoResult<>0 then 
   writeln ('Ошибка, введено не число!')

  else if (a<min) or (a>max) then 
   writeln ('Ошибка, введенное число ',

   'принадлежит интервалу [',min, ',', 
   max, ']')

  else break;

 until false;

 readNumber:=a;

end;

procedure readMatrix1 (var n:integer; 
 var a:matrix);

var i,j:integer; s,s1:string;

begin

 n:=readNumber ('Введите число строк ',

  'и столбцов матрицы: ',2,cities);

 for i:=1 to n do
 for j:=i+1 to n do begin
  s:='A['; str(i,s1); s:=s+s1+',';

  str(j,s1); s:=s+s1+']=';

  a[i,j]:=readNumber (s,0,maxInt);

 end;

end;

procedure writeMatrix1 (s:string; 
 n:integer; var a:matrix);

var i,j:integer;

begin

 writeln (s);

 for i:=1 to n do begin

  for j:=1 to n do write (a[i,j]:7);

  writeln;

 end;

end;

var a:matrix;

    n,gorod1,gorod2:integer;

begin

 readMatrix1 (n,a);

 max1 (n,a,gorod1,gorod2);

 writeMatrix1 ('A=',n,a);

 writeln ('Самый большой пассажиропоток ',

  'между городами ',gorod1,' и ',gorod2);

 readln;

end.
18.4. Открытые массивы

Недостатки изученного ранее способа передачи массивов-параметров очевидны: во-первых, необходимость описания типа данных массива оператором type нарушает правило переносимости подпрограмм (действие подпрограммы становится зависимым от внешнего оператора), во-вторых, для указания реальной размерности передаваемых в подпрограмму массивов приходится использовать дополнительные параметры-значения, в-третьих, при обработке массивов меньшей, чем указано в операторе type размерности, неэффективно теряется оперативная память. В какой то мере исправить эти недостатки позволяет использование открытых массивов.

Способ подходит только для одномерных массивов. Использовать его с матрицами возможно, если интерпретировать матрицу как вектор (см. гл. 17).

Имеющиеся в программе векторы описываются в разделе var обычным способом, без определения типа type. В списке формальных параметров подпрограммы параметр-вектор указывается без диапазона размерности:

function sum(var x : array of real) : real;

При вызове подпрограммы фактический параметр-массив подставляется на место формального:

var a:array [1..5] of real;

 s:real;

...

s:=sum(a);

Открытым остается вопрос -- как отслеживать из подпрограммы размерность переданного массива? Для этого в Паскале существуют стандартные функции Low и High. Их единственным параметром передается идентификатор массива, Low возвращает самое низкое допустимое значение индекса, а High -- самое высокое. Если A -- одномерный массив, величины Low(A) и High(A) можно непосредственно применять как границы цикла for:

function sum(var x : array of real) : real;

var i:word; s:real;

begin
 s:=0;

 for i:=Low(x) To High(x) Do s:=s+x[I];

 sum:=s;

end;

Чтобы завершить пример, вызовем написанную функцию sum:

const a:array [1..5] of real=(1,2,3,4,5.5);

begin

 writeln (sum(a):6:1);

end.

Как правило, номер первого элемента открытого массива равен нулю, однако, надежнее все-таки указывать Low. Приведем пример программы, включающей подпрограмму с открытыми массивами в качестве параметров.

Найти количество элементов вектора x[7], попадающих в интервал [0, 3] и  количество элементов вектора y[5], попадающих в интервал [-1, 1].

Для ввода элементов массива с клавиатуры напишем процедуру Input, которой все-таки придется передавать размерность массива-параметра (ведь вводятся два вектора различной размерности). Поэтому в Input использован тот факт, что нумерация элементов открытого массива по умолчанию выполняется в нуля. Функции kol, вычисляющей количество элементов открытого массива, попадающих в интервал [x1,x2], достаточно стандартного указания Low и High:

var x:array [1..7] of real;

    y:array [1..5] of real;

    k1,k2,i:integer;

procedure Input (n:integer; 
 var a:array of real);

var i:integer;

begin

 writeln ('Enter ',n,' items of array:');

 for i:=0 to n-1 do read (a[i]);

end;

function Kol (var a:array of real;
 x1,x2:real):integer;

var i,k:integer;

begin

 k:=0;

 for i:=Low(a) to High(a) do

  if (a[i]>=x1) and (a[i]<=x2) then k:=k+1;

 Kol:=k;

end;

begin

 Input (7,x);

 Input (5,y);

 k1:=Kol(x,0,3);

 k2:=Kol(y,-1,1);

 writeln ('k1=',k1,' k2=',k2);

end.

Процедура Input могла бы быть реализована и без передачи фактической размерности отдельным параметром:

procedure Input (var a:array of real);

var i:integer;

begin

 writeln ('Enter ',High(a)-Low(a)+1,
  ' items of array:');

 for i:=Low(a) to High(a) do read (a[i]);

end;

{ . . . }

Input (x);

Input (y);
19. Множества и перечислимые типы

Материал этой и последующих глав пособия является отчасти "факультативным", он может выходить за пределы, предусмотренные "программой-минимум" изучения основ программирования, однако, для тех, кто собирается программировать всерьез, тщательное изучение этих глав будет весьма полезным.

Множества представляют собой заранее заданные программистом наборы возможных значений переменной. Применение множеств позволяет удобно обращаться с данными, имеющими фиксированный набор возможных значений, а также облегчает проверку того, попадают ли значения в нужный диапазон.

Для объявления множества достаточно записать оператор

set of тип;

где тип -- один из определенных программистом или предустановленных типов данных:

type charset: set of char;

var symbols: charset;

Вопреки этому примеру, стандартные типы данных мало применимы для множеств -- ведь исходный тип набора должен быть порядковым (см. п. 2.3) и не иметь более чем 256 различных значений с нижним и верхним пределом от 0 до 255 соответственно. Это связано с тем, что для хранения количества элементов множества выделяется только один байт оперативной памяти. Приведем другой пример для множества:

type charset= set of char;

var symbols:charset;

    c1:char;

begin

 symbols:=['A'..'Z','a'..'z'];

 write ('Put one symbol:');

 readln (c1);

 if c1 in symbols then writeln ('OK')

 else writeln ('Error');

end.

Из примера видно, что переменным типа множества можно присваивать список диапазонов  соответствующего типа (мы уже знакомились с диапазонами в п. 7.8). В дальнейшем переменная типа множества может быть использована для контроля правильности входных данных с помощью оператора in:

if c1 in symbols then writeln ('OK')
Слева от оператора in может быть указано выражение любого перечислимого типа T, а справа -- набор с типом, совместимым с типом T.

Любые совместимые по типам данных множества можно объединять операций "+", вычитать операцией "-" и пересекать операцией "*". При этом результаты операций с множествами  соответствуют правилам логики множеств:

· порядковое значение c находится в множестве A+B только в том случае, если c находится в A или B;

· порядковое значение c находится в множестве A-B только в том случае, если c находится в A и не находится в B;

· порядковое значение c находится в множестве A*B только в том случае, если c находится и в A, и в B.

Если самое маленькое порядковое значение, являющееся элементом результата операции с множеством обозначить A, а самое большое -- за B, то тип результата становится равным A..B.

В следующем примере множество латинских букв получается операцией сложения подмножеств латинских прописных и латинских строчных букв.

type Latin = set of 'A'..'z';

const SmallLatin : Latin = ['A'..'Z'];

      BigLatin : Latin = ['a'..'z'];

var LatinLetters : Latin;

 c:char;

begin

 LatinLetters := BigLatin + smallLatin;

 repeat
  write ('Введите символ или ',

         'пробел для выхода:');

  reset (input);

  readln (c);

  if c in LatinLetters then 
   writeln (c,' - латинская буква');

 until c=' ';

end.

Как в примере выше, полезно бывает создавать из множеств подмножества при указании конструктора, содержащего выражения диапазонов в квадратных скобках [ ... ]:

type Digits = set of 0..9; {Множество цифр}

     Letters = set of 'A'..'Z'; 
      {Множество латинских букв}

const EvenDigits : Digits = 
 [0, 2, 4, 6, 8]; 
 {Подмножество четных цифр}

      Vowels : Letters = 
 ['A', 'E', 'I', 'O', 'U', 'y'];

 {Подмножество гласных букв}

      HexDigits : set of '0'..'z' =
 ['0'..'9', 'A'..'F', 'a'..'f']; 
 {Символы 16-ричных чисел}

type shortWeekDays =
 (Pn,Vt,sr,ch,Pt,sb,Vs); 
 {Перечислимый тип "дни недели"}

const Holidays : set of shortWeekDays 
 = [sb, Vs];
 {Подмножество "Выходные" дней недели}

var wd:shortWeekDays; 
 {Переменная типа "Дни недели"}

    i:integer;

begin

 wd:=Pn;

 for i:=1 to 7 do begin

  if wd in Holidays then 
   writeln (ord(wd), ' - Выходной день')

  else writeln (ord(wd), ' - Будний день');

  Inc(wd);

 end;

end.

Тип данных shortWeekDays в этом примере является перечислимым типом. Перечислимые типы определяют упорядоченные наборы значений, перечисляя идентификаторы, которые обозначают эти значения. Их порядок следует из последовательности, в которой они были перечислены. Оператор перечисления имеет общий вид

type имя = (идентификатор,
 идентификатор,..., идентификатор);

Возможные значения перечисления, заданные оператором type, должны быть идентификаторами Паскаля, поэтому назвать дни недели по-русски в последнем описании type было бы невозможно.

Идентификаторы, указанные в определении типа, становятся константами перечислимого типа, первая константа имеет порядковый номер 0, вторая -- номер 1, и так далее:

type suit = (Club, Diamond, Heart, Spade);

При этом объявлении Heart является константой типа suit. Стандартная функция ord возвращает порядковый номер перечислимой константы, в нашем примере

ord(Club)     = 0

ord(Diamond)  = 1

ord(Heart)    = 2

Как показано в листинге, переменным перечислимого типа можно присваивать константы, входящие в описание типа и увеличивать их значения как любые порядковые числа оператором Inc(wd), но эти значения нельзя читать или записывать "напрямую" операторами семейства read/write. В качестве альтернативы их можно приводить к целочисленным значениям стандартной функцией ord, при этом всегда первая константа списка имеет значение 0 (в нашем случае -- константа Pn). Оператор in в листинге позволяет проверить, попадает ли величина в подмножество, созданное для элементов исходного типа множества. Таким образом, основное назначение множеств и перечислимых типов -- удобная для человека запись выражений с "понятными" названиями констант вместо чисел. С точки зрения компилятора данные типа множества и перечисления являются целочисленными величинами.
Для ограничения диапазона исходных данных можно также непосредственно объявить тип-диапазон:

type Hour=0..23; minute=0..59;

Здесь объявлены переменные типов "час" и "минута", для переменных этих типов будут проверяться ограничения на попадание в указанные при описании диапазоны значений. Если переменной типа-диапазона присвоено недопустимое значение, программа отреагирует на это сообщением “Constant out of range”.

20. Обработка символьных и строковых данных

Компьютер способен обрабатывать не только числовые данные. Задачи обработки символьных данных распространены не менее, а возможно и более чем чисто арифметические расчеты. В этой главе мы изучим работу с символами и строками средствами Паскаля.

20.1. Работа с символами 
Для работы с отдельными символами описываются переменные типа char:
var ch:char;

{ ... }

ch:='y';

Одна переменная типа char хранит информацию об одном коде ASCII-символа. Это можно использовать, например, для отслеживания действий пользователя по нажатию клавиш. Приведем пример одного из вариантов такой программы:

var ch:char;

begin

 repeat

  writeln;

  write ('Продолжить? (Y/N)');

  readln (ch);

  if (ch='Y') or (ch='y') then begin

   {Здесь программируется 
    нужный вычислительный процесс}

  end

  else if (ch='N') or (ch='n') then halt

   {Завершение программы}

  else writeln ('Нажмите Y или N');

 until false;

end.

Для работы с кодами символов существуют 2 основных функции:

function ord(x):char; 
— возвращает ASCII-код символа x
function chr(x : byte):char;
—  возвращает символ с указанным ASCII-кодом x.
Применим последнюю функцию для того, чтобы узнать, как выглядит таблица символов кодировки DOS, используемой Паскалем:

var i,j:integer;

begin

 writeln;

 write ('ASCII-коды [32..255]');

 for i:=2 to 15 do begin

  writeln;

  write (i*16:3,' ');

  for j:=0 to 16 do write(chr(i*16+j),' ');

 end;

 writeln;

 write ('ENTER для выхода...');

 readln;

end.

Здесь печатаются только символы с кодами от 32 до 255 включительно, т. к. первые 32 символа с кодами от 0 до 31 -- непечатаемые (например, табуляция, перевод строки).

Для определения того, попадает ли код символа в некоторый диапазон значений, удобно использовать оператор in, как это делает следующий фрагмент программы:

 write ('Введите символ: '); readln (ch);

 if ch in ['A'..'Z'] then 
  write ('Большая латинская;')

 else if ch in ['a'..'z'] then 
  write ('Малая латинская;')

 else if (ch in ['А'..'Я']) then 
  write ('Большая русская;')

 else if (ch in ['а'..'п']) or 
         (ch in ['р'..'я']) then 
  write ('Малая русская;')

 else if ch in ['0'..'9'] then 
  write ('Цифра;')

 else write 
  ('Это не алфавитно-цифровой символ;');

 write (' Код Вашего символа= ',ord(ch));

Работая с алфавитными символами, приходится отдельно учитывать ввод строчных и прописных букв. Удобнее сначала преобразовать все символы к прописным с помощью функции upcase:

var ch:char; i:integer;

begin

 repeat

  for i:=1 to random(72) do write ('*');

  writeln;

  write ('Продолжить? (Y/N)');

  readln (ch);

 until upcase(ch)='N';

end.

К сожалению, эта функция бесполезна при работе с символами русского и других национальных алфавитов, для ее замены напишем и протестируем собственную подпрограмму c названием upcase_ru:

procedure upcase_ru (var s:string);

var i,l,c:integer;

begin

 l:=length(s);

 for i:=1 to l do begin

  c:=ord(s[i]);

  if (c>=ord('а')) and (c<=ord('п')) 
   then c:=c-32

  else if (c>=ord('р')) and (c<=ord('я'))
   then c:=c-80;

  s[i]:=Upcase(chr(c));

 end;

end;

var s:string;

begin
 writeln ('Введите строку текста:');

 readln (s);

 upcase_ru (s);

 writeln ('Преобразованная строка:');

 writeln (s);

end.

Программа учитывает, что в кодировке DOS не все символы кириллицы закодированы идущими подряд числами (см. Приложение 1).

Кроме того, в программе уже применяется массив символов, которому в Паскале соответствует тип данных string (строка). Мы упоминали этот тип данных, но еще не работали с ним. Как раз строкам посвящен п. 20.2.
20.2. Работа со строками

Строка -- это массив символов, т. е., элементов типа char. Нумерация символов в строке всегда выполняется с единицы. В Паскале строке соответствует тип данных string. Строка описывается оператором следующего вида:
var Имя_строки : string [длина];

Если положительная целочисленная величина "длина" не указана, выделяется память под строку длиной до 255 символов. Приведем примеры описания строк:
var s1:string;

s2:string[20];

s3:array [1..20] of string;

Здесь s1 -- строка с длиной по умолчанию, s2 -- строка из 20 символов, s3 -- массив из 20 строк, каждая из которых может занимать до 256 байт памяти (дополнительный байт нужен для хранения длины строки).
Со строками можно выполнять операцию присваивания. Покажем это на примере описанных выше строк.
s1:='А.И. Иванов';

— строке s1 присвоено значение строковой константы.

s1[3]:=’В’;
— отдельному символу строки s1 присвоили символьную константу.
s2:='2009';
— строке s2 присвоили строку, являющуюся записью целого числа. Позднее такая строка может быть преобразована в число стандартной процедурой Val.

s3[1]:='Информатика';
s3[2]:='';

— s3 является строковым массивом. Его первому элементу присвоена строковая константа, второму -- пустая строка.

Для ввода строк с клавиатуры следует применять оператор readln, т. к. ввод строки должен завершиться нажатием клавиши Enter:
writeln ('Введите имя пользователя:');

readln (s1);

Для вывода строк на экран или принтер можно использовать как оператор write, так и writeln:
s2:='SUMMA';

write (s2)   ;
— на экран будет выведена строка "SUMMA".
writeln ('Сумма':10);
— будет выведена строка "_____Сумма" (5 пробелов перед словом) и курсор переведен на следующую строку экрана.
Оператор сложения "+" переопределен для строк таким образом, что выполняет их сцепление (конкатенацию):
s1:='2009' + ' год'; s2:='н.э.';
s3[3]:=s1+' '+s2;

После этих действий значение строки s1 будет равно ''2009_год", а строка s3[3] -- ''2009_год_н.э.".
Если при сложении строк превышена максимальная длина результирующей строки, лишние символы отсекаются. Для сцепления строк можно также использовать стандартную функцию concat.
Операция отношения "=" позволяет посимвольно сравнить строки. При этом действуют следующие правила:
· строки считаются равными только при одинаковом наборе символов и одинаковой длине;

· иначе происходит поэлементное сравнение символов по их кодам. При этом, согласно таблице ASCII-кодов (см. Приложение 1) старшинство отдельных символов следующее: '0' < '1' < ... < '9' < 'A' < ... < 'Z' < 'a' < ... < 'z' < символы кириллицы.
Остальные операции отношения также применимы к строкам.
В ядро Паскаля включен ряд стандартных подпрограмм для обработки строковых данных. Опишем их прототипы, то есть, перечислим заголовки подпрограмм с указанием типа и количества формальных параметров.
function Length (s:string):integer; 
— определяет и возвращает длину строки s в символах;

function copy 
 (s:string; N,L:integer): string;

— возвращает часть строки  s длиной L символов, начиная с позиции N;

procedure Insert 
  (s0:string; var s: string; N: integer);

— в строку s вставляет строку s0, начиная с позиции N;

procedure Delete 
  (var s:string; N,L:integer);

— в строке s удаляет L символов, начиная с позиции N;

function Pos (s0, s:string): integer;
— возвращает позицию, начиная с которой строка s0 содержится в строке s или значение 0, если s0 не содержится в s;

procedure str (x: числовой; var s:string);

— преобразует число x в строку s, параметр x может иметь любой числовой тип;
procedure Val (s:string; 
  var x: числовой; var error:integer);

— преобразует строку s в число x. Параметр x может иметь любой числовой тип. Параметр-переменная error служит для контроля правильности преобразования. Если преобразовать удалось, то error=0, иначе error будет равен номеру первого непреобразуемого символа строки s.

Приведенных стандартных подпрограмм достаточно для решения большинства несложных задач, связанных с обработкой текстовых данных. Рассмотрим ряд типовых задач на примерах.

1. Разобрать предложение на слова и вывести каждое слово на новой строке экрана. 
Алгоритм работы этой программы очень прост -- до тех пор, пока в исходной строке предложения s есть хотя бы один пробел, вся часть строки до пробела копируется в строковую переменную w (слово). Если пробелов уже нет (или не было изначально), то вся строка -- это одно слово. После обработки очередного слова (в нашем случае -- это вывод его на новую строку экрана оператором writeln) обработанная часть строки вместе с пробелом удаляются из s -- чтобы следующий шаг цикла не нашел то же самое слово. 
var s,w:string; {предложение и слово}
    p:integer;  {позиция пробела}
begin
 writeln ('Введите текст');

 readln (s);

 repeat
  p:=pos (' ',s);

  if p>0 then w:=copy (s,1,p-1)

  else w:=s;

  writeln (w);

  delete (s,1,p);

 until p=0;

end. 
Приведенная программа имеет ряд недостатков. Самый заметный из них -- не учтены дополнительные пробелы между словами, которые будут восприниматься программой как лишние "пустые слова". Избавимся от них с помощью следующего примера.

2. Удалить лишние пробелы между словами. 
Алгоритм решения задачи также несложен. Пока в строке есть два подряд идущих пробела, следует удалять один из них. После этого остается проверить, нет ли в начале и конце строки по одному лишнему пробелу. Приведем только основную часть программы:
repeat
  p:=pos ('  ',s);

  if p>0 then delete (s,p,1);

until p=0;

if s[1]=' ' then delete (s,1,1);

if s[length(s)]=' ' then 
  delete (s, length(s),1);

writeln (s);

Однако, в примерах 1 и 2 есть более существенный недостаток -- проходы по строке неоднократны, тогда как разбор на слова можно сделать и за один цикл сканирования строки. Попытаемся реализовать это в следующей программе.

3. Разобрать предложение на слова за один цикл сканирования строки.
Приведем полный текст программы, а затем -- комментарии.
var s,word:string;

 c,c0:char;

 i,l,start:integer;

 inword:boolean;

begin

 writeln ('Enter string:');

 reset (input); readln (s);

 s:=' '+s+' ';

 l:=Length (s);

 inword:=false;

 for i:=2 to l do begin

  c0:=s[i-1];

  c:=s[i];

  if (c0=' ') and (c<>' ') then begin

   inword:=true;

   start:=i;

  end;

  if c=' ' then begin

   if inword=true then begin

    word:=copy (s,start,i-start);

    writeln ('''',word,''' is word');

   end;

   inword:=false;

  end;

 end;

end.

По сути дела, у нашей программы всего 2 состояния -- внутри слова и вне его. Переключением состояний управляет флаг inword. Номер символа, с которого начинается очередное слово, запоминается в переменной start. Программа не учитывает знаки препинания и возможность разделения слов другими символами, кроме пробела. Тем не менее, избавившись еще и от функции copy, совершающей лишний проход по части строки  (например, сразу же накапливая слово word по мере сканирования), можно было бы получить действительно эффективный алгоритм. Как обычно, платой за эффективность алгоритма является сложность программы.

4. Подсчитать количество пробелов в строке.

Это пример реализует работу со строкой как с массивом символов.
var s:string; k,i:integer;

begin

 writeln ('text?');

 readln (s);

 k:=0;

 for i:=1 to length (s) do

  if s[i]=' ' then k:=k+1;

 writeln ('k=',k);

end.

21. Текстовые файлы

Для написания большинства сложных программ требуется обмен данными с файлами. В этой главе мы рассмотрим только текстовые файлы, для работы с бинарными и типизированными файлами см. специальную литературу и гл. 22.

21.1. Общие операции

Для работы с каждым файлом описывается переменная типа text:

var f:text;

Ее называют файловой переменной. Если программа обрабатывает несколько файлов одновременно, для каждого из них описывается такая переменная. Можно использовать и одну переменную для нескольких файлов, если они обрабатываются последовательно, и каждый следующий файл открывается после завершения работы с предыдущим. Вся работа с файлом происходит через файловую переменную, имя файла указывается только один раз следующим оператором: 
assign (f,'имя_файла');

где f -- ранее описанная файловая переменная.

Этот оператор предшествует открытию файла и связывает переменную с файлом на жестком или гибком диске. В качестве имени файла может быть указан абсолютный или относительный путь к файлу на жестком или сменном диске:

assign (f,'data.txt'); 
 -- будет открываться файл с именем data.txt из текущей папки;
assign (f,'a:\my.dat');
 -- будет открыт файл с именем my.dat из корневой папки дискеты.
Имя файла также может быть введено пользователем с клавиатуры:

var name:string; f:text;

begin
 writeln ('Введите имя файла:');

 readln (name);

 assign (f,name);

   .

Наконец, имя файла можно передать программе параметром командной строки (см. п.  5.3).

После связывания следует открыть файл. Каждый файл может быть открыт только в одном из трех режимов -- для чтения данных, для записи новых данных (при этом, если файл уже существовал, его прежнее содержимое будет стерто) или для добавления данных в конец файла. Если требуется сначала прочитать данные из файла, а потом переписать этот же файл, следует открыть файл для чтения, после чтения закрыть его и вновь открыть для записи. Открытие файла выполняется одним из трех операторов:

reset (f); -- открыть для чтения;
rewrite (f);  -- открыть для записи;
append (f);  -- открыть для добавления.
Чтение или запись данных осуществляется знакомыми нам операторами read, readln, write и writeln, но первым параметром этих стандартных процедур указывается имя файловой переменной:

var a,b,c:real; f1,f2:text;

begin
 assign (f1,'read.txt');

 assign (f2,'write.txt');

 reset (f1);   
  {открыли файл read.txt для чтения,}

 rewrite (f2); 
  {а файл write.txt для записи}

 read (f1,a,b); 
  {Прочитали 2 числа из файла read.txt}

 c:=(a+b)/2;

 writeln (f2,c:6:2);  {записали значение c
         и перевод строки в файл write.txt}

После того, как все операции с файлом выполнены, его следует закрыть, особенно если происходила запись или добавление данных:

close(f);
— закрыли файл, связанный с файловой переменной f.
При работе с файлами могут возникнуть ошибки, связанные как с отсутствием нужного файла на диске, так и с проблемами чтения или записи (например, мы пытаемся открыть файл для записи на защищенном диске). Поэтому операторы открытия файла и чтения или записи данных следует защищать директивой компилятора {$I-}...{$I+}, а после оператора проверять статус операции ввода-вывода с помощью стандартной функции IoResult:

var f:text; name,s:string;

begin

 writeln ('Введите имя файла:');

 readln (name);

 assign (f,name);

 {$I-}reset(f);{$I+}

 if IoResult<>0 then begin

  writeln ('Не могу открыть файл ',name,
           ' для чтения!');

  writeln ('Нажмите Enter для выхода');

  readln;

  halt;

 end;

 readln(f,s);

 writeln ('Первая строка файла:');

 writeln (s);

 close(f);

 writeln ('Нажмите Enter для выхода');

 readln; end.

В дальнейшем мы для краткости не всегда будем выполнять эти проверки, но хороший стиль программирования предполагает, что в своих программах вы будете их делать.

При чтении из файлов зачастую объем читаемых данных неизвестен заранее. Поэтому необходима функция, умеющая определять, прочитаны ли уже все данные:

function Eof(var F : text) : boolean; 
 -- возвращает true, если при чтении достигнут конец файла.
function Eoln (var F : text) : boolean;
 -- возвращает true, если при чтении достигнут конец строки.
Как правило, основной цикл чтения файла с заранее неизвестным количеством строк выглядит так:

while not eof (f) do begin

 {операторы для чтения строк файла 
  и работы с ними}

end;

При чтении из одного файла "смешанных" строковых и числовых данных следует проверять, действительно ли в нужных строках файла содержатся числа. Для этого можно использовать стандартную процедуру val. Как правило, формат файлов, понимаемых той или иной программой, выбирается программистом, и важно предусмотреть реакцию программы на ошибочный формат исходных данных. Допустим, наша программа представляет собой простейший телефонный справочник, включающий имена абонентов и по одному телефонному номеру на каждое имя. Формат файла, хранящего данные справочника, также выберем простейшим:

Фамилия1

Номер1

Фамилия2

Номер2

и т. д., то есть, в строках 1, 3, 5, ... файла содержатся фамилии абонентов, а в строках 2, 4, 6, ... -- их номера телефонов. Примем также, что файл справочника называется phones.txt, существует и находится в той же папке, откуда запускается программа. Полный листинг программы приводится ниже.
var f:text;

    name,phone,search:string;

    number,strings:longint; 
    error:integer;    
    found:boolean;

begin

 assign (f,'phones.txt');

 reset(f);

 writeln (Фамилия абонента для поиска:');

 readln (search);

 strings:=1; {Счетчик прочитанных строк}

 found:=false; {Переключатель 
                "найдено"-"не найдено"}

 while not eof(f) do begin

  readln (f,name); {Прочитали фамилию}

  readln (f,phone); {Прочитали номер}

  val (phone,number,error); {Пробуем 
   номер-строку преобразовать в число}

  if error<>0 then begin {если это 
          не удалось сделать - ошибка}

   writeln('Ошибка - нет номера телефона!',

    ' Номер строки=', strings);

   writeln ('Нажмите Enter для выхода');

   reset (input); readln; halt;

  end;

  if name=search then begin

   writeln ('Телефон найден:',number);

   found:=true;

   break;

  end; 
  strings:=strings+1;

 end;

 close (f);

 if found=false then 
  writeln ('Телефон не найден!');

 writeln ('Нажмите Enter для выхода');

 reset (input); readln; 
end.

Этой учебной программе недостает как умения редактировать открываемый ей справочник, так и гибкости при поиске -- например, она различает как разные символы  строчные и прописные буквы во вводимой фамилии.
21.2. Примеры работы с файлами

Рассмотрим несколько типовых задач на обработку текстовых файлов.
1. Чтение числовых данных из файла. Массив данных имеет фиксированную размерность, примем ее равной 5.
var f,w:text; 
 a: array [1..5] of integer; i:integer;

begin

   assign (f,'data.txt');

   reset (f);

   for i:=1 to 5 do read (f, a[i]);

   close (f);

   assign (w, 'result.dat');

   rewrite (w);

   writeln (w,'Результаты:');

   for i:=1 to 5 do 
    writeln (w,a[i]:5,sqr(a[i]):5);

   close(w); end.

Файл с именем data.txt может быть, например, таким:

1 2 3

4 5

Файл результатов result.dat будет таким:

Результаты:

    1    1

    2    4

    3    9

    4   16

    5   25

2. Просмотр любого текстового файла на экране. По заполнении экрана до 24 строк программа делает паузу.

var f:text; s:string; count:integer;

begin
repeat
 write ('Имя файла или 0 для выхода: ');

 readln (s);

 if s='0' then halt;

 assign (f,s);

 {$I-}reset (f);{$I+}

 if IoResult<>0 then begin

  writeln ('Не могу открыть файл ',s);

  write ('Нажмите Enter для продолжения');

  readln;

  continue; {повторяем цикл с начала}

 end;

 count:=1;

 while not eof(f) do begin

   readln (f,s);

   writeln (s);

   count:=count+1;

   if count=24 then begin

    count:=1;

    write('Нажмите Enter для продолжения');

    readln;

   end;

 end;

 write ('Нажмите Enter для нового ввода');

 readln;

 close (f);

until false;

end.

Строка s здесь используется как для ввода имени файла, так и для чтения строки файла -- ведь после выполнения связывания оператором assign имя файла нам больше не нужно. Обратите внимание также на оператор continue, в данном случае он позволяет не завершать выполнение программы после неверного ввода пользователя.

3. Работаем со "смешанным" файлом данных, строки которого содержат значения разных типов.
Пусть файл data.txt имеет следующий вид:

Иванов 2

Петров 1

Сидоров 3

Попов 2
...
В каждой строке файла находится фамилия рабочего и через пробел -- номер участка, на котором он работает.
Напишем программу для вывода фамилий всех работников выбранного участка и подсчета их количества. 
var f:text; s,fam:string;
    u,uch,p,kol,i:integer;

begin

 writeln ('Участок?');  read (uch);

 assign (f,'data.txt'); reset (f);

 kol:=0;

 while not eof (f) do begin

  readln (f,s);

  p:=pos(' ',s);

  if p>0 then begin

   fam:=copy (s,1,p-1);

   delete (s,1,p);

   val (s,u,i);

   if i<>0 then begin

    writeln ('Ошибка в числе ',s,
     ' – нет номера участка');

    halt;

   end;

   if u=uch then begin 
    {подсчет рабочих на участке}

    writeln (fam);

    kol:=kol+1;

   end;

  end

  else begin

   writeln ('Ошибка в строке ',s,
            ' – нет пробела');

   halt;

  end;

 end;

 close (f);

 writeln ('kol=',kol);

end.

Значительная часть этой программы посвящена анализу прочитанной из файла строки "смешанных" данных. Программу можно существенно упростить, изменив формат исходных данных -- например, на одной строке файла может быть набрана фамилия, а на другой -- номер участка. Тогда основной цикл чтения файла будет выглядеть так:

while not eof (f) do begin
  readln (f,fam);

  readln (f,u);

  if (u=uch) then begin

   { обработка }

  end;

end;

Приведенный пример иллюстрирует, как часто выбор программистом более или менее удобного формата хранения данных влияет на сложность программы.

21.3. Работа с параметрами командной строки

Программа в DOS или Windows может запускаться с параметрами командной строки, через которые обычно передают имена файлов или указывают режимы работы программы:
turbo.exe vasya.pas

— здесь программе Турбо Паскаль передано имя файла vasya.pas;
my /s /a

— программе с именем my переданы параметры /s и /a.
Существует две стандартных функции для работы с параметрами:

Paramcount -- вернет общее число параметров командной строки;
Paramstr(i) -- вернет строку, содержащую параметр номер i.
В качестве примера реализуем программу просмотра текстового файла на экране, которой параметром передается имя нужного файла. 
var fil:text; name:string;

begin

 if Paramcount<>1 then begin

  writeln ('Запускайте так: ',
           'FILEVIEW имя_файла');

  halt; end;

 assign (fil,Paramstr(1));

 reset (fil);

 while not eof(fil) do begin

   readln (fil,name);

   writeln (name);

 end; 
end.

Второе полезное свойство функции Paramstr -- вызванная из исполняемого файла с аргументом 0, она возвращает полный путь к нему.
Из оболочки Турбо Паскаля параметры командной строки можно передать запускаемой программе через пункт Parameters меню Run.

22. Записи. Бинарные файлы

Как и массивы, записи относятся к составным типам данных. Запись состоит из фиксированного числа элементов, называемых полями. Каждое поле представляет собой объект простого типа данных -- как правило, строку или число. Существенно то, что в одну запись могут входить поля различного типа, в отличие от массива, все элементы которого однотипны. Например, записью можно считать строку экзаменационной ведомости:

Борисова Ю.А.
4
5
5

Данная запись состоит из четырех полей: одно поле -- строкового типа (фамилия и инициалы студента), остальные три поля -- числовые (оценки студента по трем дисциплинам). 
Описанию переменной типа запись предшествует определение типа с помощью оператора type. Для нашей записи это описание могло бы выглядеть следующим образом:

type zap = record

 fam: string;

 m, inf, h: integer;

end;

Здесь zap -- имя нового типа данных, а fam, m, inf и h -- имена полей записи. Служебные слова record ... end в данном случае играют роль операторных скобок, внутри которых записываются поля записи с указанием их типов.

Не очень удобно то, что для каждой дисциплины введено отдельное поле. Введем тип данных student, где эта проблема решена использованием в качестве поля массива balls (баллы студента по трем дисциплинам):

type student = record

 fam: string;

 balls: array [1..3] of integer;

end;

После сделанного таким образом определения в разделе описания переменных можно объявить переменную типа "запись":

var str: student;

или массив таких переменных:

var students: array [1..20] of student;

Для обращения к отдельным полям переменной типа запись используется селектор записи -- символ точки, разделяющий наименования записи и поля:

str.fam
— обращение к полю "фамилия" записи str;
students[1].fam
— обращение к полю "фамилия" первого элемента массива записей students;
students[i].balls[2]
— вторая оценка i-го элемента массива записей students.
С полем записи, как и с элементом массива, разрешены все допустимые в языке операции над переменной соответствующего типа.

Для того чтобы узнать объем в байтах, занимаемый записью в оперативной памяти или на жестком диске, следует воспользоваться стандартной функцией sizeof, передав ей в качестве аргумента имя типа записи:

sizeof (student) -- вернет размер памяти в байтах, занимаемый одной записью типа student.

Чаще всего записи используются при работе с базами данных, хранящимися на жестком диске. Мы уже умеем читать и записывать текстовые файлы, но хранение записей в текстовом файле, как правило, неудобно и требует каждый раз сложного разбора прочитанных строк. Поэтому для эффективной работы с файлами записей следует познакомиться с остальными типами файловых переменных Паскаля.

В общем виде файл данных в Паскале представляет собой совокупность однотипных компонент. В зависимости от способа объявления файловой переменной можно выделить три вида файлов.

· Типизированные файлы. Для них тип компонент указывается непосредственно в описании файловой переменной. Описание в этом случае имеет вид: var файловая_переменная: file of тип_компонент_файла;

Например, если компоненты файла имеют тип записи zap, то разделе описания переменных можно объявить файловую переменную следующим образом:

var f: file of zap;

Естественной "порцией данных" для такого файла будет одна запись типа zap, что очень удобно. Типизированный файл может быть объявлен и как совокупность записей простого типа:

var f2: file of real;

Здесь объявлен файл, содержащий вещественные числа во внутреннем представлении. Размер одной записи такого файла будет равен 6 байт (размер памяти, отводимый под величину типа real).

· Нетипизированные файлы. Задаются стандартным типом file без указания типа компонент, например:

var f: file;

Чтение и запись данных в такой файл может осуществляться произвольными "порциями" данных, что удобно, например, при организации временных и буферных файлов. Рассмотрение нетипизированных файлов выходит за рамки нашего курса.

· Текстовые файлы. Знакомые нам по гл. 21 текстовые файлы задаются стандартным типом text, например:

var f: text;

Компонентами текстового файла являются строки переменной длины. После ввода каждой строки нажимается клавиша Enter. Исторически сложилось так, что при этом в конец каждой строки дописывается два невидимых символа: символ с кодом 13 (CR, возврат каретки) и символ с кодом 10 (LF, перевод строки). Доступ к строкам осуществляется последовательно, начиная с первой. Число строк в текстовом файле может быть произвольным. Последним символом файла может быть специальный маркер EOF (End Of File) с кодом #26.

Существенно подчеркнуть то, что первый и второй вид файлов -- бинарные, а не текстовые:

Текстовый файл содержит только алфавитно-цифровые символы и ряд специальных (такие, как возврат каретки, перевод строки, табуляция). Содержимое файла непосредственно доступно для чтения в любом редакторе текста или в окне консоли.

Бинарный файл содержит произвольные символы, чаще всего он представляет собой набор данных в машинном коде и недоступен для непосредственного чтения.

Например, следующая несложная программа объявляет бинарный файл, состоящий из вещественных чисел и записывает в него числа от 1 до 10 включительно:

var f:file of real; r:real;

begin

 assign (f,'real.dat');

 rewrite (f);

 r:=1;

 while r<11 do begin

  write (f,r); r:=r+1;

 end;

 close (f);

end.

Так как размер величины типа real в памяти равен 6 байтам, полученный файл real.dat будет иметь размер 60 байт и содержать внутренние машинные представления указанных чисел (рис. 22.1).

Рис. 22.1. Бинарный файл

Никакого "текстового" смысла в этой записи нет, чтобы ее интерпретировать, нужно открыть файл в редакторе, поддерживающем 16-ричное "машинное" представление чисел или дамп (рис. 22.2).

Рис. 22.2. Шестнадцатеричный дамп бинарного файла

В этом представлении видно, что каждое число занимает по 6 байт.

Переписав программу так, чтобы числа писались в текстовом представлении, получаем следующее:

var f:text; r:real;

begin

 assign (f,'real.txt');

 rewrite (f);

 r:=1;

 while r<11 do begin

  write (f,r); r:=r+1;

 end;

 close (f);

end.

Файл real.txt состоит из одной строки (так как мы писали только оператором write), в этой строке приведены вещественные числа в экспоненциальной форме (поскольку мы не указывали ширину и точность вывода). Файл изображен на рис. 22.3.

Рис. 22.3. Текстовый файл

Размер файла real.txt равен 170 байт, он текстовый и его можно открыть в "Блокноте" Windows.

Важно усвоить и то, что для бинарных файлов не имеет смысла понятие "строки текста". Поэтому чтения и запись при работе с бинарными файлами осуществляются только процедурами read и write, но не readln и writeln.

Все изученные нами ранее приемы и средства работы с файлами (кроме цикла с функцией eof), вообще говоря, применимы лишь к текстовым файлам, так как существенно зависели от понятия строки данных. Для работы с бинарными файлами необходимы иные средства, позволяющие удобно перемещаться по их записям.

Считается, что в каждый момент времени в файле установлен указатель, определяющий текущую позицию чтения или записи. После чтения или записи компоненты файла указатель сдвигается на его следующую компоненту. Узнать текущую позицию указателя или установить его новое положение позволяют следующие стандартные подпрограммы:

function Filesize(var F) : longint;
 -- возвращает текущий размер файла в компонентах (не байтах! Последнее верно только для file of byte). Параметр F -- файловая переменная. Filesize(F) возвращает число компонентов в F. Если файл пустой, то Filesize(F) возвращает ноль. В режиме {$I-} функция IOResult вернет ноль, если операция была успешна, иначе она вернет отличный от ноля код ошибки. Функция Filesize не может использоваться для текстовых файлов. Файл должен быть открыт;
function FilePos(var F) : longint;
 -- возвращает текущую позицию указателя файла. Параметр F -- файловая переменная. Если указатель текущей позиции файла находится в его начале, то FilePos(F) возвращает ноль. Если указатель текущей позиции файла находится в конце файла, то есть, Eof(F)=true, то значение FilePos(F) равно значению Filesize(F). В режиме    {$I-} функция IOResult вернет ноль, если операция была успешна, иначе она вернет отличный от ноля код ошибки. Функция FilePos не может использоваться для текстовых файлов. Файл должен быть открыт;
procedure seek(var F; N : longint);
 -- перемещает текущий указатель позиции файла на определенный компонент. F -- переменная любого файлового типа за исключением текстового, N -- выражение типа longint. Указатель позиции файла F перемещается на номер компонента N. Номер первого компонента файла равен нулю (соответственно, последнего -- p-1, где p -- общее число компонент в файле). Чтобы расширить файл, вы можете передвинуть указатель на один компонент за последний компонент в файле. То есть, оператор seek(F, Filesize(F)); перемещает текущий указатель позиции файла на конец файла. В режиме {$I-} функция IOResult вернет ноль, если операция была успешна, иначе, она вернет отличный от ноля код ошибки. Процедура  seek не может быть использована для текстовых файлов. Файл должен быть открыт.

Сама процедура чтения или записи бинарного файла состоит из тех же шагов, что для текстового: файл связывается с файловой переменной оператором assign, затем открывается в режиме reset или rewrite, чтение и запись выполняются операторами read и write соответственно, по завершении работы с файлом его следует закрыть оператором close.

Для бинарных файлов запрещен режим открытия append, зато режим reset позволяет как читать записи файла, так и писать в него. Режим rewrite с бинарными файлами используется так же, как с текстовыми -- то есть, только при необходимости переписать файл заново.

Как для текстовых, так и для бинарных файлов могут понадобиться следующие стандартные средства:

procedure Erase(var F);
— стирает внешний файл с диска. Параметр F -- файловая переменная любого файлового типа. Внешний файл, связанный с переменной F удаляется. В режиме {$I-} функция IOResult вернет ноль, если операция была успешна, иначе она вернет отличный от ноля код ошибки. Перед выполнением данной процедуры файл надо закрыть, если он ранее был инициализирован процедурами reset, rewrite или append.

procedure rename(var F; NewName:string);
— переименовывает внешний файл. Параметр F -- переменная любого файлового типа. Внешний файл, связанный с переменной F переименовывается в NewName. Дальнейшие операции над файлом F происходят уже с внешним файлом с новым именем. В режиме {$I-} функция IOResult вернет ноль, если операция была успешна, иначе, она возвращает отличный от ноля код ошибки.

Развернутый пример на работу с файлом записей приведен в Приложении 4 (листинг 4). Его тщательное изучение поможет вам освоить основные приемы работы с бинарными файлами. Программа, приведенная в листинге -- учебная, поэтому "узких мест", вызывающих сомнения с точки зрения профессиональности кода, в ней немало, по возможности они закомментированы с соответствующими пояснениями.
23. Модули. Создание модулей
Применение готовых и разработанных программистом модулей позволяет эффективно решать задачу повторного использования однажды написанного кода.

23.1. Назначение и структура модулей

Модулями называют заранее скомпилированные библиотеки подпрограмм, которые программист может использовать для создания новых программ. При программировании на Паскале модули подключаются из специальных библиотечных файлов, имеющих тип *.tpu (Turbo Pascal Unit). Модуль сам по себе не является выполняемой программой, но его подпрограммы используются другими программами.

Для того чтобы создать модуль, нужно написать файл с расширением *.pas, соблюдающий описанную далее структуру модуля, затем в меню Compile оболочки Паскаля установить переключатель Destination в значение Disk (как мы помним из гл. 6, это же нужно сделать, если вы собираетесь получить из своей программы на Паскале приложение *.exe), затем собрать модуль, выбрав в меню Compile пункт Build. После устранения ошибок компиляции готовый модуль (файл с расширением *.tpu) будет находиться в папке Паскаля.

Подробно рассмотрим общую структуру модуля.
unit ИмяМодуля;
Модуль открывается заголовком, именующим его. По этому имени модуль может быть подключен из программы оператором uses ИмяМодуля;. Имена составляются по обычным для языка правилам.
interface 
Этим ключевым словом открывается интерфейсная часть, в которой объявляются константы, типы данных, переменные, процедуры и функции модуля. Тела общих процедур и функций находятся в разделе реализации (см. далее).

Раздел интерфейса является общим. В нем можно определить то, что будет видимо и доступно для любой другой программы (или модуля), использующей данный модуль. В интерфейсной части может находиться раздел uses, если модуль подключает другие модули. В таком случае ключевое слово uses должно следовать сразу за словом interface.
implementation

В разделе реализации модуля находятся тела процедур и функций, объявленных в интерфейсной части. Раздел реализации является частным. Все объявления, сделанные здесь, могут быть видимы только внутри данного раздела модуля. При этом все константы, типы, переменные, процедуры и функции, объявленные в интерфейсной части видимы и в разделе реализации.

В разделе реализации могут находиться его собственные дополнительные объявления, невидимые любым программам, использующим модуль.

Раздел uses может находиться в части реализации сразу после зарезервированного слова implementation.

Заголовки процедур или функций в разделе реализации должна соответствовать их объявлениям в разделе интерфейса.
Наконец, главная программа модуля, ограниченная операторными скобками begin ... end., обычно пуста. Тем не менее, в ней можно давать начальные значения данным модуля, открывать используемые им файлы, если таковые есть и т. п.

В качестве примера создадим простейший модуль для работы с координатами точек на плоскости и вызовем его подпрограммы при написании новой программы.
unit points;

interface

 type point = array [1..2] of real;

 procedure put (var p:point;x,y:real);

 function distance (p1,p2:point):real;

 function corner (p1:point):integer;

implementation

 procedure put (var p:point;x,y:real);

 begin

  p[1]:=x; p[2]:=y;

 end;

 function distance (p1,p2:point):real;

 begin

  distance:=sqrt(
   sqr(p1[1]-p2[1])+sqr(p1[2]-p2[2]));

 end;

 function corner (p1:point):integer;

 begin

  if p1[1]>0 then begin

   if p1[2]>0 then corner:=1

   else if p1[2]<0 then corner:=4

   else corner:=0;

  end

  else if p1[1]<0 then begin

   if p1[2]>0 then corner:=2

   else if p1[2]<0 then corner:=3

   else corner:=0;

  end

  else corner:=0;

 end;

begin

end.

Наш модуль определяет тип данных Point ("точка") как массив из двух вещественных чисел. Процедура put позволяет задать для точки значения x- и y-координаты, функция distance возвращает расстояние между двумя точками, а функция corner -- номер координатной четверти, в которой находится точка, или 0, если точка лежит на одной из осей координат. Разумеется, реальные модули могут включать сотни функций, если предметная область, которую они моделируют, достаточно сложна. Теперь напишем маленькую тестовую программу, использующую наш модуль:

uses points;

var a,b:Point;

begin

 put (a,1,1);

 put(b,0,0);

 writeln('Расстояние от A до B=',
  distance(a,b):8:3);

 writeln ('Номер четверти для A=',
  corner(a));

end.

Оператор uses, подключающий модуль, указан в первой строке программы. Во время компиляции этой программы в текущем каталоге должен присутствовать файл points.tpu, содержащий созданный ранее модуль points.

При сборке сложной программы Паскаль ищет модули только в тех папках, которые перечислены в поле ввода Unit directories окна Directories верхнего меню Options, поэтому все готовые модули следует либо помещать в одну из этих папок, либо дописать в поле ввода нужные пути к папкам. В Приложении 4 приводится полный листинг модуля для работы с "мышью" из программы на Паскале и тесты для него.

23.2. Стандартные модули Паскаля

В состав Паскаля входит ряд стандартных модулей, перечислим некоторые из них:

· dos -- в модуле dos находятся подпрограммы взаимодействия с операционной системой и обработки файлов;

· strings -- модуль  обеспечивает поддержку символьных строк, завершающихся нулевым байтом;

· graph -- в модуле находится библиотека, состоящая из более, чем 50 графических подпрограмм для рисования различных геометрических фигур;

· crt -- подпрограммы модуля обеспечивают контроль над текстовыми режимами экрана, расширенными кодами клавиатуры, цветами, окнами и звуком;

· printer -- знакомый нам ранее модуль, служащий для программного вывода на принтер.

Чтобы использовать любой из этих модулей, его достаточно подключить оператором uses.

В следующей главе мы изучим основные возможности модуля crt, позволяющие добавлять к своим программам удобный пользовательский интерфейс.
24. Модуль crt и создание консольных интерфейсов

Модуль crt содержит процедуры и функции, предназначенные для работы с экраном консоли в текстовом режиме. Как и ряд других стандартных модулей, crt встроен в компилятор и содержится в файле turbo.tpl.

Экран в текстовом режиме разбивается на отдельные строки, а каждая строка -- на позиции, причем в каждую позицию может быть помещен только 1 символ из набора ASCII.

Для полного описания экранной позиции кроме символа следует задать еще и атрибут, содержащий информацию о цвете символа и фона на экране. Символ и атрибут занимают в памяти по 1 байту. Структура байта-атрибута показана на рис. 24.1.
[image: image53.png]BUTEH
76543210

EROREEEE

M- Mepuianme; @ - ok L| - useT




Рис. 24.1. Структура байта-атрибута консоли

Старший бит 7 управляет мерцанием символа (символ на экране мерцает, если он установлен в 1), биты 4-6 содержат цвет фона (кодируется двоичными числами от 0 до 7 включительно), а биты 0-3 -- цвет  символа  (от 0 до 15). Разумеется, программисту обычно не приходится заполнять байт атрибута по битам, для этого есть стандартные коды цветов. Основные цвета кодируются цифрами от 0 до 15, причем цвет текста может быть любым, а цвет фона -- только из первых 8 цветов. Все цвета описаны в табл. 24.1.
Табл. 24.1. Коды и наименования стандартных цветов

	Код
	Наименование
	Цвет

	0
	BLACK
	черный

	1
	BLUE
	синий

	2
	GREEN
	зеленый

	3
	CYAN
	циановый

	4
	RED
	красный

	5
	MAGENTA
	фиолетовый

	6
	BROWN
	коричневый

	7
	LIGHTGRAY
	светло-серый

	8
	DARKGRAY
	темно-серый

	9
	LIGHTBLUE
	голубой

	10
	LIGHTGREEN
	светло-зеленый

	11
	LIGHTCYAN
	светло-циановый

	12
	LIGHTRED
	светло-красный

	13
	LIGHTMAGENTA
	светло-фиолетовый

	14
	YELLOW
	желтый

	15
	WHITE
	белый


Можно обращаться к цвету как по цифровому коду, так и по англоязычному имени.

Широко используемые текстовые режимы имеют в окне консоли 25 строк по 80 столбцов (позиций) в строке. Нумерация строк и позиций начинается с 1 и считается слева направо и сверху вниз. Весь экран в текстовом режиме может быть описан  парой координат (1, 1), (80, 25). Обратите внимание на порядок записи -- столбец, затем строка.

Ниже рассмотрены основные процедуры и функции модуля. Везде для краткости введены следующие обозначения:
x,x1,x2 -- координаты столбцов экрана;
y,y1,y2 -- координаты строк экрана;
c -- значение цвета.

Особенность модуля crt состоит в том, что он позволяет работать не только со всем экраном, но и с выделенным на нем прямоугольным окном. При этом весь ввод, вывод и прокрутка текста происходят в пределах окна. По умолчанию размеры окна совпадают с размерами экрана, но можно явно установить их обращением к стандартной процедуре Window (x1,y1,x2,y2);, где (x1, y1) и (x2, y2) -- соответственно, левый верхний и правый нижний угол окна.
Цвет фона окна c задает процедура textbackground ( c );, а цвет символов -- textcolor ( c );.
Процедура без параметров clrscr; очищает текущее окно цветом фона.
Для установки текстового курсора в позицию окна с координатами (x, y) определена процедура gotoxy (x,y);.
Программно определить текущее положение курсора позволяют 2 стандартные функции Wherex:char; и Wherey:char;, возвращающие, соответственно, текущие x- и y-координату курсора.

Процедура ClrEol; удаляет все символы от позиции курсора до конца  строки включительно, заполняя этот участок цветом фона.
Процедура Delline; полностью удаляет строку, в которой находится курсор, а Insline; вставляет  пустую строку на экране в месте расположения курсора и заполняет ее цветом фона. Обе процедуры обеспечивают прокрутку содержимого окна.
Процедура Sound (F:word); включает встроенный динамик с частотой F герц, обеспечивая выдачу звукового сигнала. 
Процедура Delay (T:word); задает задержку выполнения программы, равную T миллисекунд (1000 мс = 1 сек.). Эта процедура используется для организации задержек выполнения программы, а также всегда вызывается после sound, чтобы определить время звучания динамика.
Процедура без параметров NoSound; выключает динамик. Обязательно используется после пары Sound и Delay.
Наконец, в модуле crt определены 2 стандартных функции для работы с кодами нажатых клавиш. Функция readkey:char; возвращает код символа, прочитанный из буфера клавиатуры. Функция keyPressed:boolean; возвращает значение true, если была нажата клавиша на клавиатуре (за исключением вспомогательных клавиш Alt, Shift, Ctrl и т. д.). Использование последней функции позволяет организовать циклы, выполняющиеся до нажатия какой-либо клавиши.
При запуске программы из оболочки Паскаля монитор находится обычно в текстовом режиме и устанавливать его не нужно. Тем не менее, существует стандартная процедура textMode (Mode:integer), устанавливающая текстовый режим с номером Mode.

Стандартный цветной текстовый режим 25*80 позиций имеет номер 3, цветной текстовый режим 25*40 позиций -- номер 1.

Модуль crt содержит также системные переменные,  которые можно изменять в соответствии с указанным для них типом.
Переменная CheckBreak:boolean; управляет реакций программы на прерывание  по сочетанию клавиш Ctrl+Break. По умолчанию переменная имеет значение true (реакция включена).

Если переменная DirectVideo:boolean; имеет значение true, процедуры вывода на экран пишут данные непосредственно в видеопамять, не используя операционную систему. Это ускоряет вывод, но может использоваться только на полностью IBM-совместимых ЭВМ.

Переменная textAttr:integer; содержит текущий атрибут текста,  сформированный  по  описанным  выше правилам.

Приведем пример программы, определяющей коды нажатых клавиш. Конструкция repeat ... until в этой программе является образцом обработки ввода с клавиатуры. Проблема состоит в том, что функция readkey возвращает однобайтовый код клавиши, а ряд клавиш и сочетаний клавиш имеют двухбайтовые коды. С этим связан второй вызов функции readkey в программе.

uses crt;

var ch : char; {Символ, который вводим}

begin
 clrscr;   {Очистили экран}

 writeln ('Программа выводит коды клавиш;',

          ' Esc - выход.');

 repeat
  writeln('Нажмите клавишу:');

  ch := readkey;   {Ждем ввода символа}

  if ch = #0 then {Если нажата спец. 
         клавиша, то функция вернула 0,}

  begin
   ch := readkey; {и нужно прочитать код
                   символа дополнительно}

   writeln('Нажата специальная клавиша ',

           'с кодом ', ord(ch));

  end
  else {Иначе если нажата обычная клавиша –
        сразу видим ее код}

   writeln('Нажата клавиша с ASCII-кодом',

           ' ',ord(ch));

 until ch=#27;   {Значение 27 – 
                  это код клавиши Escape}

 writeln ('До свидания.');

end.

Как правило, в реальных программах широко используются небуквенные клавиши, такие как Enter, F1, Esc и т. д. Узнать их коды можно из таблиц ASCII-символов. Например, код клавиши Escape равен #27. Для записи клавиатурного кода на Паскале перед его значением ставится символ #, как сделано в этом примере. Более подробно об обработке нажатий клавиш рассказано в Приложении 5. Листинги 5-8 из Приложения 4 также иллюстрирует основные аспекты обработки нажатий клавиш.

В качестве развернутого примера использования функций модуля crt напишем программу, которая заполняет экран случайными цветными окнами, а также является примером проигрывания несложной "музыки" через встроенный динамик компьютера. Для рисования рамок в этой программе используются символы псевдографики, которые есть только в кодировке DOS (см. Приложение 1).
Program crt_example;
uses crt;   
const minLen=10; {минимальная длина окна}

  pause=500; {задержка при выводе звука}

  blink=128; {установка бита мерцания}

var x1,y1,x2,y2 :integer; 
    {координаты окна}

  background, {цвет фона окна}

  color, {цвет текста}

  freq, {частота звука}

  setblink :integer; {есть/нет мерцание}

procedure doubleFrame (x1,y1,x2,y2:integer;
 Header: string);

{Процедура рисует двойной рамкой окно 
 с заголовком и подготавливает его
 внутреннюю часть для ввода текста}

{ x1,y1,x2,y2 - координаты окна}

{ header  - заголовок окна}

var i,j: integer;

begin

 Window (1,1,80,25);
 {Рисуем верхнюю строку рамки }
 gotoxy (x1,y1); write ('╔'); 
 for i:=x1+1 to x2-1 do write('═');
 write ('╗'); 
 {Перебираем строки внутри окна}

 for i:=y1+1 to y2-1 do begin 
  gotoxy (x1,i); write('║');
  for j:=x1+1 to x2-1 do write (' ');
   {Внутренность окна - пробелы}

  write('║'); {Правая граница}

 end;
 {Аналогично рисуем нижнюю строку}

 gotoxy (x1,y2);  write('╚');   
 for i:=x1+1 to x2-1 do write('═');
 write('╝');
 gotoxy (x1+(x2-x1+1-Length(Header)) 
  div 2,y1);

  {Ставим курсор в середину верхней строки}

 write (Header); {Выводим заголовок}

 Window (x1+1,y1+1,x2-1,y2-1);     
  {Устанавливаем текущее окно внутри рамки}

 gotoxy (1,1);{Ставим курсор в левый 
               верхний угол нового окна}

end;

begin 
 textbackground (BLACK);   
 Window (1,1,80,25);
  {окно вывода - весь экран}

 clrscr; 
 {Инициализируем генератор случайных чисел}

 randomize;
  DirectVideo:=true;

 while not keyPressed do begin 
  {Пока не нажата клавиша,
   выполняется цикл}

  x1:= 1 + random(80-minLen);

  x2:=x1 + minLen + random (80-x1-minLen);

  y1:= 1 + random(25);

  y2:= 1 + y1 + random (25-y1);
  {Выбрали случайные координаты окна }
  background:=random(8);
  color:=random(16);
  {Выбрали цвет фона и текста}

  setblink:=random(2);
  {Выбрали установку мерцания ДА или НЕТ}

  textbackground (background);
  textcolor(color+blink*setblink);
   {Цвет текста с учетом мерцания}

  doubleFrame (x1,y1,x2,y2,' Hello! ');

   {Рисуем окно с помощью процедуры}

  background := (textAttr and 112) shr 4;

   { Из байта цвета, содержащегося в
     переменной textAttr, выделяем цвет

     фона. Операция shr xx сдвигает 
     байт вправо на xx бит, а 112 в 
     двоичной системе это 01110000 
     (включены биты, отвечающие за фон) }

  case background of   
   { В зависимости от цвета фона выбираем
     частоту звучания динамика }

  0: freq:=262;  {Частота ноты До}

  1: freq:=294;  {    -"-  Ре}

  2: freq:=330;  {    -"-  Ми}

  3: freq:=349;  {    -"-  Фа}

  4: freq:=392;  {    -"-  Соль}

  5: freq:=440;  {    -"-  Ля}

  6: freq:=494;  {    -"-  Си}

  7: freq:=524;  {    -"-  До}

  end;

  sound (freq);  {Включаем динамик}

  Delay (pause); 
   {Ждем, пока не истечет задержка}

  Nosound;       {Выключаем динамик!}

 end; {Конец основного цикла}
 {Восстанавливаем атрибуты текста и окно}

 textbackground (BLACK); 
 textcolor (LIGHTGRAY); 
 Window (1,1,80,25); 
 clrscr;
end.
Использование этой программы на современном быстром процессоре может и не дать вам насладиться "космической музыкой" -- проблема в реализации функции Delay, учитывающей не реально прошедшее время в миллисекундах, а "условное" время, связанное с тактовой частотой процессора. Для исправления ситуации следует написать и применять собственную реализацию Delay, привязанную к функции GetTime модуля dos, позволяющей получить "абсолютное" системное время в часах, минутах, секундах и сотых долях секунды. Ниже приводится одна из возможных версий такой функции с комментариями основных действий и тестом:

uses crt,dos;

function getlongintTime:longint; 
{Вернет системное время как longint}

 var Hour,minute,second,sec100: word;

 var k,r:longint;

begin

 GetTime (Hour, minute, second, sec100);

 {Прямое вычисление по формуле
  Hour*360000+minute*6000+second*100+sec100

  не сработает из-за неявного 
  преобразования word в longint:}

 k:=Hour; r:=k*360000;

 k:=minute; Inc (r,k*6000);

 k:=second; Inc(r,k*100);

 Inc(r,sec100); getlongintTime:=r;

end;

procedure MyDelay (ms:word); 
 {Корректно работает с задержками 
  до 65 сек.!}

var endTime,curTime : longint;

    cor:boolean; {признак коррекции времени 
             с учетом перехода через сутки}

begin

 cor:=false;

 endTime:=getlongintTime + ms div 10;

 if endTime>8639994 then cor:=true;

 {Учитываем возможный переход через сутки;

  23*360000+59*6000+59*100+99=8639999 и 
  отняли 5 мс с учетом частоты срабатывания 
  системного таймера BIOS}

 repeat
  curTime:=getlongintTime;

  if cor=true then begin

   if curTime<360000 then 
    Inc (curTime,8639994);

  end;

 until curTime>endTime;

end;

var Hour,minute,second,sec100: word;

begin
 clrscr;

 {setTime (23,59,58,99);} 
  {если раскомментарить – может изменить 
   системное время!}

 repeat
  gotoxy (1,1);
  GetTime (Hour, minute, second, sec100);

  write (Hour:2, ':', minute:2, ':', 
         second:2, ':',sec100:2, ' ');

  MyDelay (500);

until keypressed;

end.

В Приложении 4 приведены также листинги программ для вывода кодов часто используемых клавиш, движения по экрану "прицела" с помощью клавиш со стрелками, а также программа создания несложного двухуровневого меню пользователя (листинги 5-7).

25. Модуль graph и создание графики на Паскале

Для работы с графикой из программы на Паскале в папке, откуда она запускается, должен присутствовать файл egavga.bgi. Он представляет собой графический драйвер, предназначенный для управления видеопамятью в режимах поддержки мониторов типов EGA и VGA. Разумеется, современные мониторы давно "переросли" эти два исторически распространенных класса дисплеев. Однако, на любом современном компьютере поддержка видеорежимов EGA и VGA по-прежнему возможна, если не напрямую, то через специальную программу-эмулятор (см. конец главы).

В поставку Паскаля могут входить и другие файлы с расширением *.bgi, отвечающие за работу с мониторами различных типов.

Кроме того, при компиляции программы, имеющей графический вывод, должен быть доступен модуль graph.tpu, содержащий подпрограммы отрисовки графических объектов.

Библиотека graph.tpu подключается стандартным способом с помощью директивы uses в разделе описаний программ:

uses graph;

В графическом  режиме, который в современных операционных системах типа Windows является основным, экран представляет собой матрицу точек (пикселов), причем имеется возможность высветить любой пиксел любым цветом. Координаты каждого пиксела определяются парой целых чисел:

· координата x -- номер пиксела в строке. Нумерация выполняется слева направо, начиная с 0;

· координата y -- номер строки пикселов. Нумерация строк производится сверху вниз, начиная с 0.

Таким образом,  координаты левого верхнего угла экрана равны (0, 0). 
Любой объект, высвечиваемый на экране, является совокупностью отдельных пикселов. Количество воспроизводимых пикселов по горизонтали и вертикали зависит от типа монитора и установленного графического режима. Каждый монитор может использовать множество режимов, отличающихся количеством поддерживаемых цветов и разрешением графического экрана в пикселах.
Классический Паскаль поддерживает монитор CGA, имеющий разрешение до 320(200 пикселов, монитор EGA с разрешением 640(350, монитор VGA с разрешением до 640(480. Работу с более современными и мощными графическими устройствами, относящимися к классу superVGA, Паскаль непосредственно не поддерживает, хотя существуют созданные независимыми разработчиками графические драйверы этих режимов.

Графический режим работы экрана кроме количества пикселов характеризуется определенной палитрой -- набором видимых цветов. Каждая палитра состоит из 4 цветов для монитора CGA или 16 цветов для EGA и VGA.

Установка графического режима осуществляется путем обращения к процедуре initgraph:

initgraph(var gd:integer, var gm:integer,
 pt:string);

Целочисленные переменные gd и gm задают тип графического драйвера и режим его работы, строковая переменная pt -- путь к файлу *.bgi. Например, при выборе основного для Паскаля видеорежима VGA с разрешением 640(480 пикселов и поддержкой 16 цветов подойдет следующий код:
uses graph;

var  gd,gm,error: integer;

begin

  gd:=VGA;   {адаптер VGA}

  gm:=VGAHi; {режим 640*480пикс.*16 цветов}

  initgraph(gd,gm,'');

  error:=graphresult;

  if error <> grOk then begin
   write ('Ошибка графики: ',
     grapherrormsg(error));

   readln; halt;

  end;

  line (0,0,getmaxx,getmaxy); 
  readln; closegraph;

end.

Так как путь к файлу egavga.bgi указан пустым, предполагается, что он находится в текущей папке. После перехода в графический режим процедурой line рисуется линия из левого верхнего в правый нижний угол экрана, затем, после нажатия Enter, графический режим закрывается и происходит выход из программы.

Для автоматического выбора максимально возможного режима переменной gd необходимо присвоить значение detect, при этом переменные gm и pt не определяются, если в текущем каталоге, в котором находится система Турбо Паскаль, имеются файлы *.bgi. Пример:

uses graph; var gd,gm: integer;

begin
 gd:=detect; initgraph(gd,gm,''); ...

Рассмотрим основные стандартные процедуры и функции модуля graph.

closegraph;
— процедура без параметров, завершает работу в графическом режиме. Следует выполнять эту процедуру перед завершением любой графической программы на Паскале.

cleardevice;
— процедура без параметров, очищает экран. При переходе в графический режим экран очищается автоматически, так что перед началом вывода эта операция не требуется.
function getmaxx:integer;
— функция возвращает максимальную координату пиксела по оси x.
function getmaxy:integer;
— функция возвращает максимальную координату пиксела по оси y.
setcolor(color:word); 
— процедура устанавливает цвет рисования линий, точек и текста (аналог "пера" в программах для рисования). Цвета кодируются так же, как в текстовом режиме (см. табл. 24.1).
setfillstyle (style:word, color:word);
— процедура устанавливает цвет заполнения областей экрана (параметр color) и способ наложения цвета (параметр style). Является аналогом "кисти" в программах для рисования. Параметр color принимает значения, указанные в табл. 24.1, параметр style -- значения от 1 до 11. При style=1 происходит сплошное заполнение цветом, другие стили позволяют создать различные штриховки. Здесь и далее вместо цифр можно использовать символические имена стилей, узнать о них можно в справочной системе.
Приведем примеры.
setfillstyle (linefill,GREEN); 
 {установили заполнение зелеными линиями}

setfillstyle (solidfill,RED); 
 { установили сплошную заливку красным}

Следующая процедура определяет стиль рисования линий:
setlinestyle (linestyle:word, pattern:word,
 thickness:word); 
Параметр linestyle (стиль линии) принимает значения от 0 до 4, значение 0 соответствует сплошной линии, параметр pattern при использовании готовых стилей со значением linestyle от 0 до 3 игнорируется, толщина линии thickness указывается значением 1 или 3 (в пикселах). Например, оператор setlinestyle (0,0,1); устанавливает стиль сплошной тонкой линии, а setlinestyle (1,0,3); -- толстую пунктирную линию. Для цифровых значений linestyle и thickness в библиотеке также определены символические имена, при значении linestyle=4 можно определить собственный стиль, задав его параметром pattern с помощью битовой маски.
Перейдем к стандартным подпрограммам, связанным с отображением на экране основных графических примитивов.
putpixel(x,y:integer,color:word);
— процедура высвечивает на экране пиксел с координатами (x, y) цветом color;
function getpixel (x,y:integer):word;
— функция вернет код цвета пиксела с координатами (x, y).
line(x1,y1,x2,y2:integer);
— процедура рисует текущим цветом прямую линию с экранными координатами начала (x1, y1), и конца (x2, y2).

moveto(x,y:integer);
— процедура устанавливает текущую позицию рисования (пера, графического курсора) в точку с экранными координатами (x, y).
lineto(x,y:integer); 
— процедура проводит прямую линию из текущей позиции пера в точку с экранными координатами (x, y). Линия проводится текущим цветом пера.

linerel(dx,dy:integer);
— процедура проводит прямую линию из текущей позиции в точку с приращением координат от текущих на dx и dy, приращения могут быть как положительными так и отрицательными. Таким образом, процедура linerel позволяет указывать, в отличие от line и lineto, не абсолютные, а относительные координаты точки, куда нужно провести линию.
rectangle(x1,y1,x2,y2:integer);
— процедура рисует прямоугольник с координатами левого верхнего угла (x1, y1) и правого нижнего угла (x2, y2). Цвет прямоугольника, как и других незакрашенных фигур, определяется установкой, сделанной процедурой setcolor.
bar(x1,y1,x2,y2);
— процедура рисует закрашенный прямоугольник с координатами углов (x1, y1) и (x2, y2). Цвет и стиль заливки определяются процедурой setfillstyle.

bar3d (x1, y1, x2, y2, depth :integer;
  top:boolean);
— процедура рисует трехмерный параллелепипед. Параметр depth определяет глубину фигуры по оси x, top указывает, рисовать ли верхнюю грань:
bar3d (50,50,100,100,20,true);

Следующая процедура рисует многоугольник или ломаную линию:

drawpoly (numpoint:integer; 
 var polypoints);
Аналогичная процедура fillpoly создает закрашенный цветом заливки многоугольник. Покажем работу процедуры на примере:

var poly: array [1..10] of integer;

poly[1]:=20; poly[2]:=20;
poly[3]:=60; poly[4]:=30; 
poly[5]:=60; poly[6]:=60;
poly[7]:=40; poly[8]:=80; 
poly[9]:=20; poly[10]:=20;
drawpoly (5,poly);

Элементы с нечетными номерами массива poly задают x-координаты точек, а с четными -- y-координаты. Таким образом, в данном случае нарисован пятиугольник.
floodfill (x,y,bordercolor:integer);
— мощная процедура, позволяющая закрасить любую замкнутую область, которой принадлежит точка (x, y) и которая ограничена по краям цветом bordercolor.

circle(x,y:integer,r:word);
— несложная процедура рисует окружность с центром в точке с координатами (x, y) и радиусом r.

arc(x,y:integer,sa,ea,r:word);
— процедура рисует дугу окружности с центром в точке с координатами (x, y), радиусом r, начальным углом sa и конечным углом ea. Углы sa и ea измеряются в градусах и отсчитываются против часовой стрелки от оси абсцисс.

Существуют также процедуры для рисования эллипсов и секторов.
Для вывода текста на графический экран имеются 2 основные функции.
outtextxy(x,y:integer,text:string);
— процедура выводит текст на экран, начиная с точки с координатами (x, y). Здесь text -- константа или переменная строкового типа, содержащая нужное сообщение. Текст выводится установленным цветом рисования линий. Заметим, что применение стандартных процедур write и writeln для вывода текста в графическом режиме нежелательно, так как они не позиционируют текст по пикселам и не учитывают установок цвета и фона графического экрана.
outtext(text:string);
— процедура выводит текст, заданный параметром, на экран, начиная с текущей позиции графического курсора.
Для краткости мы не рассматриваем методы привязки текста к позициям на экране. 
В библиотеке graph нет процедур для вывода численных данных. Для этого необходимо сначала преобразовать число в строку с помощью процедуры str, а затем посредством операции '+' подключить строку к сообщению, выводимому процедурой outtextxy. Например:

max:=34.56; {Число}

str(max:6:2,smax); 
 {Преобразование числа max в строку smax}

outtextxy(400,40,'Максимум=' + smax); 
 {Вывод строки smax с комментарием}

Узнать ширину и высоту строки в пикселах можно с помощью стандартных функций function textwidth (s: string):word; и function textheight (s: string):word; соответственно.
Существуют также процедуры для управления внешними графическими шрифтами, хранящимися в файлах *.chr.

Приведем примеры программ, реализующих типовые графические построения.
1. Реализация процедуры, выводящей строку текста в центр прямоугольного окна на экране.
procedure centerstring 
 (x1,y1,x2,y2,color :integer; str: string);

var cx,cy:integer;

begin

 setcolor (color); {Устанавливаем цвет}

 rectangle (x1,y1,x2,y2);
{Рамка}

 rectangle (x1+2,y1+2,x2-2,y2-2);
 cx:=(x1+ x2) div 2;
{Координаты}

 cy:=(y1+ y2) div 2;
{центра}

 settextJustify (centertext,centertext);
  {Выравнивание текста по центру}

 outtextxy (cx,cy,str); {Вывод строки}

end;
... 
{ Обращение к данной процедуре: }

centerstring (100, 100, 200, 200, yELLOW,
  'Hello!');

2. В следующем примере мы нарисуем на экране как "линейный" объект (домик с переменным числом окон и этажей), так и "радиальный" (солнце с лучами), для которого нужен пересчет из декартовых координат в полярные. Схема, поясняющая принцип перевода из декартовых координат в полярные, приведена на рис. 25.1.

Рис. 25.1. Пересчет из декартовых координат в полярные
program SunHouse;

uses graph,crt;

var Driver, Mode: integer;

    i,j,u,N,K,x2,y2:integer;

    rad:real; sunx,suny:integer;

begin 
 {Не проверяем правильность ввода}

 writeln ('Сколько этажей?'); read (N);

 writeln ('Сколько окон на этаж?'); 
 read (K);

 Driver := VGA; Mode:=VGAHi; 
 initgraph(Driver, Mode,'');

 {Домик}

 setcolor (15);

 rectangle (20, getmaxy-20-70*n,
  20+k*50+(k+1)*20, getmaxy-20);

 {Крыша}

 moveto (20,getmaxy-20-70*n);

 lineto(10,getmaxy-20-70*n);

 lineto (20,getmaxy-40-70*n);

 lineto (20+k*50+(k+1)*20,getmaxy-40-70*n);

 lineto (30+k*50+(k+1)*20,getmaxy-20-70*n);

 lineto (20+k*50+(k+1)*20,getmaxy-20-70*n);

 {Линии между этажами}

 for i:=1 To N Do

  line (20, getmaxy-20-70*i,
    20+k*50+(k+1)*20, getmaxy-20-70*i);

 setfillstyle (solidfill, YELLOW);

 {Окна на каждом этаже}

 for i:=1 To N Do {Цикл по этажам}

 for j:=1 To K Do begin {Цикл по окнам}

 bar(20+(j-1)*70+20,getmaxy-20-(i-1)*70-
60,20+(j-1)*70+70, getmaxy-20-(i-1)*70-10);

 end;

sunx:=getmaxx-50; suny:=50;

 {Центр солнца – координаты на экране}

 FillEllipse (sunx, suny, 30, 30); 
 {Рисуем контур солнца}

setcolor (YELLOW);

{Рисуем лучи}

 u:=0;

 while u<=360 Do begin 
  {угол u меняем от 0 до 360 градусов}

  rad:=u*pi/180; 
  {перевод в радианы для функций sin,cos }

  x2:=round(sunx+50*cos(rad));  
  y2:=round(suny+50*sin(rad)); 
{перевод из полярных координат в декартовы}

  line (sunx,suny,x2,y2);

  u:=u+12; {шаг по углу=12 градусов}

 end;

 repeat until keypressed;

 closegraph;

end.

3. Этот пример реализует программу построения графика функции f(x), заданной отдельной подпрограммой, в границах [a, b] изменения аргумента x. 
Схема пересчета значений (x, f(x)) при 

в экранные координаты приведена на рис. 25.2. Пересчет выполняется в 2 этапа.
Узнав с помощью процедур getmaxx, getmaxy размеры графического экрана и определив значение xstep -- шаг по x, соответствующий одному пикселу на экране, мы сможем обеспечить масштабирование графика по оси X. Для масштабирования по оси Y на первом этапе пересчета требуется также определить максимальное и минимальное значения f(x) на интервале [a, b] при изменении x с шагом xstep.

Второй этап связан с непосредственным пересчетом значений (x, f(x)) в экранные координаты (cx, cy). Для решения этой задачи воспользуемся формулой, согласно которой значение x, принадлежащее интервалу [a, b], можно линейно преобразовать в значение y, принадлежащее интервалу [c, d]: 

. Эта формула позволит получить коэффициенты преобразования величин (x, f(x)) к экранным координатам. Дополнительно придется учесть то, что экранная ось Y проведена сверху вниз.

Рис. 25.2. Пересчет из декартовых координат в экранные

program graphOfFun;

uses graph,crt;

function f(x:real):real; 
 { Функция, график которой строим }

begin
 f:=sin(x)+cos(x);

end;

function getreal(s:string):real;

var f:real; {Ввод числа с контролем ошибок}

begin

 repeat

  write (s);

  {$I-}readln(f);{$I+}

  if IoResult=0 then break

  else writeln 
   ('Ошибка! Введено не число');

 until false;

 getreal:=f;

end;

procedure Init; 
{Инициализация графического режима }

{VGA 640*480 пикселов, 16 цветов}

var driver,mode,error:integer;

begin

 driver:=VGA; mode:=VGAHi;

 initgraph(driver,mode,'');

 error:=graphresult;

 if error<>0 then begin  
  {Не ноль означает ошибку!}

  writeln;

  write ('Не могу инициализировать ',

  'графику! Ошибка ',grapherrormsg(error));

  halt(1)

 end;

end;

var a,b: real;  { Границы изменения x }

xmax,ymax: integer; { Размеры графического
                экрана по длине и высоте }

xstep:real; { Шаг по x }

x,fx:real;  
fmax,fmin:real;

cx,cy:integer;  { Экранные координаты }

oldx,oldy:integer;{В этих переменных будем
запоминать координаты последней точки,
чтобы соединить ее с текущей }

xcoef,ycoef:real; {Коэффициенты пересчета к
 экранным координатам }

ss:string; 
begin
 clrscr;

 repeat

  a:=getreal ('Левая  граница по x=');

  b:=getreal ('Правая граница по x=');

  if a>b then write('Ошибка!Левая граница',

   ' должна быть меньше правой');

 until a<b;

 Init;     { Инициализировать графику    }

 xmax:=getmaxx; ymax:=getmaxy; 
  { размеры графического экрана  }

 xstep:=(b-a)/(xmax-19);   
  { шаг по x, соответствующий 1 пикселу.}

 x:=a; fmax:=f(a); fmin:=fmax;

 while x<=b do begin 
  fx:=f(x);

  if fx>fmax then fmax:=fx
  else if fx<fmin then fmin:=fx;
  x:=x+xstep;

 end; 
 xcoef:=(xmax-19)/(b-a);

 ycoef:=(ymax-19)/(fmax-fmin);

 { обрамление графика: }

 setfillstyle (solidfill,CYAN);

 bar (10,10,xmax-10,ymax-10);

 setcolor (YELLOW);

 rectangle (9,9,xmax-9,ymax-9);

 str (a:8:3,ss); outtextxy (2,ymax-8, ss);
 str (b:8:3,ss); outtextxy 
   (xmax-66,ymax-8, ss); {Границы по x }

 settextstyle (DefaultFont,VertDir,1); 
  { Границы по y выводим вертикально }

 str(fmax:8:3,ss); outtextxy(9,2, ss);

 str(fmin:8:3,ss);outtextxy(9,ymax-66, ss);
 setcolor (White);{цвет рисования графика}

 x:=a;

 while x<=b do begin

  fx:=f(x);

  cx:=10+round((x-a)*xcoef);
  cy:=ymax-10-round((fx-fmin)*ycoef);
  putpixel (cx,cy,LightRED);
  if x>a then line (oldx,oldy,cx,cy); 
   { Соединяем две последние точки }

  oldx:=cx; oldy:=cy;    
   { Запоминаем текущую точку }

  x:=x+xstep;

 end;
 repeat until keyPressed;  
 closegraph;
end.

Недостаток этой программы -- отсутствие пропорционального масштабирования по осям x и y. Подумайте, как ее можно улучшить. Листинг 12 из Приложения 4 представляет более объемную графическую программу, реализующую несложную компьютерную игру. Функция Draw этого листинга может также служить примером обработки 16-цветного изображения в формате BMP из программы на Паскале.

Решение задач, связанных с выводом графики на экран, часто требует сохранения участка экранного изображения в оперативной памяти с последующим восстановлением прежней картинки. Для решения этой проблемы в библиотеке graph предусмотрен набор соответствующих функций.

В первую очередь требуется оценить объем памяти, требуемый для сохранения участка экрана. Стандартная функция библиотеки graph, имеющая вид imagesize (x1,y1,x2,y2:integer):word, где x1, y1 -- экранные координаты верхнего левого, а x2, y2 -- правого нижнего угла, возвращает число байт памяти, необходимых для сохранения заданной прямоугольной области. Эта функция может определить объем памяти до 64 Кб включительно, так как тип возвращаемого значения -- беззнаковое целое типа word. Если количество требуемой памяти больше либо равно 64 Кб, возвращается значение ошибки -11 (grError). Разумеется, вызов функции предполагает, что монитор находится в графическом режиме.

После того, как требуемое количество байт определено, программа должна позаботиться о выделении участка оперативной памяти, предназначенного для сохранения изображения. Это легко сделать, используя системную процедуру getmem (var p:pointer; size:word), где объем памяти size ранее определен функцией imagesize, а переменная p представляет собой указатель. Ранее незнакомый нам тип данных "указатель" служит для косвенного вызова одних переменных через другие. Фактически, переменная-указатель хранит адрес другой типизированной переменной и может обратиться к ней, используя синтаксис p^, где p -- имя указателя. Применение указателей позволяет создавать динамические переменные, способные в разные моменты времени адресовать различные участки оперативной памяти, в которых хранятся данные. Самый большой блок памяти, который может выделить getmem, также равен 64 Кб. Освободить ранее занятую память можно процедурой Freemem (var p:pointer; size:word).
Наконец, третий шаг -- сохранить участок экрана, используя только что сформированный в оперативной памяти буфер. Для этой цели достаточно использовать процедуру Getimage (x1,y1,x2,y2:integer; var p). Здесь параметры x1,y1,x2,y2 имеют тот же смысл, что для функции imagesize, а нетипизированный параметр-указатель p получен процедурой getmem.
Теперь требуемый участок экрана сохранен в памяти и может быть занят новым изображением. После того, как изображение выполнило свои функции и нужно восстановить прежний фрагмент экрана, достаточно вызвать процедуру putimage (x,y:integer; var p; mode:word), где x,y -- экранные координаты левого верхнего угла восстанавливаемой области, p -- указатель на сохраненную память, а переменная mode определяет, какая двоичная операция будет использована при выводе изображения на экран. Для неизмененного восстановления изображения следует передавать в качестве mode значение NormalPut, другие возможные значения параметра -- copyPut, XORPut, ORPut, ANDput и NOTPut. Все описанные функции использованы в листинге, приведенном ниже. Его изучение поможет вам в написании аналогичных программ, поддерживающих движение по экрану графических объектов.

uses graph,crt;

var Gd, Gm : integer;

    P : pointer;

    size : word;

    x,y,width,height: integer;

    ch:char;

    changed:boolean;

begin
 {Инициализация графики}

 Gd:=VGA; Gm:=VGAHi;

 initgraph(Gd, Gm, '');

 if graphresult <> grOk then halt(1);

 {Отрисовка фона}

 setfillstyle(xHatchFill, CYAN);

 bar(0, 0, getmaxx, getmaxy);

 {Параметры активного окна}

 x:=getmaxx div 2;

 y:=getmaxy div 2;

 width:=40;

 height:=30;

 {Выделение памяти для сохранения 
  фона под окном}

 size:=imagesize(x,y,x+width-1,y+height-1);

 getmem(P, size);

 getimage(x, y, x+width-1, y+height-1, P^);

 {Первая отрисовка активного окна}

 setfillstyle(solidfill, RED);

 bar (x,y,x+width-1,y+height-1);

 {Признак изменения положения окна}

 changed:=false;

 repeat {Цикл движения объекта}

  ch:= readkey; {Читаем код клавиши}

  if ch=#0 then begin 
   {Если это расширенный код...}

   ch:= readkey; {то читаем второй байт}

   case ch of 
    {Реагируем только на 4 клавиши:}

    #72: if y>0 then changed:=true;
          {стрелка вверх}

    #80: if y+height<getmaxy then
       changed:=true; {стрелка вниз}

    #75: if x>0 then changed:=true;
          {стрелка влево}

    #77: if y+width<getmaxx then
       changed:=true; {стрелка вправо}

   end;

   if changed=true then begin 
    {если флаг реакции выставлен}

    PutImage(x, y, P^, NormalPut); 
    {восстанавливаем экран под окном}

    case ch of 
     {и меняем нужную координату окна}

     #72: dec(y);

     #80: inc(y);

     #75: dec(x);

     #77: inc(x);

    end;

    getimage(x,y,x+width-1,y+height-1,P^);

{сохраняем экран под новым положением окна}

    bar (x,y,x+width-1,y+height-1); 
{и перерисовываем окно}

    changed:=false; 
    {сбросить флаг изменения}

   end;

  end;

 until ch=#27; {...пока не нажата Esc}

 Freemem (p,size); {освобождаем память}

 closegraph; {и закрываем графику}

end.

Говоря о написании графических программ применительно к Паскалю, нельзя обойти стороной вопрос о поддержке DOS-графики современными компьютерами. К сожалению, многие современные платформы не поддерживают графические видеорежимы DOS. Помочь может эмулятор DOS-машины, такой как свободно распространяемая программа DOSBox. Скачав DOSBox по адресу http://dosbox.sourceforge.net и установив ее, мы можем запускать приложения DOS в любом видеорежиме с поддержкой (эмуляцией) многочисленных устаревших программных и аппаратных решений.

Желательно также установить оболочку эмулятора, позволяющую создавать для DOS-приложений настраиваемые ярлыки. Оболочка DOSShell доступна по адресу http://www.loonies.narod.ru/dosshell.htm, а узнать об эмуляторах DOS больше Вы можете по адресам http://ru.wikipedia.org/wiki/DOSBox и http://gh.gameslife.ru/text/dosbox.htm.
Заключение
Изучение алгоритмических языков высокого уровня было и остается важным элементом в подготовке специалиста, чья профессиональная деятельность связана с информационными технологиями.

Теоретический материал и многочисленные примеры программ, приведенные в этом пособии, позволяют автору рассчитывать, что оно окажется полезным для всех, кто приступает к изучению увлекательной науки программирования.
Приложение 1. Таблицы ASCII-кодов символов для операционных систем DOS и Windows
Чтобы понять, как хранится информация в ЭВМ, нам придется вспомнить ряд терминов. 
Минимальная единица измерения информации -- один бит. Бит -- это двоичный разряд со значением "0" или "1". Очевидно, почему разработчики первых ЭВМ остановились на двоичной системе счисления. Числа в этой системе легче всего представить физически -- допустим, нулю соответствует состояние "не намагничено" участка магнитной ленты, а единице -- "намагничено", или нулю -- состояние "нет сигнала", а единице -- "есть сигнал" в некоторой линии связи.

Вся информация в компьютере хранится в числовой форме и двоичной системе счисления. Поскольку с помощью одного бита можно представить всего 2 различных значения, минимальной передаваемой или адресуемой единицей информации является байт, представляющий собой совокупность 8 бит. Более крупными единицами измерения данных являются килобайт (Кб) =1024 (210) байта, мегабайт (Мб) =1024 килобайта и гигабайт (Гб) =1024 мегабайта. Для ориентировки можно сказать, что если на странице текста помещается в среднем 2500 знаков, то 1 Мб -- это примерно 400 страниц, а 1 Гб -- 400 тысяч страниц.

Легко понять, сколько различных значений может быть представлено с помощью N бит -- это число равно 2N. Таким образом, в один байт "уместится" 28 = 256 различных значений.
Для обработки на компьютере вся нечисловая информация должна быть преобразована в числовую форму. Так, для компьютерной обработки текста каждая буква при вводе кодируется определенным числом, а при выводе на внешние устройства, такие как монитор или принтер, по кодам символов строятся соответствующие изображения букв. Соответствие между набором символом и кодирующими их числами называется кодировкой символов. Как правило, код символа хранится в одном байте, поэтому коды символов могут принимать значения от 0 до 255. Такие кодировки называются однобайтовыми. Основной символьный набор компьютера -- это стандартная для IBM-совместимых машин однобайтовая кодировка ANSI, называемая также ASCII-кодом (читается "аски-код").

В двухбайтовой кодировке Unicode (Юникод), предлагаемой в настоящее время в качестве общемирового стандарта, символ кодируется двумя байтами, таким образом, коды символов могут принимать значения от 0 до 65535=216 различных символов. В этой кодировке имеются коды для всех букв алфавитов множества языков, математических, декоративных символов и т. д.

На рис. П1 представлены две основные русскоязычные кодировки, известные как DOS-866 и Windows-1251. С первой работает среда Турбо-Паскаль и все программы русифицированных версий DOS, со второй -- все приложения русифицированных версий Windows. Чтобы узнать код символа, достаточно к числу десятков из первого столбца приписать число единиц из первой строки. Так, код буквы "Z" в обеих кодировках равен 90. Символы с кодами меньше 32 -- непечатаемые, это такие символы, как перевод строки, возврат каретки, табуляция, поэтому они не вошли в таблицу. Код пробела равен 32. Обратите внимание, что первые половины кодовых таблиц (символы с кодами меньше 128) совпадают как в этих двух кодировках, так и во всех остальных.

Рис. П1. Кодировки Dos и Windows
Приложение 2. Основные директивы компилятора Паскаля

{$A+} -- включить/выключить выравнивание по словам.
{$B+} -- включить/выключить полное вычисление булевых выражений.
{$С MOVEABLE DEMANDLOAD DISCARDABLE} --управление сегментом кода (только режимы Windows и Protected):
· MOVEABLE -- система может изменить положение сегмента кода в памяти;
· FIXED -- система не может изменить положение сегмента кода в памяти;
· PRELOAD -- сегмент кода загружается с началом исполнения программы;
· DEMANDLOAD -- сегмент кода загружается только при обращении;
· PERMANENT -- сегмент кода остается в памяти после загрузки;
· DISCARDABLE -- сегмент кода может быть выгружен после обращения.
{$D+} -- включить/выключить отладочную информацию.
{$E+} -- включить/выключить эмуляцию сопроцессора вещественных чисел.
{$F+} -- включить/выключить FAR-вызовы по умолчанию.
{$G Имя_модуля1, Имя_модуля2, ...}  -- включить в проект указанные модули Unit (только режимы Windows и Protected).
{$G+} -- включить/выключить генерацию кода процессора 80286.
{$I Имя_файла} -- включить исходный текст файла *.pas в программу.
{$I+} -- включить/выключить контроль операций ввода-вывода.
{$K+} -- включить/выключить оптимизацию вызовов подпрограмм (только Windows).
{$L Имя_файла} -- включить файл *.obj в программу на этапе сборки.
{$L+} -- включить/выключить генерацию MAP-файла.
{$M Стек, Хип-минимум, Хип-максимум} -- указать размеры стека (1024-65520) и хипа (0-655360) для программы в байтах.
{$N+} -- включить/выключить поддержку сопроцессора 80x87.
{$O+} -- включить/выключить поддержку оверлеев.
{$O Имя_модуля} -- подключить оверлейный модуль (unit).
{$P+} -- если директива включена, строки "открыты" ("закрыть" для совместимости со старыми версиями).
{$Q+} -- включить/выключить контроль переполнения для арифметических операций.
{$R+} -- включить/выключить контроль переполнения для порядковых величин.
{$R Имя_файла} -- подключить файл ресурсов *.res (только Windows и Protected).
{$S Размер} -- указать размер сегмента кода (только Windows и Protected).
{$S+} -- включить/выключить проверку переполнения стека.
{$T+} --включить/выключить контроль типов указателей.
{$V+} -- включить/выключить строгий контроль длины строк.
{$W+} -- если режим включен, генерируются начальный и завершающий код для far-функций и процедур.
{$X+} -- включить/выключить расширенный синтаксис.
{$Y+} -- включить/выключить генерацию таблицы перекрестных ссылок.
Приложение 3. Основные сообщения об ошибках Паскаля

Сообщения компилятора о синтаксических ошибках:

1 -- выход за границы памяти;

2 -- не указан идентификатор;

3 -- неизвестный идентификатор;

4 -- повторный идентификатор;

5 -- синтаксическая ошибка;

6 -- ошибка в вещественной константе;

7 -- ошибка в целочисленной константе;

8 -- строковая константа превышает размеры строки;

9 -- слишком много вложенных файлов;

10 -- неправильный конец файла;

11 -- строка слишком длинная;

12 -- требуется идентификатор типа;

13 -- слишком много открытых файлов;

14 -- неверное имя файла;

15 -- файл не найден;

16 -- диск заполнен;

17 -- неправильная директива компилятора;

18 -- слишком много файлов;

19 -- неопределенный тип в определении ссылки;

20 -- нужен идентификатор переменной;

21 -- ошибка в определении типа;

22 -- слишком большая структура;

23 -- базовый тип множества нарушает границы;

24 – компонентами файла не могут быть файлы или объекты;

25 -- неверная длина строки;

26 -- несоответствие типов;

27 -- неправильный базовый тип отрезка типа;

28 -- нижняя граница больше верхней;

29 -- нужен порядковый тип;

30 -- нужна целая константа;

31 -- нужна константа;

32 -- нужна целая или действительная константа;

33 -- нужен идентификатор типа;

34 -- неправильный тип результата функции;

35 -- нужен идентификатор метки;

36 -- нужен begin;

37 -- нужен end;

38 -- нужно выражение типа integer;

39 -- нужно выражение перечисляемого типа;

40 -- нужно выражение типа boolean;

41 -- типы операндов не соответствуют оператору;

42 -- ошибка в выражении;

43 -- неверное присваивание;

44 -- нужен идентификатор поля;

45 -- объектный файл слишком большой (больше 64 Кб);

46 -- неопределенная внешняя процедура;

47 -- неправильная запись объектного файла;

48 -- сегмент кода слишком большой (больше 65520 байт);

49 -- сегмент данный слишком велик;

50 -- нужен оператор do;

51 -- неверное определение public;

52 -- неправильное определение extrn;

53 -- слишком много определений типа extrn (больше 256);

54 -- требуется of;

55 -- требуется интерфейсная секция;

56 -- недействительная перемещаемая ссылка;

57 -- требуется then;

58 -- требуется to или downto;

59 -- неопределенное опережающее описание;

60 -- слишком много процедур (больше 512 в одном модуле);

61 -- неверное преобразование типа;

62 -- деление на нуль;

63 -- неверный файловый тип;

64 – невозможно прочитать или записать переменные данного типа;

65 – требуется использование переменной-указателя;

66 -- нужна строковая переменная;

67 -- нужно выражение строкового типа;

68 -- программный модуль не найден;

69 -- несоответствие времен программных модулей;

70 -- несоответствие версий программных модулей;

71 -- повторное имя программного модуля;

72 -- ошибка формата файла программного модуля;

73 -- требуется секция реализации;

74 -- типы константы и тип выражения оператора case не соответствуют друг другу;

75 -- нужна переменная типа запись;

76 -- константа нарушает границы;

77 -- нужна файловая переменная;

78 -- нужно выражение типа указатель;

79 -- нужно выражение типа real или integer;

80 -- метка не находится внутри текущего блока;

81 -- метка уже определена;

82 -- неопределенная метка в предыдущем разделе операторов;

83 -- недействительный аргумент оператора @;

84 -- требуется ключевое слово unit;

85 -- требуется указать ";";

86 -- требуется указать ":";

87 -- требуется указать ",";

88 -- требуется указать "(";

89 -- требуется указать ")";

90 -- требуется указать "=";

91 -- требуется указать ":=";

92 -- требуется "[" или "(.";

93 -- требуется "]" или ".)";

94 -- требуется ".";

95 -- требуется "..";

96 -- слишком много переменных;

97 -- неправильная переменная цикла оператора for. Переменная должна быть перечислимого типа;
98 -- нужна переменная целого типа;

99 -- здесь не допускаются файлы;

100 -- несоответствие длины строковой переменной или константы;

101 -- неверный порядок полей;

102 -- нужна константа строкового типа;

103 -- нужна переменная типа integer или real;

104 -- нужна переменная перечисляемого типа;

105 -- ошибка в операторе inline;

106 -- предшествующее выражение должно иметь символьный тип;

107 -- слишком много перемещаемых элементов;

108 -- недостаточно памяти для выполнения программы;

109 -- нет возможности найти файл .EXE;

110 -- модуль выполнять нельзя;

111 -- компиляция прервана с помощью клавиш Ctrl+Break;

112 -- константа оператора case находится вне границ;

113 -- ошибка в операторе. Данный символ не может быть первым символом в операторе;

114 -- невозможно вызвать процедуру прерывания;

115 -- для компиляции необходимо наличие сопроцессора 8087;

116 -- для компиляции необходим режим 8087;

117 -- адрес назначения не найден;

118 -- в такой ситуации включаемые файлы не допускаются;

119 -- ошибка формата файла .TPU;

120 -- требуется NIL;

121 -- неверный квалификатор переменной;

122 -- недействительная ссылка на переменную;

123 -- слишком много символов (больше 64 Кб);

124 -- слишком большой раздел операторов (больше 64 Кб);

125 -- в модуле нет отладочной информации;

126 -- параметры файлового типа должны быть параметрами var;

127 -- слишком много условных символов;

128 -- пропущена условная директива;

129 -- пропущена директива endif;

130 -- ошибка в начальных условных определениях;

131 -- заголовок не соответствует предыдущему определению;

132 -- критическая ошибка диска;

133 -- нельзя вычислить данное выражение;

134 -- некорректное завершение выражения;

135 -- неверный спецификатор формата;

136 -- недопустимая косвенная ссылка;

137 -- здесь не допускается использование структурной переменной;

138 -- нельзя вычислить без блока system;

139 -- доступ к данному символу отсутствует;

140 -- недопустимая операция с плавающей запятой;

141 -- нельзя выполнить компиляцию оверлеев в память;

142 -- должна использоваться переменная-процедура или функция;

143 -- недопустимая ссылка на процедуру или функцию;

144 -- этот модуль не может использоваться в качестве оверлейного.

Сообщения об ошибках времени исполнения программы:;

1 -- не найден файл;

3 -- не найден путь;

4 -- слишком много открытых файлов;

5 -- отказано в доступе к файлу;

6 -- недоступный файловый канал;

12 -- недействительный код доступа к файлам;

15 -- недопустимый номер дисководов;

16 -- нельзя удалить текущий каталог;

17 -- нельзя при именовании указывать разные дисководы;

100 -- ошибка чтения диска;

101 -- ошибка записи на диск;

102 -- файлу не присвоено имя;

103 -- файл не открыт;

104 -- файл не открыт для ввода;

105 -- файл не открыт для вывода;

106 -- неверный числовой формат;

150 -- диск защищен от записи;

151 -- неизвестный модуль;

152 -- дисковод находится в состоянии "не готов";

153 -- неопознанная команда;

154 -- в исходных данных;

155 -- при запросе к диску неверная длина структуры;

156 -- ошибка при операции установки головок на диске;

157 -- неизвестный тип носителя;

158 -- сектор не найден;

159 -- кончилась бумага на устройстве печати;

160 -- ошибка при записи на устройство;

161 -- ошибка при чтении с устройства;

162 -- сбой аппаратуры;

200 -- деление на нуль;

201 -- ошибка при проверке границ;

202 -- переполнение стека;

203 -- переполнение динамически распределяемой области памяти;

204 -- недействительная операция ссылки;

205 -- переполнение операции с плавающей запятой;

206 -- исчезновение порядка при операции плавающей запятой;

207 -- недопустимая операция с плавающей запятой;

208 -- не установлена подсистема управления оверлеями;

209 -- ошибка чтения оверлейного файла.

Приложение 4. Дополнительные листинги программ

1. Решение системы линейных алгебраических уравнений Ax=b методом Гаусса.

program Slau;

uses crt;

const size=30; {максимально допустимая размерность}

type matrix=array [1..size,1..size+1] 
 of real;

type vector=array [1..size] of real;

function GetNumber (s:string;
  a,b:real):real;

{Ввод числа из интервала a,b. 
 Если a=b, то число любое}

var n:real;

begin

 repeat

  write (s);

  {$I-}readln (n);{$I+}

  if (IoResult<>0) then 
    writeln ('Введено не число!')

  else if (a<b) and ((n<a) or (n>b)) then

   writeln ('Число не в интервале от ',
    a,' до ',b)

  else break;

 until false;

 GetNumber:=n;

end;

procedure GetMatrix (n,m:integer; 
 var a:matrix); {ввод матрицы}

var i,j:integer; si,sj: string [3];

begin

 for i:=1 to n do begin

  str (i,si);

  for j:=1 to m do begin

   str (j,sj);

   a[i,j]:=GetNumber ('a['+ si+ ','+ sj+
   ']=', 0,0);

  end;

 end;

end;

procedure GetVector (n:integer; 
  var a:vector); {ввод вектора}

var i:integer; si:string [3];

begin

 for i:=1 to n do begin

  str (i,si);

  a[i]:=GetNumber ('b['+si+']=',0,0);

 end;

end;

procedure PutVector (n:integer; 
  var a:vector); {вывод вектора}

var i:integer;

begin

 writeln;

 for i:=1 to n do writeln (a[i]:10:3);

end;

procedure MV_Mult (n,m:integer;
  var a:matrix;var x,b:vector);

{умножение матрицы на вектор}

var i,j:integer;

begin

 for i:=1 to n do begin

  b[i]:=0;

  for j:=1 to m do b[i]:=b[i]+a[i,j]*x[j];

 end;

end;

function Gauss (n:integer; var a:matrix;
   var x:vector):boolean;

{метод Гаусса решения СЛАУ}

{a - расширенная матрица системы}

const eps=1e-6; {точность расчетов}

var i,j,k:integer;

    r,s:real;

begin

 for k:=1 to n do begin  {перестановка 
        для диагонального преобладания}

  s:=a[k,k];

  j:=k;

  for i:=k+1 to n do begin

   r:=a[i,k];

   if abs(r)>abs(s) then begin

    s:=r;

    j:=i;

   end;

  end;

  if abs(s)<eps then begin 
   {нулевой определитель, нет решения}

   Gauss:=false;

   exit;

  end;

  if j<>k then

  for i:=k to n+1 do begin

   r:=a[k,i];

   a[k,i]:=a[j,i];

   a[j,i]:=r;

  end;  {прямой ход метода}

  for j:=k+1 to n+1 do a[k,j]:=a[k,j]/s;

  for i:=k+1 to n do begin

   r:=a[i,k];

   for j:=k+1 to n+1 do 
     a[i,j]:=a[i,j]-a[k,j]*r;

  end;

 end;

 if abs(s)>eps then begin {обратный ход}

  for i:=n downto 1 do begin

   s:=a[i,n+1];

   for j:=i+1 to n do s:=s-a[i,j]*x[j];

   x[i]:=s;

  end;

  Gauss:=true;

 end

 else Gauss:=false;

end;

var a,a1:matrix;

    x,b,b1:vector;

    n,i,j:integer;

begin

 n:=trunc(GetNumber 
 ('Введите размерность матрицы: ',2,size));

 GetMatrix (n,n,a);

 writeln ('Ввод правой части:');

 GetVector (n,b);

 for i:=1 to n do begin 
  {делаем расширенную матрицу}

  for j:=1 to n do a1[i,j]:=a[i,j];

  a1[i,n+1]:=b[i];

 end;

 if Gauss (n,a1,x)=true then begin

  write ('Решение:');

  PutVector (n,x);

  write ('Проверка:');

  MV_Mult (n,n,a,x,b1);

  PutVector (n,b1);

 end

 else write ('Решения нет');

 reset (input); readln;

end.

2. Процедурно-ориентированная реализация задачи сортировки одномерного массива по возрастанию.
program sort;

const size=100;

type vector=array [1..size] of real;

procedure GetArray (var n:integer; 
 var a:vector);

var i:integer;

begin
 repeat
  writeln ('Введите размерность массива:');

  {$I-}readln (n); {$I+}

  if (IoResult<>0) or (n<2) or (n>size)
  then writeln
 ('Размерность должна быть от 2 до ',size);

 until (n>1) and (n<size);

 for i:=1 to n do begin

  write (i,' элемент=');

  readln (a[i]);

 end;

end;

procedure PutArray (n:integer; 
  var a:vector);

var i:integer;

begin

 writeln;

 for i:=1 to n do writeln (a[i]:10:3);

end;

procedure sortArray (n:integer; 
  var a:vector);

var i,j:integer; buf:real;

begin

 for i:=1 to n do

 for j:=i+1 to n do if a[i]>a[j] then begin

   buf:=a[i]; a[i]:=a[j]; a[j]:=buf;

 end;

end;

var a:vector;

    n:integer;

begin

 GetArray (n,a);

 sortArray (n,a);

 write ('Отсортированный массив:');

 PutArray (n,a);

end.

3. Вычисление всех миноров второго порядка в квадратной матрице.
program minor2_count;

const size=10;

type Matrix= array [1..size,1..size] 
 of real;

function minor2 (n:integer;
   i,j,l,k:integer; a:matrix):real;

begin

 minor2:=a[i,j]*a[l,k]-a[l,j]*a[i,k];

end;

procedure Input2 (var n:integer;
  maxn:integer; var a:matrix);

var i,j:integer;

begin

 repeat

  writeln;

  write ('Введите размерность матрицы ',

   '(от 2 до ',size,' включительно):');

  readln (n);

 until (n>1) and (n<size);

 for i:=1 to n do begin

  writeln;

  write ('Введите ',i,' строку матрицы:');

  for j:=1 to n do read (a[i,j]);

 end;

end;

var i,j,k,l,n:integer;

    s:real;

    a:matrix;

begin

 Input2 (n,size,a);

 for i:=1 to n do

 for j:=1 to n do

 for l:=i+1 to n do

 for k:=j+1 to n do begin

  s:=minor2 (n,i,j,l,k,a);

  writeln;

  writeln ('Минор [',i,',',j,']');

  writeln ('   [',l,',',k,']=',s:8:3);

 end;

end.

4. Учебная база данных "Студенты".
type student = record 
 {Определение записи "Студент"}

 name:string[20];

 balls:array [1..4] of integer;

end;

const filename='students.dat'; 
                {Имя базы данных}

var s:student; {Текущая запись}

    f:file of student; {Файл базы данных}

    kol,current:longint; 
    {Количество записей и текущая запись}

    size:integer; {Размер записи в байтах}

    st1,st2:string; 
     {Буферные строки для данных}

procedure Warning (msg:string); 
 {Сообщение-предупреждение}

begin

 writeln; writeln (msg);

 write ('Нажмите Enter для продолжения');

 reset (input); readln;

end;

procedure out; {Закрытие базы и выход}

begin

 close (f); halt;

end;

procedure Error (msg:string); 
{Сообщение об ошибке + выход из программы}

begin
 writeln; writeln (msg);

 write ('Нажмите Enter для выхода');

 reset (input); readln; out;

end;

procedure open; {открыть, при необходимости

  создать файл записей}

begin

 assign (f,filename);

 repeat

  {$I-} reset (f); {$I+}

  if IoResult <> 0 then begin

   Warning 
  ('Не могу открыть файл '+filename+

   '... Будет создан новый файл');

   {$I-}rewrite (f);{$I+}

   if IoResult <> 0 then

    Error ('Не могу создать файл! '+

    'Проверьте права и состояние диска');

  end

  else break;

 until false;

end;

procedure getsize (var kol:longint;

  var size:integer);

{Вернет текущее число записей kol и 
 размер записи в байтах size}

begin

 reset (f);

 size:=sizeof(student);

 if filesize(f)=0 then kol:=0

 else begin

  seek(F, Filesize(F));

  kol:=filepos (f);

 end;

end;

function getname (s:string):string;

{Переводит строку в верхний регистр 
 c учетом кириллицы DOS}

var i,l,c:integer;

begin

 l:=length(s);

 for i:=1 to l do begin

  c:=ord(s[i]);

  if (c>=ord('а')) and (c<=ord('п')) 
   then c:=c-32

  else if (c>=ord('р')) and (c<=ord('я'))

   then c:=c-80;

  s[i]:=Upcase(chr(c));

 end;

 getname:=s;

end;

procedure prints;

{Вспомогательная процедура печати -
 печатает текущую s}

var i:integer;

begin

 write (getname(s.name),': ');

 for i:=1 to 4 do begin

  write (s.balls[i]);

  if i<4 then write (',');

 end;

 writeln;

end;

procedure print (n:integer); {Вывести 
 запись номер n (с переходом к ней)}

begin

 seek (f,n-1); read (f,s); prints;

end;

procedure go (d:integer); {Перейти на d 
  записей по базе}

begin

 writeln;

 write ('Текущая запись: ');

 if current=0 then writeln ('нет')

 else begin

  writeln (current);

  print (current);

 end;

 current:=current+d;

 if current<1 then begin

  Warning ('Не могу перейти на запись '+

   'с номером меньше 1');

  if kol>0 then current:=1

  else current:=0;

 end

 else if current>kol then begin

  str (kol,st1);

  Warning ('Не могу перейти на запись '+

   'с номером больше '+st1);

  current:=kol;

 end

 else begin

  writeln ('Новая запись: ',current);

  print (current);

 end;

end;

procedure search; 
 {Поиск записи в базе по фамилии}

var i,found,p:integer;

begin

 if kol<1 then

  Warning ('База пуста! Искать нечего')

 else begin

  writeln;

  write ('Введите фамилию (часть фамилии)',

   ' для поиска, регистр символов любой:');

  reset (input);

  readln (st1);

  st1:=getname(st1);

  seek (f,0);

  found:=0;

  for i:=0 to kol-1 do begin

   read (f,s);

   p:=pos(st1,getname(s.name));

   if p>0 then begin

    writeln ('Запись номер ',i+1);

    prints;

    found:=found+1;

    if found mod 10 = 0 then 
     Warning ('Пауза...');

    {Пауза после вывода 10 найденных}

   end;

  end;

  if found=0 then 
    Warning ('Ничего не найдено...');

 end;

end;

procedure add; 
  {Добавить запись в конец базы}

var i,b:integer;

begin
 repeat
  writeln;

  write ('Введите фамилию студента ',

   'для добавления:');

  reset (input);

  readln (st1);

  if length(st1)<1 then begin

   Warning ('Слишком короткая строка!'+

    ' Повторите ввод');

   continue;

  end

  else if length(st1)>20 then begin

   Warning ('Слишком длинная строка! '+

    'Будет обрезана до 20 символов');

   st1:=copy (st1,1,20);

  end;

  s.name:=st1;

  break;

 until false;

 for i:=1 to 4 do begin

  repeat

   writeln; {следовало бы предусмотреть

       возможность ввода не всех оценок}

   write ('Введите оценку ',i,' из 4:');

   {$I-}readln (b);{$I+}

   if (IoResult<>0) or (b<2) or (b>5) 
   then begin

    Warning ('Неверный ввод! Оценка - '+

      'это число от 2 до 5! Повторите.');

    continue;

   end

   else begin

    s.balls[i]:=b; break;

   end;

  until false;

 end;

 seek (f,filesize(f));

 write (f,s); kol:=kol+1; current:=kol;

end;

procedure delete; {Удаление текущей записи}

var f2:file of student; i:integer;

begin

 if kol<1 then

  Warning ('База пуста! Удалять нечего')

 else begin

  assign (f2,'students.tmp');

  {$I-}rewrite(f2);{$I+}

  if IoResult<>0 then begin

   Warning ('Не могу открыть новый файл '+

     'для записи!'+#13+#10+

    ' Операция невозможна. Проверьте '+

      'права доступа и текущий диск.');

   Exit;

  end;

  seek (f,0);

  for i:=0 to kol-1 do begin

   if i+1<>current then begin 
   {переписываем все записи, кроме текущей}

    read (f,s); write (f2,s);

   end;

  end;

  close (f); {закрываем исходную БД}

  erase (f); {Удаляем исходную БД, 
    проверка IoResult опущена!}

  rename (f2,filename); {Переименовываем f2

    в имя БД}

  close (f2); {Закрываем 
       переименованный f2}

  open; {Связываем БД с прежней 
         файловой переменной f}

  kol:=kol-1;

  if current>kol then current:=kol;

 end;

end;

procedure sort; 
 {сортировка базы по фамилии студента}

var i,j:integer;

 s2:student;

begin

 if kol<2 then

  Warning ('В базе нет 2-х записей!'+

   ' Сортировать нечего')

 else begin

  for i:=0 to kol-2 do begin 
   {Обычная сортировка}

   seek (f,i); {только в учебных целях -
               работает неоптимально}

   read (f,s);{и много обращается к диску!}

   for j:=i+1 to kol-1 do begin

    seek (f,j);

    read (f,s2);

    if getname(s.name)>getname(s2.name)

    then begin

     seek (f,i); write (f,s2);

     seek (f,j); write (f,s);

     s:=s2;  {После перестановки в s уже

              новая запись!}

    end;

   end;

  end;

 end;

end;

procedure edit; {редактирование записи
   номер current}

var i,b:integer;

begin

 if (kol<1) or (current<1) or (current>kol)

  then  Warning ('Неверный номер '+

  'текущей записи! Не могу редактировать')

 else begin

  seek (f,current-1);

  read (f,s);

  repeat

   writeln ('Запись номер ',current);

   writeln ('Выберите действие:');

   writeln ('1. Фамилия (',s.name,')');

   for i:=1 to 4 do

    writeln (i+1,'. Оценка ',i,

     ' (',s.balls[i],')');

   writeln ('0. Завершить редактирование');

   reset (input);

   {$I-}readln (b);{$I+}

   if (IoResult<>0) or (b<0) or (b>5) then

    Warning ('Неверный ввод! Повторите')

   else begin

    if b=1 then begin
     write ('Введите новую фамилию:');

     {для простоты здесь нет}

     {проверок корректности}

      reset (input); readln (s.name); 
    end

    else if b=0 then break

    else begin

     write ('Введите новую оценку:');

     reset (input); readln (s.balls[b-1]);

    end;

   end;

  until false;

  seek (f,current-1); 
  {Пишем, даже если запись не менялась -}

  write (f,s); {в реальных проектах 
                так не делают}

 end;

end;

procedure menu; {Управление главным меню и

   вызов процедур}

var n:integer;

begin

 repeat

  writeln;

  writeln ('Выберите операцию:');

  writeln ('1 - вперед');

  writeln ('2 - назад');

  writeln ('3 - поиск по фамилии');

  writeln ('4 - добавить в конец');

  writeln ('5 - удалить текущую');

  writeln ('6 - сортировать по фамилии');

  writeln ('7 - начало базы');

  writeln ('8 - конец базы');

  writeln ('9 - изменить текущую');

  writeln ('0 - выход');

  reset (input);

  {$I-}read (n);{$I+}

  if (IoResult<>0) or (n<0) or (n>9) 
  then begin

   Warning ('Неверный ввод!');

   continue;

  end

  else break;

 until false;

 case n of

  1: go (1);

  2: go (-1);

  3: search;

  4: add;

  5: delete;

  6: sort;

  7: go (-(current-1));

  8: go (kol-current);

  9: edit;

  0: out;

 end;

end;

begin {Главная программа}

 open;

 getsize (kol,size);

 str(kol,st1);

 str(size,st2);

 writeln;

 writeln('==============================');

 writeln('Учебная база данных "Студенты"');

 writeln('==============================');

 Warning ('Файл '+FileName+

  ' открыт'+#13+#10+

  'Число записей='+st1+#13+#10+

  'Размер записи='+st2+#13+#10);

  {+#13+#10 - добавить к строке символы

    возврата каретки и первода строки}

 if kol=0 then current:=0

 else current:=1;

 repeat

  menu;

 until false;

end.
5. Программа содержит коды часто используемых клавиш и печатает их названия.
uses crt;

const ESC=#27; ENTER=#13; F1=#59;  
  F10=#68; TAB=#9;  SPACE=#32;

  UP=#72;  DOWN=#80; LEFT=#75; RIGHT=#77;
  HOME=#71; END_=#79;

  PAGE_UP=#73; PAGE_DN=#81;

var ch:char;

begin

 clrscr;

 repeat

  ch:=Upcase(readkey);

  case ch of

   'A'..'z': write ('Letter');

   SPACE: write ('SPACE');

   ENTER: write ('ENTER');

   TAB: write ('TAB');

   #0: begin

    ch:=readkey;

    case ch of

     F1: write ('F1');
     F10: write ('F10');

     LEFT: write ('LEFT');
     RIGHT: write ('RIGHT');

     UP: write ('UP');
     DOWN: write ('DOWN');

     HOME: write ('HOME');
 
     END_: write ('END');

     PAGE_UP: write ('PgUp');
     PAGE_DN: write ('PgDn');

    end;

   end;

   else begin

   end;

  end;

 until ch=Esc;

end.

6.1. Программа позволяет двигать по текстовому экрану "прицел" с помощью клавиш со стрелками.
uses crt; 
{$V-} {отключили строгий контроль типов}

const ESC=#27; UP=#72;  DOWN=#80;
      LEFT=#75; RIGHT=#77;

var ch:char;

procedure Draw (x,y:integer;mode:boolean);
{mode определяет, нарисовать или стереть}

var sprite:array [1..3] of string [3]; 
 {"прицел", заданный массивом sprite}

 i:integer;

begin

 sprite[1]:='/|\';

 sprite[2]:='-=-';

 sprite[3]:='\|/';

 if mode=true then textcolor (White)

 else textcolor (Black);

 for i:=y to y+2 do begin

  gotoxy (x,i); write (sprite[i-y+1]);

 end;

 gotoxy (x+1,y+1);

end;

procedure status (n:integer; s:string);
{рисует строку статуса 
  внизу или вверху экрана}

begin

 textcolor (Black); textbackground (White);

 gotoxy (1,n); write (' ':79);

 gotoxy (2,n); write (s);

 textcolor (White); textbackground (Black);

end;

var x,y:integer;

begin

 textMode (cO80);

 status (1,'Пример управления движением!');

 status(25,'Стрелки-управление;ESC-выход');

 x:=10; y:=10;

 repeat

  Draw (x,y,true);

  ch:=Upcase(readkey);

  case ch of

   #0: begin

    ch:=readkey;

    Draw (x,y,false);

    case ch of

 LEFT:  if x>1 then x:=x-1;

 RIGHT: if x<77 then x:=x+1;

 UP:    if y>2 then y:=y-1;

 DOWN:  if y<22 then y:=y+1;

    end;

   end;

  end;

 until ch=ESC;

 clrscr;

end.

6.2. Эта версия программы 6.1 позволяет "прицелу" продолжать движение до тех пор, пока он не натолкнется на край экрана.

uses crt;

{$V-}

const ESC=#27; UP=#72; DOWN=#80; 
  LEFT=#75; RIGHT=#77;

const goleft=1; GoRight=2; goup=3; 
  godown=4; gostop=0;

 {возможные направления движения}

const myDelay=1000; {задержка для Delay}

var ch:char; LastDir:integer; 
{последнее направление движения}

procedure Draw (x,y:integer;mode:boolean);

var sprite:array [1..3] of string [3];

    i:integer;

begin

 sprite[1]:='/|\';

 sprite[2]:='-=-';

 sprite[3]:='\|/';

 if mode then textcolor (White)

 else textcolor (Black);

 for i:=y to y+2 do begin

  gotoxy (x,i);

  write (sprite[i-y+1]);

 end;

 gotoxy (x+1,y+1);

end;

procedure status (n:integer; s:string);

begin

 textcolor (Black); textbackground (White);

 gotoxy (1,n); write (' ':79);

 gotoxy (2,n); write (s);

 textcolor (White); textbackground (Black);

end;

var x,y:integer;

begin

 clrscr;

 status(1,'Управление движением-2');

 status(25,'Стрелки-управление;ESC-выход');

 x:=10; y:=10; LastDir:=goleft;

 repeat {бесконечный цикл работы программы}

  repeat {цикл до нажатия клавиши}

   Draw (x,y,true); Delay (myDelay);
   Draw (x,y,false);

   case LastDir of

    goLeft:

     if x>1 then Dec(x)

     else begin

      x:=1; LastDir:=gostop;

     end;

    GoRight:

     if x<77 then inc(x)

     else begin

      x:=77; LastDir:=gostop;

     end;

    goUp:

     if y>2 then Dec(y)

     else begin

      y:=2; LastDir:=gostop;

     end;

    goDown:

     if y<22 then inc(y)

     else begin

      y:=22; LastDir:=gostop;

     end;

   end;

  until keyPressed;

  {обработка нажатия клавиши}

  ch:=Upcase(readkey);

  case ch of

   #0: begin

    ch:=readkey;

    case ch of

     LEFT:  LastDir:=goLeft;

     RIGHT: LastDir:=GoRight;

     UP:    LastDir:=goUp;

     DOWN:  LastDir:=goDown;

    end;

   end;

   ESC: halt;

  end;

 until false;

end.

7. Демо-программа для создания несложного двухуровневого меню пользователя. Переопределив пользовательскую часть программы, на ее основе можно создать собственный консольный интерфейс.

uses crt; { Глобальные данные: }

const 
maxmenu=2; {количество меню}  

maxpoints=3; {макс. количество пунктов}

var x1,x2,y: array [1..maxmenu] of integer; 
{x1,x2- начало и конец каждого меню, 
 y- строка начала каждого меню}

 kolpoints, points: array [1..maxmenu] of 
 integer;{Кол-во пунктов и текущие пункты }

 text: array [1..maxmenu,1..maxpoints] 
  of string[12]; { Названия пунктов }
 txtcolor, textback, cursorback:integer; 
  { Цвета текста, фона, курсора}

 mainhelp:string[80]; { Строка помощи }

procedure DrawMain (s:string); {Очищает 
   экран, рисует строку главного меню s }

begin   Window (1,1,80,25);

 textcolor (txtcolor);  
 textbackground (textback);

 clrscr; gotoxy (1,1); write (s);

end;

procedure DrawHelp (s:string); 
 { Выводит подсказку s }

var i:integer; begin

 textcolor (txtcolor);  
 textbackground (textback); gotoxy (1,25);

 for i:=1 to 79 do write (' ');

 gotoxy (1,25); write (s);

end;

procedure doubleFrame (x1,y1,x2,y2:integer;
  Header: string);

{ Процедура рисует двойной рамкой окно }

var i,j: integer;

begin gotoxy (x1,y1); 
write ('╔');

 for i:=x1+1 to x2-1 do write('═');
 write ('╗');

 for i:=y1+1 to y2-1 do begin 
  gotoxy (x1,i);  write('║');

  for j:=x1+1 to x2-1 do write (' ');
  write('║');

 end;

 gotoxy (x1,y2);  write('╚');   
 for i:=x1+1 to x2-1 do write('═'); 
 write('╝');

 gotoxy (x1+(x2-x1+1-Length(Header)) 
  div 2,y1);

 write (Header); {Выводим заголовок}

 gotoxy (x1+1,y1+1);

end;

procedure clearFrame (x1,y1,x2,y2:integer);

var i,j:integer;

begin textbackground (textback);

 for i:=y1 to y2 do begin

  gotoxy (x1,i);

  for j:=x1 to x2 do write (' ');

 end;

end;

procedure cursor (Menu,Point: integer;
 Action: boolean);{ Подсвечивает (если 
 Action=true) или гасит п. Point меню Menu}

begin textcolor (Txtcolor);

 if Action=true then 
  textbackground (cursorBack)

 else textbackground (textBack);

 gotoxy (x1[Menu]+1,y[Menu]+Point);

 write (text[Menu][Point]);

end;

procedure DrawMenu (Menu:integer; 
 Action: boolean);{Рисует меню с номером
 Menu, если Action=true, иначе стирает }

var i:integer;

begin

 if Action=true then textcolor (Txtcolor)

 else textcolor (textBack);

 textbackground (textBack);

 doubleFrame (x1[Menu], y[Menu], x2[Menu],
  y[Menu]+1+KolPoints[Menu],'');

 for i:=1 to KolPoints[Menu] do begin

  gotoxy (x1[Menu]+1, y[Menu]+i);

  writeln (text[Menu][i]);

 end;

end;

{Часть, определяемая пользователем}

procedure Init; { Установка глобальных
  данных и начальная отрисовка }

begin
 txtcolor:=yELLOW; textback:=BLUE; 
 cursorback:=LIGHTcyAN;

 kolpoints[1]:=2; kolpoints[2]:=1; 
  {пунктов в каждом меню}

 points[1]:=1; points[2]:=1; 
  {выбран по умолчанию в каждом меню}

 x1[1]:=1;  x2[1]:=9;  y[1]:=2; 
 text[1,1]:='Запуск'; text[1,2]:='Выход ';

 x1[2]:=9;  x2[2]:=22; y[2]:=2; 
 text[2,1]:='О программе';

 DrawMain ('Файл   Справка');

 MainHelp:='ESC - Выход из программы '+

 'ENTER - выбор пункта меню    '+

 'Стрелки - перемещение';

 DrawHelp(MainHelp);

end;

procedure Work; { Рабочая процедура }

var i,kol:integer; ch:char;

begin

 DrawHelp('Идет расчет...'); 
 { Строка статуса }

 textcolor (LIGHTGRAY);
 textbackground  (BLACK);

 { Выбираем цвета для работы в окне }

 doubleFrame (2,2,78,24,' Расчет ');

 Window (3,3,77,23); 
{Секция действий, выполняемых программой}

 writeln;

 write ('Введите число шагов: ');

 {$I-}read (kol);{$I+}

 if IoResult<>0 then writeln 
  ('Ошибка! Вы ввели не число')

 else if kol>0 then begin

  for i:=1 to kol do 
    writeln ('Выполняется шаг ',i);

  writeln ('Все сделано!');

 end

 else writeln ('Ошибка! Число больше 0');

{Восстановление окна и выход}

 Window (1,1,80,25);

 DrawHelp('Нажмите любую клавишу...');

 ch:=readkey;

 clearFrame (2,2,78,24); { Стираем окно }

end;

procedure Out; { Очистка экрана и выход}

begin

 textcolor (LIGHTGRAY); 
 textbackground (BLACK); clrscr; halt(0);

end;

procedure Help; {Окно с информацией}

var ch:char;

begin

 textcolor (Txtcolor); 
 textbackground (textback);

 doubleFrame (24,10,56,13,' О программе ');

 DrawHelp ('Нажмите клавишу...');

 gotoxy (25,11); 
 writeln(' Демонстрация простейшего меню');

 gotoxy (25,12); 
 write (  '  Новосибирск, НГАСУ');

 ch:=readkey;

 clearFrame (24,10,58,13);

end;

procedure command (Menu,Point:integer); 
{Вызывает процедуры после выбора в меню }

begin

 if Menu=1 then begin

  if Point=1 then Work

  else if Point=2 then Out;

 end

 else begin

  if Point=1 then Help;

 end;

end;

{Конец части пользователя }

procedure MainMenu (Point, 
 HorMenu:integer); { Поддерживает систему 
  одноуровневых меню }

var ch: char; funckey:boolean;

begin

 Points[HorMenu]:=Point;

 DrawMenu (HorMenu,true);

repeat

 cursor (HorMenu,Points[HorMenu],true);

 ch:=readkey;

 cursor (HorMenu,Points[HorMenu],false);

 if ch=#0 then begin

  funckey:=true;  ch:=readkey;

 end

 else funckey:=false;

 if funckey=true then begin

  ch:=Upcase (ch);

  if ch=#75 then begin { Стрелка влево }

   DrawMenu (HorMenu,false); 
   HorMenu:=HorMenu-1;

   if (HorMenu<1) then HorMenu:=maxMenu;

   DrawMenu (HorMenu,true);

  end

  else if ch=#77 then begin 
   { Стрелка вправо }

   DrawMenu (HorMenu,false);
   HorMenu:=HorMenu+1;

   if (HorMenu>maxMenu) then HorMenu:=1;

   DrawMenu (HorMenu,true);

  end

  else if ch=#72 then begin 
   { Стрелка вверх }

   Points[HorMenu]:=Points[HorMenu]-1;

   if Points[HorMenu]<1 then
    Points[HorMenu]:=Kolpoints[HorMenu];

  end

  else if ch=#80 then begin 
   { Стрелка вниз }

   Points[HorMenu]:=Points[HorMenu]+1;

   if (Points[HorMenu]>KolPoints[HorMenu])
   then Points[HorMenu]:=1;

  end;

 end

 else if ch=#13 then begin 
  { Клавиша ENTER }

  DrawMenu (HorMenu,false);

  command (HorMenu,Points[HorMenu]);

  DrawMenu (HorMenu,true);

  DrawHelp (MainHelp);

 end;

until (ch=#27) and (funckey=false); 
{ Пока не нажата клавиша ESC }

end;

{ Основная программа }

begin 
 Init; 
 MainMenu (1,1); 
 Out; 
end.

8. Простейший "генератор" программы на Паскале. Из входного файла, содержащего текст, генерируется программа для листания этого текста.

program str2Pas;

uses crt; label 10,20;

var ch:char;str:string;

 I,J,Len,count:word; InFile,OutFile:text;

procedure Error (ErNum:char);

begin

 case ErNum of

  #1: writeln

   ('Запускайте с 2 параметрами -',#13,#10,

'именами входного и выходного файла.',
#13,#10,

'Во входном файле содержится текст',
#13,#10,

'в обычном ASCII-формате,',#13,#10,

'в выходном будет программа на Паскале');

  #2:

  writeln 
  (' Не могу открыть входной файл!');

  #3:

  writeln 
  (' Не могу открыть выходной файл!');

  else writeln (' Неизвестная ошибка!');

 end;

 halt;

end;

begin

 if Paramcount<>2 then Error (#1);

 assign (InFile,Paramstr(1));

 reset (InFile);

 if (IoResult<>0) then Error (#2);

 assign (OutFile,Paramstr(2));

 rewrite (OutFile);

 if (IoResult<>0) then Error (#3);

{ Вписать заголовок программы }

 writeln (OutFile,'uses crt;');

 write (OutFile,'const colstr=');

{ Узнать число строк текста }

 count:=0;

 while not Eof (InFile) do begin

  readLn (InFile,str);

  count:=count+1;

 end;

 reset (InFile);

 writeln (OutFile,count,';');

 { Следующий сегмент программы: }

 writeln (OutFile,'var ch:char;');

 writeln (OutFile,'    List:boolean;');

 writeln (OutFile,
  '    I,start,endstr:word;');

 writeln (OutFile,
 '    ptext:array [1..colstr] of string;');

 writeln (OutFile,'begin');

 { Строки листаемого текста: }

 for I:=1 to count do begin

  Len:=0;

  repeat

   if (Eof (InFile)=true) then goto 10;

   read (InFile,ch);

   if ch=#39 then begin

    Len:=Len+1; str[Len]:=#39; 
    Len:=Len+1; str[Len]:=#39;

   end

   else if ch=#13 then begin

    read (InFile,ch);

    if (ch=#10) then goto 10

    else goto 20;

   end

   else begin

20:

    Len:=Len+1; str[Len]:=ch;

   end;

  until false;

10:

  write (OutFile,' ptext[',I,']:=''');

  for J:=1 to Len do 
    write (OutFile,str[J]);

  writeln (OutFile,''';');

 end;

 { Сегмент программы }

 writeln (OutFile,' textcolor (YELLOW);');

 writeln (OutFile,
  ' textbackground (Blue);');

 writeln (OutFile,
  ' List:=true; start:=1;');

 { Последняя строка на экране: }

 if (count>25) then 
  writeln (OutFile,' endstr:=25;')

 else writeln (OutFile,' endstr:=colstr;');

 writeln (OutFile,' repeat');

 writeln (OutFile,
 '  if (List=true) then begin');

 writeln (OutFile,'   clrscr;');

 writeln (OutFile,
  '   for I:=start to endstr-1 do ',

  'write (ptext[I],#13,#10);');

 writeln (OutFile,
 '   write (ptext[endstr]);');

 writeln (OutFile,'   List:=false;');

 writeln (OutFile,'  end;');

 writeln (OutFile,'  ch:=readkey;');

 writeln (OutFile,
  '  if ch= #0 then begin');

 writeln (OutFile,'   ch:=readkey;');

 writeln (OutFile,'   case ch of');

 writeln (OutFile,'    #72: begin');

 writeln (OutFile,
  '     if start>1 then begin');

 writeln (OutFile,'      start:=start-1;');

 writeln (OutFile,
  '      endstr:=endstr-1;');

 writeln (OutFile,'      List:=true;');

 writeln (OutFile,'     end;');

 writeln (OutFile,'    end;');

 writeln (OutFile,'    #80: begin');

 writeln (OutFile,
  '     if endstr<colstr then begin');

 writeln (OutFile,'      start:=start+1;');

 writeln (OutFile,
  '      endstr:=endstr+1;');

 writeln (OutFile,'      List:=true;');

 writeln (OutFile,'     end;');

 writeln (OutFile,'    end;');

 { Листание PgUp и PgDn }

 if (count>25) then begin

  writeln (OutFile,'    #73: begin');

  writeln (OutFile,
  '     if start>1 then begin');

  writeln (OutFile,
  '      start:=1; endstr:=25;');

  writeln (OutFile,'      List:=true;');

  writeln (OutFile,'     end;');

  writeln (OutFile,'    end;');

  writeln (OutFile,'    #81: begin');

  writeln (OutFile,
   '     if endstr<colstr then begin');

  writeln (OutFile,
'      start:=colstr-24; endstr:=colstr;');

  writeln (OutFile,'      List:=true;');

  writeln (OutFile,'     end;');

  writeln (OutFile,'    end;');

 end;

 { Заключительный сегмент }

 writeln (OutFile,'    else begin  end;');

 writeln (OutFile,'   end;');

 writeln (OutFile,'  end');

 writeln (OutFile,'  else begin');

 writeln (OutFile,'   case ch of');

 writeln (OutFile,'    #27: begin');

 writeln (OutFile,
 '     textcolor (LightGray);');

 writeln (OutFile,
  '     textbackground (Black);');

 writeln (OutFile,'     clrscr;');

 writeln (OutFile,'     halt;');

 writeln (OutFile,'    end;');

 writeln (OutFile,'    else begin');

 writeln (OutFile,'    end;');

 writeln (OutFile,'   end;');

 writeln (OutFile,'  end;');

 writeln (OutFile,' until false;');

 writeln (OutFile,'end.');

 close (InFile); close (OutFile);

 writeln ('OK.');

end.

9. Шаблон программы для работы с матрицами и текстовыми файлами.
program Files;{ Программа демонстрирует
 работу с текстовыми файлами и матрицами }

const rows=10; cols=10;

type matrix=array [1..rows,1..cols] 
 of real;

var f1,f2:text; a,b:matrix; 
 Name1,Name2:string; n,m:integer;

procedure Error (msg:string);

begin

 writeln; writeln (msg);

 writeln ('Нажмите Enter для выхода');

 reset (Input); readln; halt;

end;

procedure readDim (var f:text; 
 var n,m:integer);{ Читает из файла f 
 размерности матрицы: n - число строк, 
 m - число столбцов. Если n<0 или n>rows 
(число строк) или m<0 или m>cols (число 
 столбцов), прервет работу. }

var s:string;

begin

 {$I-}read (f,n);{$I+}

 if (IoResult<>0) or (n<0) or (n>rows) 
 then begin

  str (rows,s);

  Error ('Неверное число строк '+

  'в файле данных!'+#13+#10+

  'должно быть от 1 до '+s);

 end;

 {$I-}read (f,m);{$I+}

 if (IoResult<>0) or (m<0) or (m>cols) 
 then begin

  str (cols,s);

  Error ('Неверное число столбцов '+

   'в файле данных!'+#13+#10+

   'должно быть от 1 до '+s);

 end;

end;

procedure readMatrix (var f:text;
  n,m:integer; var a:matrix);

{ Читает из файла f матрицу a 
   размерностью n*m }

var i,j:integer; er:boolean;

begin

 er:=false;

 for i:=1 to n do

 for j:=1 to m do begin

  {$I-}read (f,a[i,j]);{$I+}

  if IoResult<>0 then begin

   er:=true; a[i,j]:=0;

  end;

 end;

 if er=true then begin

  writeln;

  writeln 
  ('В прочитанных данных есть ошибки!');

  writeln ('Неверные элементы матрицы',

   ' заменены нулями');

 end;

end;

procedure writeMatrix (var f:text; 
 n,m:integer; var a:matrix);

{ Пишет в файл f матрицу a[n,m] }

var i,j:integer;

begin

 for i:=1 to n do begin

  for j:=1 to m do write (f,a[i,j]:11:4);

  writeln (f);

 end;

end;

procedure Proc1 (n,m:integer; 
  var a,b:matrix);

{ Матрицу a[n,m] пишет в матрицу b[n,m],
  меняя знаки элементов }

var i,j:integer;

begin

  for i:=1 to n do

  for j:=1 to m do b[i,j]:=-a[i,j]

end;

begin

 if Paramcount<1 then begin

  writeln ('Имя файла для чтения:');

  readLn (Name1);

 end

 else Name1:=Paramstr(1);

 if Paramcount<2 then begin

  writeln ('Имя файла для записи:');

  readLn (Name2);

 end

 else Name2:=Paramstr(2);

 assign (f1,Name1);

 {$I-}reset (f1);{$I+}

 if IoResult<>0 then 
  Error ('Не могу открыть '+Name1+
 ' для чтения');

 assign (f2,Name2);

{$I-}rewrite (f2);{$I+}

 if IoResult<>0 then 
  Error ('Не могу открыть '+Name2+
  ' для записи');

 readDim (f1,n,m);

 readMatrix (f1,n,m,a);

 Proc1 (n,m,a,b);

 writeMatrix (f2,n,m,b);

 close (f1); close (f2);

end.

10. Подсчет количества дней от введенной даты до сегодняшнего дня.

program Days;

uses Dos;

const mondays: array [1..12] of integer =
(31,28,31, 30,31,30, 31,31,30, 31,30,31);

var d,d1,d2,m1,m2,y1,y2:word;

function Leapyear (year:word):boolean;

begin

 if (year mod 4 =0) and (year mod 100 <>0)
 or (year mod 400 =0) then Leapyear:=true
 else Leapyear:=false;

end;

function correctDate
  (day,mon,year:integer):boolean;

var maxday:integer;

begin

 if (year<0) or (mon<1) or (mon>12) or
  (day<1) then correctDate:=false
 else begin

  maxday:=mondays[mon];

  if (Leapyear (year)=true) and (mon=2)
  then maxday:=29;

  if (day>maxday) then correctDate:=false
  else correctDate:=true;

 end;

end;

function KolDays (d1,m1,d2,m2,y:word):word;

var i,f,s:word;

begin

 s:=0;

 if m1=m2 then KolDays:=d2-d1

 else for i:=m1 to m2 do begin

  f:=mondays[i];

  if (Leapyear (y)=true) and (i=2) 
  then f:=f+1;

  if i=m1 then s:=s+(f-d1+1)

  else if i=m2 then s:=s+d2

  else s:=s+f;

  KolDays:=s;

 end;

end;

function countDays (day1, mon1, year1,
  day2, mon2, year2:word):word;

var f,i:word;

begin

 f:=0;

 if year1=year2 then countDays:=
  KolDays (day1, mon1, day2, mon2, year1)

 else for i:=year1 to year2 do begin

  if i=year1 then f:=
    KolDays (day1, mon1, 31, 12, year1)

  else if i=year2 then f:=f+
    KolDays (1,1,day2,mon2,year2)-1

  else f:=f+KolDays (1,1,31,12,i);

  countDays:=f;

 end;

end;

begin

 getdate (y2,m2,d2,d);

 writeln ('Год Вашего рождения?');

 readln (y1);

 writeln ('Месяц Вашего рождения?');

 readln (m1);

 writeln ('День Вашего рождения?');

 readln (d1);

 if correctDate (d1,m1,y1)=false then begin

  writeln ('Недопустимая дата!'); halt;

 end;

 if (y2<y1) or   ( (y2=y1) and

    ( (m2<m1) or ( (m2=m1) and (d2<d1))))
 then begin writeln ('Введенная дата',

  ' позднее сегодняшней!');    halt;

 end;

 d:=countDays (d1,m1,y1,d2,m2,y2);

 writeln ('Количество дней= ',d);

end.

11.1. Исходный текст модуля для поддержки мыши.

unit Mouse;

{Примеры использования – 
 см. mousetst.pas в графике,

 mousetxt.pas в текстовом режиме 80*25}

interface

var   MousePresent:boolean;

function MouseInit(var nb:integer):boolean;

{ Инициализация мыши - вызывать первой.
  Вернет true, если мышь обнаружена }

procedure Mouseshow; {Показать курсор мыши}

procedure MouseHide; {Скрыть курсор мыши}

procedure Mouseread(var x,y,bMask:integer); 
{Прочитать позицию мыши.
 Вернет через x,y координаты курсора 
 (для текстового режима см. пример),

 через bmask - состояние кнопок 
(0-отпущены,1-нажата левая,2-нажата правая,

 3-нажаты обе) }

procedure MousesetPos(x,y:word); 
 {Поставить курсор в указанную позицию}

procedure Mouseminxmaxx(minx,maxx:integer);
 {Установить границы перемещения по x}

procedure Mouseminymaxy(miny,maxy:integer);

 {Установить границы перемещения по y}

procedure setVideoPage(Page:integer); 
 {Установить нужную видеостраницу}

procedure GetVideoPage(var Page:integer);
 {Получить номер видеостраницы}

function MouseGetb(bMask:word; var count,
 Lastx, Lasty:word):word;

procedure MousekeyPreset
 (var key,sost,x,y:integer);

implementation

uses Dos;

var   r: registers;

      Mi:pointer;

function MouseInit(var nb:integer):boolean;

begin

 if MousePresent then begin

  r.Ax:=0; Intr($33,r);

  if r.Ax=0 then begin

   nb:=0; MouseInit:=false
  end

  else begin

   nb:=r.Ax; MouseInit:=true
  end

 end

 else begin

  nb:=0; MouseInit:=false
 end

end;

procedure Mouseshow;

begin

   r.Ax:=1; Intr($33,r)

end;

procedure MouseHide;

begin

   r.Ax:=2; Intr($33,r)

end;

procedure Mouseread(var x,y,bMask:integer);

begin

 r.Ax:=3; Intr($33,r);
 x:=r.cx; y:=r.dx; bMask:=r.Bx
end;

procedure MousesetPos(x,y:word);

begin

 r.Ax:=4; r.cx:=x; r.dx:=y;

 Intr($33,r)

end;

function MouseGetb(bMask:word;
 var count,Lastx,Lasty:word):word;

begin

 r.Ax:=5; r.Bx:=bMask;Intr($33,r);

 count:=r.Bx; Lastx:=r.cx;

 Lasty:=r.dx; MouseGetb:=r.Ax
end;

procedure Mouseminxmaxx(minx,maxx:integer);

begin

 r.Ax:=7; r.cx:=minx;

 r.dx:=maxx; Intr($33,r)

end;

procedure Mouseminymaxy(miny,maxy:integer);

begin

 r.Ax:=8; r.cx:=miny;

 r.dx:=maxy; Intr($33,r)

end;

procedure setVideoPage(Page:integer);

begin

 r.Ax:=$1D; r.Bx:=Page; Intr($33,r)

end;

procedure GetVideoPage(var Page:integer);

begin

 r.Ax:=$1E; Intr($33,r); Page:=r.Bx;

end;

procedure MousekeyPreset
 (var key,sost,x,y:integer);

begin

 r.Ax:=$6; r.Bx:=key; Intr($33,r);

 key:=r.Ax; sost:=r.Bx;
 x:=r.cx; y:=r.dx;

end;

begin

 GetIntVec($33,Mi);

 if Mi=nil then

  MousePresent:=false
 else if byte(Mi^)=$cE then

  MousePresent:=false
 else MousePresent:=true
end.

11.2. Тест модуля mouse.pas в графическом режиме (mousetst.pas).
program MouseTst; 
uses graph,Mouse,crt;

var grDriver : integer; grMode : integer;

 Errcode : integer;

procedure init;

begin
 grDriver:=VGA;grMode:=VGAHi;

 initgraph(grDriver, grMode, '');

 Errcode:=graphresult;

 if Errcode <> grOk then begin
  writeln('Ошибка инициализации графики:',
  grapherrormsg(Errcode)); halt;

 end;

end;

var n,x,y,x0,y0,b:integer; s1,s2:string;

begin

 init;

 mouseinit(n);

 mouseshow;

 setfillstyle (solidfill,BLACK);

 setcolor (WHITE);

 settextJustify(centertext, centertext);

 x0:=-1; y0:=-1;

 repeat

  mouseread (x,y,b);

  if (x<>x0) or (y<>y0) then begin

   str (x,s1); str (y,s2);

   bar (getmaxx div 2-50, 
    getmaxy-15,getmaxx div 2+50,getmaxy-5);

   outtextxy (getmaxx div 2, 
    getmaxy-10,s1+' '+s2);

   x0:=x; y0:=y;

  end;

 until keypressed;

 mousehide;

 closegraph;

end.

11.3. Тест модуля mouse.pas в текстовом режиме (mousetxt.pas).

program MouseTxt;

uses crt,mouse;

var n,x,y,b:integer;

 n1,k,lastx,lasty:word;

begin

 textmode(3);

 mouseinit (n);

 mouseshow;

 repeat

  mouseread (x,y,b);

  gotoxy (1,25);

  write ('x=',(x div 8 + 1):2,
   ' y=',(y div 8 + 1):2,' b=',b:2);

 until keypressed;

 mousehide;

end.

12.1. Учебная игра, использующая собственный файл ресурсов. Первый листинг содержит утилиту для создания файла ресурсов resfile из файлов *.bmp текущей директории, список которых находится в файле filelist.txt. Файлы *.bmp должны быть сохранены в режиме 16 цветов. При необходимости следует изменить в программе константу пути к Паскалю.
uses graph,crt;

const VGAPath='c:\TP7\egavga.bgi';

 FileList='filelist.txt';

resfile='attack.res';

const width=32; height=20;

const color: array [0..15] of byte=
 (0,4,2,6,1,5,3,7,8,12,10,14,9,13,11,15);

const maxx=639; maxy=479;

 cx=MAxx div 2; cy=maxy div 2;

type bmpinfo=record

 h1,h2:char;

 size,reserved,offset,b,width,
  height: longint;

 plans,bpp:word;

end;

var Driver, Mode: integer;

 DriverF: file; List,res:text;

 DriverP: pointer; s:string;

procedure Wait;

var ch:char;

begin

 reset (Input); repeat until keyPressed;

 ch:=readkey; if ch=#0 then readkey;

end;

procedure closeMe;

begin

 if DriverP <> nil then begin

  FreeMem(DriverP, Filesize(DriverF));

  close (DriverF);

 end;

 closegraph;

end;

procedure graphError;

begin

 closeMe;

 writeln('graphics error:',
  grapherrormsg(graphresult));

 writeln('Press any key to halt ');

 Wait;

 halt (graphresult);

end;

procedure InitMe;

begin

 assign(DriverF, VGAPath);

 reset(DriverF, 1);

 getmem(DriverP, Filesize(DriverF));

 Blockread(DriverF, DriverP^,
  Filesize(DriverF));

 if registerBGIdriver(DriverP)<0 then
  graphError;

 Driver:=VGA; Mode:=VGAHi;

 initgraph(Driver, Mode,'');

 if graphresult < 0 then graphError;

end;

procedure clearscreen;

begin

 setfillstyle (solidfill, White);

 bar (0,0,maxx,maxy);

end;

procedure Window 
 (x1,y1,x2,y2,color,Fillcolor:integer);

begin

 setcolor (color);

 setfillstyle (1,Fillcolor);

 bar (x1,y1,x2,y2);

 rectangle (x1+2,y1+2,x2-2,y2-2);

 rectangle (x1+4,y1+4,x2-4,y2-4);

 setfillstyle (1,DArKGrAy);

 bar (x1+8,y2+1,x2+8,y2+8);

 bar (x2+1,y1+8,x2+8,y2);

end;

procedure Error (code:integer; str:string);

begin

 Window (cx-140,cy-100,cx+140,
  cy-70,Black,YELLOW);

 case code of

  1: s:='Файл '+str+' не найден!';

  2: s:='Файл '+str+' не формата BMP-16';

  3: s:='Файл '+str+' испорчен!';

 end;

 settextjustify (Lefttext, toptext);

 settextstyle(DefaultFont, HorizDir, 1);

 outtextxy (cx-136,cy-92,s);

 Wait;

 halt(code);

end;

function Draw (x0,y0:integer; fname:string;
 transparent:boolean):integer;

var f:file of bmpinfo;

 bmpf:file of byte;

 res:integer; info:bmpinfo;

 x,y:integer; b,bh,bl:byte;

 nb,np:integer; tpcolor:byte;

 i,j:integer;
begin

 assign(f,fname);

 {$I-} reset (f); {$I+}

 res:=IoResult;

 if res <> 0 then Error (1,fname);

 read (f,info);

 close (f);

 if info.bpp<>4 then Error(2,fname);

 x:=x0;

 y:=y0+info.height;

 nb:=(info.width div 8)*4;

 if (info.width mod 8) <> 0 then nb:=nb+4;

 assign (bmpf,fname);

 reset (bmpf);

 seek (bmpf,info.offset);

 if transparent then begin

  read (bmpf,b);

  tpcolor:=b shr 4;

  seek (bmpf,info.offset);

 end

 else tpcolor:=17;

 for i:=1 to info.height do begin

  np:=0;

  for j:=1 to nb do begin

   read (bmpf,b);

   if np<info.width then begin

    bh:=b shr 4;

    if bh <> tpcolor then 
     putpixel (x,y,color[bh]);

    inc (x);

    inc(np);

   end;

   if np<info.width then begin

    bl:=b and 15;

    if bl <> tpcolor then 
     putpixel (x,y,color[bl]);

    inc(x);

    inc(np);

   end;

  end;

  x:=x0;

  dec(y);

 end;

 close (bmpf);

 Draw:=info.height;

end;

var i,j:word;

 b:char;

 r:integer;

begin

 InitMe;

 clearscreen;

 assign (List,FileList);

 {$I-}

 reset (List);

 {$I+}

 if IoResult <> 0 then Error (1,FileList);

 assign (res,resfile);

 {$I-}

 rewrite (res);

 {$I+}

 if IoResult <> 0 then Error (1,resfile);

 settextjustify (centertext,toptext);

 while not eof(List) do begin

  readLn (List,s);

  clearscreen;

  Draw (0,0,s,true);

  for j:=1 to height do

  for i:=1 to width do begin

   b:=chr(getpixel (i,j));

   write (res,b);

  end;

  setcolor (BLACK);

  outtextxy (cx,maxy-20,'Файл '+s+' ОК');

  Wait;

 end;

 closeMe;

 close (res);

 close (List);

end.
12.2. Листинг содержит исходный текст игры в стиле Invaders. Компилировать в Паскаль 7. При необходимости изменить константу пути к Паскалю. Требует файла ресурсов, созданного утилитой из листинга 12.1. Требует установленного графического шрифта trip.chr.
uses graph,crt,Dos;

const width=32; height=20;

type Picture=array [0..width-1,0..height-1]
  of char;

type sprite=record

 state,x,y,Pnum,PREDir: word;

end;

const VGAPath='c:\TP7\egavga.bgi';

  FontPath='c:\TP7\Trip.chr';

  sprName='attack.res';

const ESC=#27; F1=#59; SPACE=#32; 
 UP=#72; DOWN=#80; LEFT=#75; RIGHT=#77;

const maxx=639; maxy=479;

 cx=maxx div 2; cy=maxy div 2;

 maxsprites=11; maxPictures=11;

 maxshoots=100;

const LeftDir=0; RightDir=1;

 UpDir=2; DownDir=3; 
 Delta=2; shootradius=5;

var ch:char; s:string;

 Hour,min,sec,sec1,secN,secN1,
 sec100,secI,secI1:word;

var Driver, Mode, Font1, 
 currentsprites, currentBottom,

 currentshoots, shootx, Lives,
 Enemyshooter, Enemies,

 shootsProbability: integer;

 score,Level:longint;

 DriverF,FontF: file;

 DriverP,FontP: pointer;

 spr: array [1..maxsprites] of sprite;

 Pict: array [1..maxPictures] of Picture;

 shoots: array [1..maxshoots] of sprite;

 shooter,DieMe,InGame,Initshoot:boolean;

procedure Wait;

var ch:char;

begin

 reset (Input); repeat until keyPressed;

 ch:=readkey; if ch=#0 then readkey;

end;

procedure closeAll;

begin

 if FontP <> nil then begin

  FreeMem(FontP, Filesize(FontF));

  close (FontF);

 end;

 if DriverP <> nil then begin

  FreeMem(DriverP, Filesize(DriverF));

  close (DriverF);

 end;

 closegraph;

end;

procedure graphError;

begin

 closeAll;

 writeln('graphics error:',
  grapherrormsg(graphresult));

 writeln('Press any key to halt');

 Wait; halt (graphresult);

end;

procedure InitAll;

begin

 assign(DriverF, VGAPath);

 reset(DriverF, 1);

 getmem(DriverP, Filesize(DriverF));

 Blockread(DriverF, DriverP^,
   Filesize(DriverF));

 if registerBGIdriver(DriverP)<0 then
   graphError;

 Driver:=VGA; Mode:=VGAHi;

 initgraph(Driver, Mode,'');

 if graphresult < 0 then graphError;

 assign(FontF, FontPath);

 reset(FontF, 1);

 getmem(FontP, Filesize(FontF));

 Blockread(FontF, FontP^, Filesize(FontF));

 Font1:=registerBGifont(FontP);

 if Font1 < 0 then graphError;

end;

procedure clearscreen;

begin

 setfillstyle (solidfill, White);

 bar (0,0,maxx,maxy);

end;

procedure Window
  (x1,y1,x2,y2,color,Fillcolor:integer);

begin

 setcolor (color);

 setfillstyle (1,Fillcolor);

 bar (x1,y1,x2,y2);

 rectangle (x1+2,y1+2,x2-2,y2-2);

 rectangle (x1+4,y1+4,x2-4,y2-4);

 setfillstyle (1,DArKGrAy);

 bar (x1+8,y2+1,x2+8,y2+8);

 bar (x2+1,y1+8,x2+8,y2);

end;

procedure outtextcxy (y:integer; s:string);

begin

 settextjustify (centertext,centertext);

 outtextxy (cx ,y,s);

end;

procedure start;

begin

 clearscreen;

 Window (10,10,maxx-10,maxy-10,Blue,White);

 settextstyle(Font1, HorizDir, 4);

 outtextcxy (25,'Атака из космоса');

 settextstyle(Font1, HorizDir, 1);

 outtextcxy (maxy-25,
  'Нажмите клавишу для начала');

 Wait;

end;

procedure restorescreen
  (sNum,Dir,Delta:word);

var x,y:word;

begin

 x:=spr[sNum].x; y:=spr[sNum].y;

 setfillstyle (solidfill,White);

 case Dir of

  LeftDir: begin

   bar(x+width-Delta,y,x+width-1,
       y+height-1);

  end;

  RightDir: begin

   bar (x,y,x+Delta,y+height-1);

  end;

  UpDir: begin

   bar (x,y+height-Delta,
        x+width-1,y+height-1);

  end;

  DownDir: begin

   bar (x,y,x+width-1,y+Delta);

  end;

 end;

end;

procedure Drawsprite (sNum:word);

var i,j,x,y,n,b:integer;

begin

 N:=spr[sNum].PNum;

 x:=spr[sNum].x; y:=spr[sNum].y;

 for j:=y to y+height-1 do

 for i:=x to x+width-1 do begin

  b:=ord(Pict[n,i-x,j-y]);

  putpixel(i,j,b);

 end;

end;

procedure GoLeft;

var x,d2:word;

begin

 x:=spr[1].x; d2:=delta*4;

 if x>d2 then begin

  restorescreen (1,LeftDir,d2);

  Dec(spr[1].x,d2); Drawsprite (1);

 end;

end;

procedure GoRight;

var x,d2:word;

begin

 x:=spr[1].x;

 d2:=delta*4;

 if x+width < maxx then begin

  restorescreen (1,RightDir,d2);

  Inc(spr[1].x,d2);

  Drawsprite (1);

 end;

end;

procedure showLives;

begin

 str(Lives,s);

 setfillstyle (solidfill,White);

 setcolor (RED); bar (80,0,110,10);

 outtextxy (82,2,s);

end;

procedure showscore;

begin

 str(score,s);

 setfillstyle (solidfill,White);

 setcolor (Blue); bar (150,0,250,10);

 outtextxy (152,2,s);

end;

procedure showshoots;

begin

 str(currentshoots,s);

 setfillstyle (solidfill,White);

 setcolor (Black); bar (20,0,50,10);

 outtextxy (20,2,s);

end;

procedure showLevel;

begin

 str(Level,s);

 setfillstyle (solidfill,White);

 setcolor (Blue); bar (251,0,350,10);

 outtextxy (253,2,'Level '+s);

end;

procedure shoot;

var i:integer;

begin

 if currentshoots>0 then begin
  for i:=1 to maxshoots do 
  if (sec<>sec1) and (shoots[i].state=0)
  then begin

   Dec(currentshoots);

   showshoots;

   spr[1].PNum:=6; Drawsprite (1);

   GetTime(Hour,min,sec,sec100);

   shootx:=spr[1].x; shooter:=true;

   shoots[i].x:=spr[1].x+ (width div 2);

   shoots[i].y:=spr[1].y - 5;

   shoots[i].PNum:=UpDir;

   shoots[i].state:=1;

   break;

  end;

 end;

end;

procedure Help(s:string);

begin

 setfillstyle (solidfill,White);

 setcolor (Blue);
 bar (10,maxy-10,maxx-10,maxy);

 outtextxy (10,maxy-9,s);

end;

procedure Error (code:integer; str:string);

begin

 Window (cx-120,cy-100,cx+120,cy-70,
  Black,YELLOW);

 case code of

  1: s:='Файл '+str+' не найден!';

 end;

 settextjustify (Lefttext, toptext);

 settextstyle(DefaultFont, HorizDir, 1);

 outtextxy (cx-116,cy-92,s);

 Wait; closeAll; halt(code);

end;

procedure DrawField;

var i,x,y:integer;

begin

 clearscreen;

 with spr[1] do begin

  state:=1; Pnum:=1;

  x:=maxx div 2;

  y:=maxy - 10 - height;

  Drawsprite (1);

 end;

 x:=100;

 y:=10;

 for i:=2 to currentsprites do begin

  spr[i].state:=1;

  spr[i].PNum:=7;

  spr[i].x:=x; spr[i].y:=y;

  Drawsprite (i);

  inc(x,50);

  if x>maxx-width then begin

   x:=100;

   if y<currentBottom-height then
    Inc(y,height)

   else y:=10;

  end;

 end;

 for i:=1 to maxshoots do 
  shoots[i].state:=0;

 shooter:=false;

 Enemyshooter:=-1;

 sec:=0; secN:=0;

 secI1:=100; sec1:=100; secN1:=100;

 setfillstyle (solidfill,RED);

 FillEllipse (10,5,5,4);

 showshoots;

 setfillstyle (solidfill,Green);

 bar (60,1,72,10);

 setfillstyle (solidfill,LightGreen);

 bar (62,3,70,8);

 showLives;

 setfillstyle (solidfill,YELLOW);

 setcolor (Black);

 for i:=1 to 3 do begin

  circle (126+i*2,5,4);

  FillEllipse (126+i*2,5,4,4);

 end;

 showscore;

 showLevel;

 InGame:=true;

end;

procedure Loadsprites;

var F:text;

 n,i,j,r:integer;

 b:char;

begin

 assign (f,sprName);

 {$I-}

 reset (f);

 {$I+}

 if IoResult<>0 then Error (1,sprName);

 for n:=1 to maxPictures do

 for j:=0 to height-1 do

 for i:=0 to width-1 do begin

  read (f,b);

  Pict [n,i,j]:=b;

 end;

 close (f);

end;

procedure Deltas (sNum,Dir:integer; 
 var dx,dy:integer);

var x,y:integer;

begin

 x:=spr[sNum].x; y:=spr[sNum].y;

 case Dir of

  LeftDir: begin

   Dec(x,Delta);

   if x<0 then x:=0;

  end;

  RightDir: begin

   Inc(x,Delta);

   if x>maxx-width then x:=maxx-width;

  end;

  UpDir: begin

   Dec (y,Delta);

   if y<10 then y:=10;

  end;

  DownDir: begin

   Inc(y,Delta);

   if y>currentBottom then 
    y:=currentBottom;

  end;

 end;

 dx:=x; dy:=y;

end;

function Between (a,x,b:integer):boolean;

begin

 if (x>a) and (x<b) then Between:=true

 else Between:=false;

end;

procedure shootMovies;

var i,d,n:integer;

 x,y:word;

 found:boolean;

begin

 for i:=1 to maxshoots do 
 if shoots[i].state=1 then begin

  x:=shoots[i].x; y:=shoots[i].y;

  d:=shoots[i].PNum;

  setfillstyle (solidfill,White);

  setcolor (White);
  fillellipse(x,y,shootradius,shootradius);

  if d=updir then begin

   setfillstyle (solidfill,RED);

   if y<15 then begin

    shoots[i].state:=0; continue;

   end;

   found:=false;

   for n:=2 to currentsprites do begin

    if spr[n].state=1 then begin

     if (Between(spr[n].x,x, 
         spr[n].x+width)) and
         (Between(spr[n].y,y,
         spr[n].y+height)) then begin

      shoots[i].state:=0;

      found:=true;

      spr[n].state:=2;

      Inc(spr[n].PNum);

      Inc(score,10+5*n);

      showscore;

      break;

     end;

    end;

   end;

   if not found then Dec(y,Delta);

  end

  else begin

   setfillstyle (solidfill,Blue);

   if y>maxy-10-(height div 2) then begin

    shoots[i].state:=0;

    continue;

   end;

   found:=false;

   if Between(spr[1].x,x,spr[1].x+width)
      and

      Between(spr[1].y,y,spr[1].y+height) 
      then begin

    shoots[i].state:=0; found:=true;

    Inc(spr[1].Pnum); DieMe:=true;

    Help ('you are missed one life :-(');

    Drawsprite (1);

   end;

   if not found then Inc(y,Delta);

  end;

  if not found then begin
  fillellipse(x,y,shootradius,shootradius);

   shoots[i].x:=x; shoots[i].y:=y;

  end;

 end;

end;

procedure Enemiesstep;

var i,k,Dir,dx,dy,n:integer;

begin

 Enemies:=0;

 for i:=2 to currentsprites do begin

  if spr[i].state=1 then begin

   Inc(Enemies);

   for k:=1 to 3 do begin

    dir:=random(4);

    if dir=spr[i].pREDir then break;

   end;

   spr[i].pREDir:=dir;

   Deltas (i, dir, dx, dy);

   restorescreen (i,Dir,Delta);

   spr[i].x:=dx; spr[i].y:=dy;

   Drawsprite (i);

   Initshoot:=false;

   GetTime(Hour,min,secN1,sec100);

   if (secN1<>secN) and
   (1+random(100)<shootsProbability) then 
    Initshoot:=true;

   if Initshoot then begin

    secN:=secN1;

    for n:=1 to maxshoots do

    if (shoots[n].state=0) and 
       (Enemyshooter<>i) then begin

     Enemyshooter:=i;

     shoots[n].x:=dx+ (width div 2);

     shoots[n].y:=dy +height +5;

     shoots[n].PNum:=DownDir;

     shoots[n].state:=1;

     break;

    end;

   end;

  end

  else if spr[i].state=2 then begin

   GetTime (Hour,min,secI,sec100);

   Drawsprite (i);

   if secI<>secI1 then begin

    secI1:=secI;

    if (spr[i].PNum<11) then 
     Inc(spr[i].PNum)

    else begin

     spr[i].state:=0;

     setfillstyle (solidfill, White);

     bar (spr[i].x,spr[i].y,
       spr[i].x+width-1,spr[i].y+height-1);

    end;

   end;

  end;

 end;

end;

procedure Timefunctions;

var i:integer;

begin

 if not InGame then Exit;

 GetTime(Hour,min,sec1,sec100);

 if (shooter) and (sec<>sec1) then begin

  spr[1].PNum:=1;

  if shootx=spr[1].x then Drawsprite (1);

  shooter:=false;

 end;

 if (DieMe) and (sec<>sec1) then begin

  if spr[1].Pnum<5 then begin

   sec:=sec1; Inc(spr[1].PNum);

   Drawsprite (1); DieMe:=true;

  end

  else begin

   DieMe:=false;

   if Lives>0 then begin

    Dec(Lives); showLives;

    spr[1].PNum:=1;

    Drawsprite (1);

   end

   else InGame:=false;

  end;

 end;

end;

function getlongintTime:longint; 
 {Вернет системное время как longint}

var Hour,minute,second,sec100: word;

var k,r:longint;

begin

 GetTime (Hour, minute, second, sec100);

 k:=Hour; r:=k*360000;

 k:=minute; Inc (r,k*6000);

 k:=second; Inc(r,k*100);

 Inc(r,sec100); getlongintTime:=r;

end;

procedure Delay (ms:word); 
var endTime,curTime : longint;

    cor:boolean;

begin

 cor:=false;

 endTime:=getlongintTime + ms div 10;

 if endTime>8639994 then cor:=true;

  repeat

  curTime:=getlongintTime;

  if cor=true then begin

   if curTime<360000 then 
    Inc (curTime,8639994);

  end;

 until curTime>endTime;

end;

label 10,20;

begin

 randomize; InitAll; InGame:=false;

 start;

 settextstyle (DefaultFont,HorizDir,1);

 settextjustify (Lefttext,toptext);

 Loadsprites;

 currentBottom:=200; currentshoots:=50;

 Lives:=3; score:=0; Level:=1;

 shootsProbability:=5;

 currentsprites:=5;

10:

 DrawField;

 if Level>1 then begin

  str(Level-1,s);

  Help ('cool, you''re complete level '+s);

 end

 else Help 
  ('Let''s go! Kill them, invaders!');

 repeat

  if InGame then repeat

   Enemiesstep;

   if Enemies=0 then begin

    Inc(score,100+Level*10);

    if shootsProbability<100 then 
     Inc (shootsProbability);

    if currentsprites<maxsprites then
     Inc(currentsprites);

    if currentBottom<maxy-10-4*height then
     Inc(currentBottom,10);

    currentshoots:=50;

    Delay(1000);{Пауза перед след. уровнем}
    Inc(Level);

    goto 10;

   end;

   shootMovies;

   if not InGame then begin

    Help ('sorry, you''re dead');

   end;

   Timefunctions;

  until keypressed;

  ch:=readkey;

  case ch of

   SPACE: 
    if not DieMe and InGame then shoot;

   #0: begin

    ch:=readkey;

    case ch of

     F1: Help 
      ('Sorry, there''s no help here :-)');

     LEFT: if not DieMe and InGame 
           then GoLeft;

     RIGHT: if not DieMe and InGame 
           then GoRight;

     UP: if not DieMe and InGame 
           then shoot;

    end;

   end;

  end;

 until ch=ESC;

 closeAll;

end.

Приложение 5. Расширенные коды клавиатуры

Нажатие клавиши преобразуется в двухбайтовый код, называемый скан-ASCII-кодом. Этот код помещается в буфер клавиатуры, откуда ваша программа может считать его с помощью прерывания системы BIOS. Старший байт двухбайтового кода называется скан-кодом и является отображением фактически нажатой клавиши. Скан-код не отражает состояние клавиш Shift, Ctrl или Alt и не является уникальным. Помимо скан-кодов нажатия, существуют коды отпускания клавиш, отличающиеся на шестнадцатеричное значение 80 в сторону увеличения. Младший байт полного кода, называемый ASCII-кодом, также не является уникальным, но полная комбинация скан и ASCII-кода уникальна. Некоторые клавиши не имеют ASCII-кода и вместо него возвращается ноль. Такие двухбайтовые коды называются расширенными. При приеме кода нажатой клавиши через DOS последняя отделяет от общего значения скан-код. Кроме того, работающий в системе русификатор может дополнительно транслировать скан-коды буквенных клавиш в ASCII-коды русских букв.

Исходя из сказанного, при использовании стандартной функции readkey, работающей с ASCII-кодами клавиш, в общем случае является правильной следующая схема обработки на Паскале:

ch := readkey; {Чтение символа в байт ch}

if ch = #0 then begin
   {Если нет ASCII-кода, прочитать 
    дополнительно расширенный код}

   ch := readkey;

   {Обработка расширенного кода}

end
else 
   {Обработка ASCII-кода}

На Паскале десятичный код может быть записан в виде #N, где N -- число, например, #65 ('A' латинская). ASCII-коды основных печатных символов можно узнать из Приложения 1, остальные нужные коды приводятся в табл. П5, П6. 
Таблица П5. ASCII-коды некоторых клавиш
	Enter
	13
	Пробел
	32

	Esc
	27
	BackSpace
	8

	Tab
	9
	
	


Таблица П6. Расширенные коды некоторых клавиш

	Клавиша
	Код
	Код с Shift
	Код с Ctrl
	Код с Alt

	F1
	59
	84
	94
	104

	F2
	60
	85
	95
	105

	F3
	61
	86
	96
	106

	F4
	62
	87
	97
	107

	F5
	63
	88
	98
	108

	F6
	64
	89
	99
	109

	F7
	65
	90
	100
	110

	F8
	66
	91
	101
	111

	F9
	67
	92
	102
	112

	F10
	68
	93
	103
	113

	Стрелка вверх
	72
	
	
	

	Стрелка вниз
	80
	
	
	

	Стрелка влево
	75
	
	
	

	Стрелка вправо
	77
	
	
	

	Insert
	82
	
	
	

	Delete
	83
	
	
	

	Home
	71
	
	119
	

	End
	79
	
	117
	

	Page Up
	73
	
	132
	

	Page Down
	81
	
	118
	


Приложение 6. Правила хорошего кода
Написание красивого и эффективного программного кода -- целое искусство, во многом, увы, подзабытое в связи со взрывным ростом мощности вычислительных устройств, вызвавшим снижение требований к качеству алгоритмов. Это небольшое приложение не может заменить изучения специализированных дисциплин вроде "Технологии программирования" или "Теории алгоритмов и формальных языков", но следование изложенным здесь принципам позволит начинающему программисту привыкать не просто "решать задачи", а делать это возможно более красивым и экономичным с точки зрения вычислительных затрат способом.

1. Структурируйте и выравнивайте код, по крайней мере, так, как сказано в гл. 5. Во всем пособии я тоже форматировал листинги в привычном для себя стиле. Лучше привыкнуть структурировать текст, сдвигая любые вложения и разветвления кода одним-двумя пробелами вправо. Программа при этом не "разъедется" далеко вправо на сложных блоках, а вид текста, открытого в любом редакторе, не будет зависеть от размера отступа табуляции.

2. Давайте переменным осмысленные имена. Переменная с именем Length, или, в крайнем случае, Dlina, сама напомнит о своем назначении, в отличие от L. С другой стороны, не возбраняется использовать стандартные сокращения -- например, S для площади, P для периметра, a, b и c -- для сторон треугольника. Любые индексы естественно выглядят с именами i, j, k и т. д. Но если индекс обозначает номер месяца в году, куда естественней назвать его month, чем i. Хотя Паскаль и не различает регистр букв в именах переменных и служебных словах -- соблюдайте его везде. Большинство профессиональных языков регистр символов различают.

3. Существует множество соглашений об именах переменных -- можно спорить об их достоинствах и недостатках, но бесспорно одно -- соблюдение единообразного стиля именования намного облегчает понимание и модификацию программы. В сложных проектах осмысленных имен переменных может оказаться недостаточно, тогда на помощь придут префиксы. Так, если все имена всех переменных, относящихся к таблице "Студенты", начинаются на st_, а все динамические указатели имеют в имени префикс ptr_ (от англ. "pointer" -- указатель), читать такую программу будет намного проще.
4. Создавая любую переменную, обратите внимание на следующие моменты: 
· какой тип значений может принимать переменная, нельзя ли заменить ее перечислением, множеством или иным "сокращенным" типом данных? 
· есть ли ограничения на допустимые значения, если да, где и как они будут учтены? 
· что произойдет при переполнении значения или попытке дать переменной недопустимое значение?

5. Закрывайте блоки сразу. Такой блок, как

if условие then begin
end

else begin

end;

или
while условие do begin

end;

пишется сразу, а только потом ветвь алгоритма или тело цикла наполняются содержимым. Это поможет не запутаться в сложном коде и облегчит соблюдение следующего принципа.

6. Не оставляйте неработающее приложение "на завтра". Блочно-модульная структура программы позволяет всегда избежать этого. Подпрограмма может быть пустой "заглушкой", вы можете использовать ничего не делающие условия, пустые блоки, комментарии, но текущий код должен компилироваться, если завтра вы не хотите половину рабочего дня затратить на восстановление в памяти недоделанного сегодня.

7. Доводите программу до отсутствия предупреждений компилятора, а не только ошибок. Неизвестно, как скажутся на самом деле эти "невинные" напоминания. В языке Си конструкция вида if a:=0 допустима и вызовет лишь предупреждение "Possibly incorrect assignment" -- хотя в результате переменная a всегда будет получать значение 0 и ветвь алгоритма, привязанная к этому условию, будет всегда выполняться.

8. Выбирайте более короткие типы данных там, где это уместно: часто byte может заменить word или integer, а string[20]  -- просто string.

9. Применяйте возможно более эффективный алгоритм решения -- прежде всего, оценка эффективности связана с зависимостью числа выполняемых операций от размерности данных. Двойной цикл полной обработки всех элементов матрицы прост в изучении, но далеко не всегда является лучшим решением при работе с реальной задачей -- ведь трудоемкость этого алгоритма равна n2, где n -- размерность матрицы.

10. Выбирайте менее трудоемкие операции. Так, n div k лучше, чем Trunc(n/k), а Inc(i); лучше, чем i:=i+1;. Во всех случаях порядковые операторы и операнды работают быстрее, чем вещественные. Поэтому обходитесь порядковыми данными везде, где это возможно. Особенно избегайте без необходимости деления на вещественные числа. 
11. Не забывайте о погрешностях при работе с вещественными числами. Хрестоматийное while x<=2.5 do ... -- плохо, если x -- вещественный. С другой стороны, while abs(x-2.5)<eps выглядит громоздко и требует лишних вычислений. Лучше всего while x<=2.5+eps, оптимизирующий компилятор все равно преобразует 2.5+eps в константу.

12. Используйте стандартные функции языка, на котором пишете. Мнимая экономия ресурсов, достигнутая написанием собственных подпрограмм нижнего уровня, обернется трудноуловимыми ошибками в больших проектах. После пяти-десяти лет практики это становится ясно каждому программисту, хотя эксперименты с собственными интерфейсами и машинно-ориентированными программами бывают интересны и даже полезны на этапе обучения.

13. Следите за условиями. Если вы проверяете одно и то же условие неоднократно -- скорей всего, у вашей программы не в порядке с логикой. Пример из п. 7.7 показывает это наглядно.

14. Не забывайте о взаимоисключающих условиях. Составной условный оператор if ... else if или же case в таких случаях намного лучше набора коротких условных операторов.

15. Зачастую при написании длинных фрагментов кода удобнее обрабатывать ошибки в виде

if ошибка then завершение;

обработка;

чем по схеме

if верно then обработка

else завершение;

Вообще, избегайте else, находящихся строк через 100 после своего if -- это затрудняет восприятие даже хорошо структурированной программы.

16. Избегайте в циклах вычислений, не зависящих от их параметров! Выражение вроде sin(Pi/n), помещенное в цикл, где n не меняется, выглядит нелепо. Ведь каждое вычисление синуса (как и других стандартных функций) -- это трудоемкое разложение в ряд Фурье, выполняемое машиной.

17. Используйте математику там, где это уместно для сокращения трудоемкости кода и числа сравнений. Проверить, что переменные x и y имеют один знак, можно так: 
if (x>0) and (y>0) or (x<0) and (y<0) then ..., 
а можно и в виде if (x*y>0) then ....
18. Прекращайте циклы, когда результат уже достигнут. Приоритет средств при этом следующий:

· использование циклов repeat-until или while-do вместо for;

· операторы break или exit;

· в последнюю очередь -- goto, и только в случаях, описанных в п. 16.2.

19. Временный "рабочий" массив того же порядка, что создаваемый или прочитанный из файла -- почти всегда не лучшее решение. Его использование говорит о том, что задача решается "в лоб" и не оптимальным способом. Допустимым считается использование одномерного рабочего массива при обработке матричных данных -- если размерность его не превышает большей из размерностей матрицы.

20. Именуйте размерные константы массивов. Никому не нужны несколько циклов с верхними границами-"близнецами". А что, если размерность обрабатываемых данных придется изменить?

21. Передавайте значения подпрограммам преимущественно по адресу, а не по значению. Для матричных и векторных данных старайтесь делать это всегда. Применение векторных данных имеет приоритет перед матричными.

22. Не делайте подпрограммы зависимыми от глобальных данных. Несоблюдение этого правила существенно уменьшит вероятность повторного использования кода вами или другим разработчиком.

23. Не пишите подпрограмм, возвращающих более одного объекта -- скаляра, вектора или матрицы. В крайнем случае, можно отдельным параметром передавать или возвращать размерность векторных данных. Избегайте подпрограмм, которые ничего не возвращают. Разработка сложных подпрограмм облегчается, если их "точка выхода" и возвращаемое значение указаны единственным и последним оператором. Для перехода из тела подпрограммы в точку возврата в этом случае не грешно использовать даже goto:

function Test (a,b:integer):integer;

label end_of_Test;

var error:integer;

begin

 error:=0;

 if (a<0) or (b<0) then begin

  error:=1;

  goto end_of_Test;

 end;
 . . .

end_of_Test:

 Test:=error;

end;

24. Нужно приучить себя не только правильно называть переменные и функции, но и выделять все логические части задачи при проектировании иерархии подпрограмм. Например, если функция ищет в наборе данных элемент по какому-то признаку и создает новый в том случае, когда искомый элемент обнаружен, то лучше разделить ее на две -- функцию поиска и функцию создания нового узла, которая будет вызываться из первой. Достаточно придерживаться подобных незамысловатых приемов, чтобы повысить сопровождаемость приложения в разы. Если посмотреть на вопрос с другой стороны, не заставляйте одну подпрограмму выполнять несколько функций -- она окажется бесполезна с точки зрения повторного использования кода. Так, определение длины ломаной линии можно реализовать одной подпрограммой, но гораздо лучше, если задача разбита на подпрограммы вычисления длины и вычисления расстояния между двумя точками.

25. При работе с динамическими объектами пишите код так, чтобы открытые объекты всегда закрывались, как только они станут не нужны. В идеале порядок закрытия объектов должен быть обратным по отношению к порядку открытия (последний занявший память объект освобождает ее первым). Следует также избегать функций перераспределения ранее выделенной динамической памяти.

26. Проверить логику своей программы легче всего по принципу "Одна правка -- одно место в программе". Если при добавлении в меню нового пункта или, еще хуже, простом изменении размерности одномерного массива приходится переписывать несколько удаленных друг от друга фрагментов кода -- программа написана плохо.

27. Если написанная вами программа не работает или работает "криво", ошибка лежит на вашей совести, а компьютер с компилятором ни в чем не виноваты. "Отладка", при которой программист хаотически меняет то одно, то другое место в коде и на которую уходит до 90% времени написания, на самом деле -- свидетельство не слишком качественной работы. Хорошо написанной программе нужна не столько отладка, сколько тестирование на различных допустимых, недопустимых и "пограничных" наборах данных. Кстати,  обдумывание и написание тестов до тестируемого кода способствует и улучшению, и большей устойчивости конечного продукта.

Резюмируя, можно сказать, что все большие программы на свете написаны только за счет одного -- правильного структурирования кода. И хороший программист -- не тот, кто "знает языки", а тот, кто умеет писать легко читаемый, понимаемый и модифицируемый код, максимально используя стандартные средства языка разработки.

